
目 次

1 変更記録

• 2016/12/22

– 作成および初期リリース (FDPS バージョン 3.0として)

• 2018/07/11

– 第 4節の以下の記述の修正・改善

∗ ユーザ定義型のソースコードの一部が端切れしていた (第 4.1節, 第 4.2節)

∗ 一部のディレクトリ名に誤植

• 2018/08/29

– N 体/SPHサンプルコードの解説を追加 (第 7.1節)

• 2018/08/31

– x86用 Phantom-GRAPEライブラリの節を追加 (第 3.4.1.7節)

• 2019/07/19

– N 体/SPHサンプルコードの記述を改訂 (第 7.1節)

• 2020/08/16

– PIKGの利用法を追加 (第 3.4.1.8節)

• 2020/08/28

– N 体/SPHコード (第 7.1節) で使用可能な初期条件ファイルの配布先情報を変更

• 2020/09/02

– ツリーオブジェクトを初期化するAPI init tree の第一引数の説明に誤解を招く表
現があったため修正。サンプルコードも合わせて修正。

2 概要

本節では、Framework for Developing Particle Simulator (FDPS) および FDPS Fortran

インターフェース の概要について述べる。FDPSは粒子シミュレーションのコード開発を支
援するフレームワークである。FDPSが行うのは、計算コストの最も大きな粒子間相互作用
の計算と、粒子間相互作用の計算のコストを負荷分散するための処理である。これらはマル
チプロセス、マルチスレッドで並列に処理することができる。比較的計算コストが小さく、
並列処理を必要としない処理 (粒子の軌道計算など)はユーザーが行う。
FDPSが対応している座標系は、2次元直交座標系と 3次元直交座標系である。また、境界
条件としては、開放境界条件と周期境界条件に対応している。周期境界条件の場合、x、y、
z軸方向の任意の組み合わせの周期境界条件を課すことができる。
ユーザーは粒子間相互作用の形を定義する必要がある。定義できる粒子間相互作用の形に
は様々なものがある。粒子間相互作用の形を大きく分けると 2種類あり、1つは長距離力、
もう 1つは短距離力である。この 2つの力は、遠くの複数の粒子からの作用を 1つの超粒子
からの作用にまとめるか (長距離力)、まとめないか (短距離力)という基準でもって分類さ
れる。
長距離力には、小分類があり、無限遠に存在する粒子からの力も計算するカットオフなし
長距離力と、ある距離以上離れた粒子からの力は計算しないカットオフあり長距離力がある。
前者は開境界条件下における重力やクーロン力に対して、後者は周期境界条件下の重力や
クーロン力に使うことができる。後者のためには Particle Mesh法などが必要となるが、こ
れは FDPSの拡張機能として用意されている。
短距離力には、小分類が 4つ存在する。短距離力の場合、粒子はある距離より離れた粒子
からの作用は受けない。すなわち必ずカットオフが存在する。このカットオフ長の決め方に
よって、小分類がなされる。すなわち、全粒子のカットオフ長が等しいコンスタントカーネ
ル、カットオフ長が作用を受ける粒子固有の性質で決まるギャザーカーネル、カットオフ長
が作用を与える粒子固有の性質で決まるスキャッタカーネル、カットオフ長が作用を受ける
粒子と作用を与える粒子の両方の性質で決まるシンメトリックカーネルである。コンスタン
トカーネルは分子動力学における LJ力に適用でき、その他のカーネルは SPHなどに適用で
きる。
ユーザーは、粒子間相互作用や粒子の軌道積分などを、 Fortran 2003 を用いて記述する。

3 入門：サンプルコードを動かしてみよう

本節では、まずはじめに、FDPS および FDPS Fortranインターフェース の動作環境、必
要なソフトウェア、インストール方法などを説明し、その後、サンプルコードの使用方法を
説明する。サンプルコードの中身に関しては、次節 (第 4節)で詳しく述べる。

3.1 動作環境

FDPSは Linux, Mac OS X, WindowsなどのOS上で動作する。

3.2 必要なソフトウェア

本節では、FDPSを使用する際に必要となるソフトウェアを記述する。まず標準機能を用
いるのに必要なソフトウェア、次に拡張機能を用いるのに必要なソフトウェアを記述する。

3.2.1 標準機能

本節では、FDPSの標準機能のみを使用する際に必要なソフトウェアを記述する。最初に
逐次処理機能のみを用いる場合（並列処理機能を用いない場合）に必要なソフトウェアを記
述する。次に並列処理機能を用いる場合に必要なソフトウェアを記述する。

3.2.1.1 逐次処理

逐次処理の場合に必要なソフトウェアは以下の通りである。

• make

• C++コンパイラ (gccバージョン 4.8.3以降なら確実, Kコンパイラバージョン 1.2.0で
動作確認済)

• Fortranコンパイラ (Fortran 2003 標準をサポートし、上記 C++コンパイラと相互運
用可能なもの。gcc 4.8.3 以降の gfortranなら確実)

• Python 2.7.5以上、または、Python 3.4以上 (これ以外での正常動作は保証しない。特
に、Python 2.7以前では動作しない)

3.2.1.2 並列処理

本節では、FDPS の並列処理機能を用いる際に必要なソフトウェアを記述する。まず、
OpenMPを使用する際に必要なソフトウェア、次にMPIを使用する際に必要なソフトウェ
ア、最後にOpenMPとMPIを同時に使用する際に必要なソフトウェアを記述する。

3.2.1.2.1 OpenMP

OpenMPを使用する際に必要なソフトウェアは以下の通り。

• make

• OpenMP対応の C++コンパイラ (gcc version 4.8.3以降なら確実, Kコンパイラバー
ジョン 1.2.0で動作確認済)

• OpenMP対応のFortranコンパイラ (Fortran 2003 標準をサポートし、上記 C++コン
パイラと相互運用可能なもの。gcc 4.8.3以降なら確実)

• Python 2.7.5以上、または、Python 3.4以上 (これ以外での正常動作は保証しない。特
に、Python 2.7以前では動作しない)

3.2.1.2.2 MPI

MPIを使用する際に必要なソフトウェアは以下の通り。

• make

• MPI version 1.3対応のC++コンパイラ (Open MPI 1.6.4で動作確認済, K コンパイラ
バージョン 1.2.0で動作確認済)

• MPI version 1.3対応のFortranコンパイラ (Fortran 2003標準をサポートし、上記 C++

コンパイラと相互運用可能なもの。Open MPI 1.6.4で動作確認済み)

• Python 2.7.5以上、または、Python 3.4以上 (これ以外での正常動作は保証しない。特
に、Python 2.7以前では動作しない)

3.2.1.2.3 MPI+OpenMP

MPIとOpenMPを同時に使用する際に必要なソフトウェアは以下の通り。

• make

• MPI version 1.3とOpenMPに対応のC++コンパイラ (Open MPI 1.6.4で動作確認済,

Kコンパイラバージョン 1.2.0で動作確認済)

• MPI version 1.3とOpenMPに対応の Fortranコンパイラ (Fortran 2003標準をサポー
トし、上記 C++コンパイラと相互運用可能なもの。Open MPI 1.6.4で動作確認ずみ)

• Python 2.7.5以上、または、Python 3.4以上 (これ以外での正常動作は保証しない。特
に、Python 2.7以前では動作しない)

3.2.2 拡張機能

本節では、FDPSの拡張機能を使用する際に必要なソフトウェアについて述べる。FDPS

の拡張機能にはParticle Meshがある。以下ではParticle Meshを使用する際に必要なソフト
ウェアを述べる。

3.2.2.1 Particle Mesh

Particle Meshを使用する際に必要なソフトウェアは以下の通りである。

• make

• MPI version 1.3とOpenMPに対応のC++コンパイラ (Open MPI 1.6.4で動作確認済)

• FFTW 3.3以降

3.3 インストール

本節では、FDPS および FDPS Fortranインターフェース のインストールについて述べ
る。取得方法、ビルド方法について述べる。

3.3.1 取得方法

ここではFDPSの取得方法を述べる。最初に最新バージョンの取得方法、次に過去のバー
ジョンの取得方法を述べる。

3.3.1.1 最新バージョン

以下の方法のいずれかで FDPSの最新バージョンを取得できる。

• ブラウザから

1. ウェブサイト https://github.com/FDPS/FDPSで”Download ZIP”をクリックし、
ファイル FDPS-master.zipをダウンロード

2. FDPSを展開したいディレクトリに移動し、圧縮ファイルを展開

• コマンドラインから

– Subversionを用いる場合：以下のコマンドを実行するとディレクトリ trunkの下
を Subversionレポジトリとして使用できる� �
$ svn co --depth empty https://github.com/FDPS/FDPS

$ cd FDPS

$ svn up trunk� �
10

– Gitを用いる場合：以下のコマンドを実行するとカレントディレクトリにディレ
クトリ FDPSができ、その下をGitのレポジトリとして使用できる� �
$ git clone git://github.com/FDPS/FDPS.git� �

3.3.1.2 過去のバージョン

以下の方法でブラウザから FDPSの過去のバージョンを取得できる。

• ウェブサイト https://github.com/FDPS/FDPS/releasesに過去のバージョンが並ん
でいるので、ほしいバージョンをクリックし、ダウンロード

• FDPSを展開したいディレクトリに移動し、圧縮ファイルを展開

3.3.2 インストール方法

C++言語で記述されたFDPS本体はヘッダライブラリ注 1)のため、configureなどを行う
必要はない。基本的にはアーカイブを展開したあと、自分のソースファイルをコンパイルす
る時に適切なインクルードパスを設定すればよい。実際の手続きは第 3.4節で説明するサン
プルコードとその Makefile をみて欲しい。
Fortranの場合、コンパイル前に Fortranソースファイルから FDPSとのインターフェー
スコードを生成する必要がある。その手順は仕様書 doc spec ftn ja.pdfの第 6章に記述され
ている。本サンプルコードのMakefileでは、インターフェースコードが make コマンド実行
中に自動的に生成されるようになっている。ユーザが自分のコードの Makefile を作る時に
はサンプルコードの Makefile を参考にすることを推奨する。

3.4 サンプルコードの使用方法

本節ではサンプルコードの使用方法について説明する。サンプルコードには重力N 体シ
ミュレーションコードと、SPHシミュレーションコードがある。最初に重力N体シミュレー
ションコード、次に SPHシミュレーションコードの使用について記述する。サンプルコード
は拡張機能を使用していない。

3.4.1 重力N 体シミュレーションコード

本サンプルコードは、 FDPS Fortranインターフェース を用いて書かれた無衝突系のN

体計算コードである。このコードでは一様球のコールドコラプス問題を計算し、粒子分布の
スナップショットを出力する。

注 1)ヘッダファイルだけで構成されるライブラリのこと

3.4.1.1 概要

以下の手順で本コードを使用できる。

• ディレクトリ$(FDPS)/sample/fortran/nbodyに移動。これ以後、ディレクトリ$(FDPS)

はFDPSの最も上の階層のディレクトリを指す ($(FDPS)は環境変数にはなっていない)。
$(FDPS)はFDPSの取得によって異なり、ブラウザからならFDPS-master, Subversion

からなら trunk, Gitからなら FDPSである。

• カレントディレクトリにある Makefileを編集

• コマンドライン上で makeを実行

• nbody.out ファイルの実行

• 結果の解析

最後に x86版 Phantom-GRAPEおよび PIKGを使う場合について述べる。

3.4.1.2 ディレクトリ移動

ディレクトリ$(FDPS)/sample/fortran/nbodyに移動する。

3.4.1.3 Makefileの編集

サンプルコードのディレクトリには 2つの Makefileがある。1つは GCC用に書かれた
Makefileであり、もう 1つは Intelコンパイラ用に書かれた Makefile.intelである。ここ
では Makefileについて詳しく解説し、Makefile.intelに関しては使用上の注意点を本節
最後で述べるのみとする。
まず、Makefileの初期設定について説明する。サンプルコードをコンパイルするにあたっ
て、ユーザが設定すべきMakefile変数は 4つあり、Fortranコンパイラを表す FC、C++コン
パイラを表す CXX、それぞれのコンパイルオプションを表す FCFLAGS, CXXFLAGSである。こ
れらの初期設定値は次のようになっている:� �
FC=gfortran

CXX=g++

FCFLAGS = -std=f2003 -O3 -ffast-math -funroll-loops -finline-functions

CXXFLAGS = -O3 -ffast-math -funroll-loops $(FDPS_INC)� �
ここで、$(FDPS INC)はFDPS本体をインクルードするために必要なインクルードPATHが
格納された変数であり、Makefile内で設定済みである。したがって、ここで変更する必要
はない。
上記 4つのMakefile変数の値を適切に編集し、makeコマンドを実行することで実行ファ
イルが得られる。OpenMPとMPIを使用するかどうかで編集方法が変わるため、以下でそ
れを説明する。

• OpenMPもMPIも使用しない場合

– 変数 FCに Fortranコンパイラを代入する

– 変数 CXXにC++コンパイラを代入する

• OpenMPのみ使用の場合

– 変数 FCにOpenMP対応の Fortranコンパイラを代入する

– 変数 CXXにOpenMP対応のC++コンパイラを代入する

– FCFLAGS += -DPARTICLE SIMULATOR THREAD PARALLEL -fopenmpの行のコメン
トアウトを外す

– CXXFLAGS += -DPARTICLE SIMULATOR THREAD PARALLEL -fopenmpの行のコメン
トアウトを外す

• MPIのみ使用の場合

– 変数 FCにMPI対応の Fortranコンパイラを代入する

– 変数 CXXにMPI対応のC++コンパイラを代入する

– FCFLAGS += -DPARTICLE SIMULATOR MPI PARALLELの行のコメントアウトを外す

– CXXFLAGS += -DPARTICLE SIMULATOR MPI PARALLELの行のコメントアウトを外
す

• OpenMPとMPIの同時使用の場合

– 変数 FCにMPI対応の Fortranコンパイラを代入する

– 変数 CXXにMPI対応のC++コンパイラを代入する

– FCFLAGS += -DPARTICLE SIMULATOR THREAD PARALLEL -fopenmpの行のコメン
トアウトを外す

– FCFLAGS += -DPARTICLE SIMULATOR MPI PARALLELの行のコメントアウトを外す

– CXXFLAGS += -DPARTICLE SIMULATOR THREAD PARALLEL -fopenmpの行のコメン
トアウトを外す

– CXXFLAGS += -DPARTICLE SIMULATOR MPI PARALLELの行のコメントアウトを外
す

次に、ユーザが本Makefileをユーザコードで使用する場合に便利な情報を記述する。ユー
ザコードで使用する場合に最も重要となるMakefile変数は、FDPS LOC,

SRC USER DEFINED TYPE, SRC USER の 3つである。まず、変数 FDPS LOCには、FDPSのトッ
プディレクトリのPATHを格納する。本Makefileでは、FDPSのソースディレクトリのPATH

やFortranとのインターフェースコードを生成するスクリプトのPATH等、FDPSに関連す
る各種な設定がこの変数の値に基いて自動的に設定されるようになっている。したがって、

ユーザは適切に設定する必要がある。次に、変数 SRC USER DEFINED TYPE, SRC USERには、
それぞれ、ユーザ定義型が記述された Fortranファイル名と、ユーザ定義型以外の部分が記
述された Fortranファイル名を格納する。FDPS の Fortran インターフェースコードはユー
ザーコードのクラス (派生型)を記述する部分から生成されるので、その部分が記述された
ファイルを SRC USER DEFINED TYPEで、それ以外を SRC USERで指定する。これにより、SRC
USERで指定したファイルが変更されても FDPS の再コンパイルは起きなくなるので、コン
パイル・リンクの時間が短くなる。但し、SRC USER DEFINED TYPE、或いは、SRC USERに格
納された (複数の)ファイルの間に依存関係がある場合、依存関係を示すルールをMakefileに
追記しなければならない点に注意して頂きたい。この記述方法に関しては、例えば、GNU

make のマニュアル等を読んで頂きたい。
最後に、Makefile.intelを使用する上での注意点について説明する。変数の初期値が異
なる点を除き、Makefile.intelの構造は Makefileと同じである。したがって、変数の値
をユーザが利用する計算機システムにおける値に適切に設定すれば、Makefileと同様に利
用可能である。以下に変更する上での注意点を述べる:

• /opt/intel/binを、利用する計算機システムにおける Intelコンパイラの格納ディレクト
リの PATHに変更する。

• /opt/intel/includeを、Intelコンパイラに付属するヘッダファイル群を格納したディレ
クトリの PATHに変更する。

• Makefile.intelの LDFLAGSは、-L/opt/intel/lib/intel64 -L/usr/lib64 -lifport

-lifcore -limf -lsvml -lm -lipgo -lirc -lirc sとなっている。
この中の -lifcore 注 2)は、C++コンパイラでC++オブジェクトとFortranオブジェクトを
リンクするため必要である注 3)。計算機システムのライブラリパスに、Intelコンパイラのラ
イブラリ群が登録されていない場合、さらに、-L/opt/intel/lib/intel64 -L/usr/lib64

-lifport -limf -lsvml -lm -lipgo -lirc -lirc s のような指定が必要である。
ここで、/opt/intel/lib/intel64は、Intelコンパイラのライブラリ群が格納されたディ
レクトリの PATHで、/usr/lib64はライブラリ libmを格納したディレクトリの PATH

である。これらは利用する計算機システムに合わせて修正する必要がある。コンパイルに
必要なライブラリ群 (-l*)は、Intelコンパイラのバージョンによって変わる可能性がある
ので確認して頂きたい。

• 本書を執筆時点 (2016/12/26)で、Intelコンパイラで OpenMPを有効にするオプション
は-openmp、或いは、-qopenmpである。これは Intelコンパイラのバージョンによって異
なり、より新しいバージョンのコンパイラは後者を使用する (前者を使用した場合、廃止
予定の警告が出る)。

• 利用する計算機システムによっては、-lifcoreの指定以外の設定が環境変数 (PATH, CPATH,

LD LIBRARY PATH等として)で既に行われていることもありえる。

注 2)libifcoreは、Fortran ランタイムライブラリである。
注 3)Intelコンパイラ (バージョン 17.0.0 20160721)において確認。

3.4.1.4 makeの実行

makeコマンドを実行する。 このとき、まず FDPSの Fortranインターフェースプログラ
ムが生成され、その後、インターフェースプログラムとサンプルコードが一緒にコンパイル
される。

3.4.1.5 実行

実行方法は以下の通りである。

• MPIを使用しない場合、コマンドライン上で以下のコマンドを実行する� �
$ ./nbody.out� �

• MPIを使用する場合、コマンドライン上で以下のコマンドを実行する� �
$ MPIRUN -np NPROC ./nbody.out� �
ここで、MPIRUNには mpirunや mpiexecなどが、NPROCには使用するMPIプロセスの
数が入る。

正しく終了すると、以下のようなログを出力する。energy errorは絶対値で 1× 10−3の
オーダーに収まっていればよい。� �
time: 9.5000000000E+000, energy error: -3.8046534069E-003

time: 9.6250000000E+000, energy error: -3.9711750200E-003

time: 9.7500000000E+000, energy error: -3.8223429428E-003

time: 9.8750000000E+000, energy error: -3.8843099298E-003

MemoryPool::finalize() is completed!

******** FDPS has successfully finished. ********� �
3.4.1.6 結果の解析

ディレクトリ resultに粒子分布を出力したファイル ”snap0000x-proc0000y.dat” ができ
ている。ここで x は整数で時刻に対応している。y はMPIプロセス番号を表しており、MPI

実行しなければ常に y=0である。
出力ファイルフォーマットは 1列目から順に粒子の ID, 粒子の質量、位置の x, y, z座標、
粒子の x, y, z軸方向の速度である。
ここで実行したのは、粒子数 1024個からなる一様球 (半径 3)のコールドコラプスである。
コマンドライン上で以下のコマンドを実行すれば、時刻 9における xy平面に射影した粒子
分布を見ることができる。

図 1:

� �
$ cd result

$ cat snap00009-proc* > snap00009.dat

$ gnuplot

> plot "snap00009.dat" using 3:4� �
他の時刻の粒子分布をプロットすると、一様球が次第に収縮し、その後もう一度膨張する
様子を見ることができる (図 1参照)。
粒子数を 10000個にして計算を行いたい場合には、ファイル f main.F90の中のサブルー
チン f main()のパラメータ変数 ntotを 10000に設定し、再度、コンパイルした上で実行す
ればよい。

3.4.1.7 x86版Phantom-GRAPEを使う場合

Phantom-GRAPEはSIMD命令を効率的に活用することで重力相互作用の計算を高速に実
行するライブラリである (詳細はTanikawa et al.[2012, New Astronomy, 17, 82] とTanikawa

et al.[2012, New Astronomy, 19, 74]を参照のこと)。

まず、使用環境を確認する。SIMD命令セットAVXをサポートする Intel CPUまたはAMD

CPUを搭載したコンピュータを使用しているならば、x86版 Phantom-GRAPEを使用可能
である。
次にディレクトリ$(FDPS)/src/phantom grape x86/G5/newton/libpg5に移動して、ファ
イルMakefileを編集し、コマンドmakeを実行してPhantom-GRAPEのライブラリlibpg5.a

を作る。
最後に、ディレクトリ$(FDPS)/sample/fortran/nbodyに戻り、ファイル Makefile内の

’’#use phantom grape x86 = yes’’の’’#’’を消す。makeを実行してコンパイルする (OpenMP,

MPIの使用・不使用どちらにも対応)と、x86版Phantom-GRAPEを使用したコードができ
ている。上と同様の方法で実行・結果の確認を行うとさきほどと同様の結果が得られる。
Intel Core i5-3210M CPU @ 2.50GHz の 2コアで性能テスト (OpenMP使用、MPI不使
用)をした結果、粒子数 8192の場合に、Phantom-GRAPEを使うと、使わない場合に比べ
て、最大で 5倍弱ほど高速なコードとなる。

3.4.1.8 PIKGを使う場合

PIKGはDSL(ドメイン固有言語)記述から様々なアーキテクチャ向けに最適化された粒子
間相互作用カーネルを生成するカーネルジェネレータである．
ディレクトリ$(FDPS)/sample/fortran/nbodyのファイル Makefile内の

’’#use pikg x86 = yes’’の’’#’’を消す。 この状態で makeを実行してコンパイルする
(OpenMP, MPIの使用・不使用どちらにも対応)と、PIKGから生成されたカーネルを使用
した実行ファイル nbody.outができている。 上と同様の方法で実行・結果の確認を行うと
さきほどと同様の結果が得られる。
デフォルトでは，通常の Fortranコードと同等の referenceモードの相互作用カーネル
が生成される．Makefile内の#COVERSION TYPEとその直後にある #CXXFLAGS の行のコメ
ントアウトを外すと，別のアーキテクチャ(AVX2やAVX-512)向けにコードを生成できる．
AVX2モードを使う場合は，利用するCPUがAVX2及びFMAに対応していなくてはならな
い．さらにAVX-512モードを使う場合には利用するCPUがAVX-512FおよびAVX-512DQ

に対応していなくてはならない．

3.4.2 SPHシミュレーションコード

本サンプルコードには標準SPH法がFDPSを使って実装されている。簡単のため、smooth-

ing lengthは一定値を取ると仮定している。コードでは、3次元の衝撃波管問題の初期条件
を生成し、衝撃波管問題を実際に計算する。

3.4.2.1 概要

以下の手順で本コードを使用できる。

• ディレクトリ$(FDPS)/sample/fortran/sphに移動

• カレントディレクトリにある Makefileを編集 (後述)

• コマンドライン上で makeを実行

• sph.outファイルの実行 (後述)

• 結果の解析 (後述)

3.4.2.2 ディレクトリ移動

ディレクトリ$(FDPS)/sample/fortran/sphに移動する。

3.4.2.3 Makefileの編集

SPHサンプルコードにも、N 体計算のサンプルコードの場合と同様、GCCと Intelコン
パイラ用に 2種類のMakefileが用意されている。編集の仕方は、N体計算の場合と同一なの
で、 第 3.4.1.3節を参照されたい。

3.4.2.4 makeの実行

makeコマンドを実行する。N体計算のときと同様、このとき、まずFDPSのFortranイン
ターフェースプログラムが生成され、その後、インターフェースプログラムとサンプルコー
ドが一緒にコンパイルされる。

3.4.2.5 実行

実行方法は以下の通りである。

• MPIを使用しない場合、コマンドライン上で以下のコマンドを実行する� �
$ ./sph.out� �

• MPIを使用する場合、コマンドライン上で以下のコマンドを実行する� �
$ MPIRUN -np NPROC ./sph.out� �
ここで、MPIRUNには mpirunや mpiexecなどが、NPROCには使用するMPIプロセスの
数が入る。

正しく終了すると以下のようなログを出力する。� �
******** FDPS has successfully finished. ********� �

3.4.2.6 結果の解析

実行するとディレクトリ resultにファイルが出力されている。ファイル名は”snap0000x-

proc0000y.dat” となっている。ここで、x,yは整数で、それぞれ、時刻とMPIプロセス番号
を表す。MPI実行でない場合には、常に y=0である。 出力ファイルフォーマットは 1列目
から順に粒子の ID、粒子の質量、位置の x, y, z座標、粒子の x, y, z軸方向の速度、密度、内
部エネルギー、圧力である。
コマンドライン上で以下のコマンドを実行すれば、横軸に粒子の x座標、縦軸に粒子の密
度をプロットできる (時刻は 40)。� �
$ cd result

$ cat snap00040-proc* > snap00040.dat

$ gnuplot

> plot "snap00040.dat" using 3:9� �
正しい答が得られれば、図 2のような図を描ける。

 0

 0.5

 1

 0  0.5  1

ρ

x

図 2: 衝撃波管問題の時刻 t = 40における密度分布

4 サンプルコードの解説

本節では、前節 (第 3節)で動かしたサンプルコードについての解説を行う。特に、ユーザ
が定義しなければならない派生データ型 (以後、ユーザ定義型と呼ぶ)や FDPSの各種API

の使い方について詳しく述べる。説明の重複を避けるため、いくつかの事項に関しては、そ

の詳細な説明がN体シミュレーションコードの節でのみ行われている。そのため、SPHシ

ミュレーションだけに興味があるユーザも、N 体シミュレーションコードの節に目を通して
頂きたい。

4.1 N体シミュレーションコード

4.1.1 ソースファイルの場所と構成

ソースファイルは$(FDPS)/sample/fortran/nbody以下にある。 サンプルコードは、次
節で説明するユーザ定義型が記述されたソースコード user_defined.F90 と、N 体シミュ
レーションのメインループ等が記述されたソースコード f_main.F90から構成される。この
他に、GCCと Intelコンパイラ用のMakefileである Makefileと Makefile.intelがある。

4.1.2 ユーザー定義型・ユーザ定義関数

本節では、FDPSの機能を用いてN体計算を行う際、ユーザーが記述しなければならない
派生データ型と サブルーチンについて記述する。

4.1.2.1 FullParticle型

ユーザーはユーザ定義型の 1つFullParticle型を記述しなければならない。FullParticle型
には、シミュレーションを行うにあたって、N体粒子が持っているべき全ての物理量が含まれ
ている。Listing 1に本サンプルコードのFullParticle型の実装例を示す ( user defined.F90

を参照)。

Listing 1: FullParticle型
1 type , public , bind(c) :: full_particle !$fdps FP ,EPI ,EPJ ,Force

2 !$fdps copyFromForce full_particle (pot ,pot) (acc ,acc)

3 !$fdps copyFromFP full_particle (id ,id) (mass ,mass) (pos ,pos)

4 !$fdps clear id=keep , mass=keep , pos=keep , vel=keep

5 integer(kind=c_long_long) :: id

6 real(kind=c_double) mass !$fdps charge

7 type(fdps_f64vec) :: pos !$fdps position

8 type(fdps_f64vec) :: vel !$fdps velocity

9 real(kind=c_double) :: pot

10 type(fdps_f64vec) :: acc

11 end type full_particle

FDPS Fortranインターフェースを使ってユーザコードを開発する場合、ユーザは派生デー
タ型がどのユーザ定義型 (FullParticle型, EssentialParticleI型, EssentialParticleJ型, Force

型)に対応するかをFDPSに教えなければならない。本インターフェースにおいて、この指示
は、派生データ型に決まった書式のコメント文を加えることによって行う (以後、この種のコメ
ント文をFDPS指示文と呼ぶ)。本サンプルコードでは、FullParticle型がEssentialParticleI

型、EssentialParticleJ型、そして、Force型を兼ねている。そのため、派生データ型がすべ
てのユーザ定義型に対応すること指示する以下のコメント文を記述している:� �
type, public, bind(c) :: full_particle !$fdps FP,EPI,EPJ,Force� �
また、FDPSはFullParticle型のどのメンバ変数が質量や位置等の必須物理量 (どの粒子計
算でも必ず必要となる物理量、或いは、特定の粒子計算において必要とされる物理量と定義
する)に対応するのかを知っていなければならない。この指示も決まった書式のコメント文
をメンバ変数に対して記述することで行う。今回の例では、メンバ変数 mass, pos, velが、
それぞれ、質量、位置、速度に対応することをFDPSに指示するため、以下の指示文が記述
されている:� �
real(kind=c_double) :: mass !$fdps charge

type(fdps_f64vec) :: pos !$fdps position

type(fdps_f64vec) :: vel !$fdps velocity� �
ただし、メンバ変数が速度であることを指示する !$fdps velocity は予約語であり、指示
は任意である (現時点で FDPSの振舞に一切影響しない)。
FullParticle型はEssentialParticleI型、EssentialParticleJ型、Force型との間でデータの移
動 (データコピー)を行う。ユーザはこのコピーの仕方を指示する FDPS指示文も記述しな
ければならない。本サンプルコードでは、以下のように記述している:� �
!$fdps copyFromForce full_particle (pot,pot) (acc,acc)

!$fdps copyFromFP full_particle (id,id) (mass,mass) (pos,pos)� �
ここで、キーワードcopyFromForceを含む指示文は、Force型のどのメンバ変数をFullParticle

型のどのメンバ変数にコピーするのかを指示するもので、FullParticle型に常に記述しなければ
ならない指示文である。一方、キーワードcopyFromFPはFullParticle型からEssentialParticleI

型およびEssentialParticleJ型へのデータコピーの仕方を指示するもので、EssentialParticleI

型と EssentialParticleJ型には必ず記述しなければならない指示文である。今、FullParticle

型はこれら 2つを兼ねているため、ここに記述している。
今、FullParticle型は Force型を兼ねている。Force型にも必ず記述しなければならない指
示文がある。それは、相互作用計算において、積算対象のメンバ変数をどのように 0クリア
するかを指示する指示文である。本サンプルコードでは、積算対象である加速度とポテン
シャルのみを 0クリアすることを指示するため、次の指示文を記述している:� �
!$fdps clear id=keep, mass=keep, pos=keep, vel=keep� �

ここで、キーワード clearの右に記述された構文 mbr =keep は、メンバ変数 mbr の値を変
更しないことを指示する構文である。
FDPS指示文の書式の詳細については、仕様書 doc_specs_ftn_ja.pdfをご覧頂きたい。

4.1.2.2 相互作用関数 calcForceEpEp

ユーザーは粒子間相互作用の仕方を記述した相互作用関数 calcForceEpEpを記述しなけ
ればならない。サブルーチン calcForceEpEpには、粒子-粒子相互作用計算の具体的な内容
を書く必要がある。Listing 2に、本サンプルコードでの実装を示す ( user defined.F90 を
参照)。

Listing 2: 関数 calcForceEpEp

1 subroutine calc_gravity_ep_ep(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

2 implicit none

3 integer(c_int), intent(in), value :: n_ip ,n_jp

4 type(full_particle), dimension(n_ip), intent(in) :: ep_i

5 type(full_particle), dimension(n_jp), intent(in) :: ep_j

6 type(full_particle), dimension(n_ip), intent(inout) :: f

7 !* Local variables

8 integer(c_int) :: i,j

9 real(c_double) :: eps2 ,poti ,r3_inv ,r_inv

10 type(fdps_f64vec) :: xi ,ai ,rij

11

12 !* Compute force

13 eps2 = eps_grav * eps_grav

14 do i=1,n_ip

15 xi = ep_i(i)%pos

16 ai = 0.0d0

17 poti = 0.0d0

18 do j=1,n_jp

19 rij%x = xi%x - ep_j(j)%pos%x

20 rij%y = xi%y - ep_j(j)%pos%y

21 rij%z = xi%z - ep_j(j)%pos%z

22 r3_inv = rij%x*rij%x &

23 + rij%y*rij%y &

24 + rij%z*rij%z &

25 + eps2

26 r_inv = 1.0d0/sqrt(r3_inv)

27 r3_inv = r_inv * r_inv

28 r_inv = r_inv * ep_j(j)%mass

29 r3_inv = r3_inv * r_inv

30 ai%x = ai%x - r3_inv * rij%x

31 ai%y = ai%y - r3_inv * rij%y

32 ai%z = ai%z - r3_inv * rij%z

33 poti = poti - r_inv

34 ! [IMPORTANT NOTE]

35 ! In the innermost loop , we use the components of vectors

36 ! directly for vector operations because of the following

37 ! reasion. Except for intel compilers with ‘-ipo ‘ option ,

38 ! most of Fortran compilers use function calls to perform

39 ! vector operations like rij = x - ep_j(j)%pos.

40 ! This significantly slow downs the speed of the code.

41 ! By using the components of vector directly , we can avoid

42 ! these function calls.

43 end do

44 f(i)%pot = f(i)%pot + poti

45 f(i)%acc = f(i)%acc + ai

46 end do

47

48 end subroutine calc_gravity_ep_ep

本サンプルコードでは、サブルーチン calc gravity ep epとして実装されている。サブ
ルーチンの仮引数は、EssentialParticleIの配列、EssentialParticleIの個数、EssentialParticleJ
の配列、EssentialParticleJの個数、Force型の配列である。本サンプルコードでは、FullPar-

ticle型がすべてのユーザ定義型を兼ねているため、引数のデータ型はすべて full particle

型となっていることに注意して頂きたい。

4.1.2.3 相互作用関数 calcForceEpSp

ユーザーは粒子-超粒子間相互作用の仕方を記述した相互作用関数 calcForceEpSpを記述
しなければならない。calcForceEpSpには、粒子-超粒子相互作用計算の具体的な内容を書く
必要があり、サブルーチンとして実装しなければならない。Listing 3に、本サンプルコード
での実装を示す ( user defined.F90 を参照)。

Listing 3: 関数 calcForceEpSp

1 subroutine calc_gravity_ep_sp(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

2 implicit none

3 integer(c_int), intent(in), value :: n_ip ,n_jp

4 type(full_particle), dimension(n_ip), intent(in) :: ep_i

5 type(fdps_spj_monopole), dimension(n_jp), intent(in) :: ep_j

6 type(full_particle), dimension(n_ip), intent(inout) :: f

7 !* Local variables

8 integer(c_int) :: i,j

9 real(c_double) :: eps2 ,poti ,r3_inv ,r_inv

10 type(fdps_f64vec) :: xi ,ai ,rij

11

12 eps2 = eps_grav * eps_grav

13 do i=1,n_ip

14 xi = ep_i(i)%pos

15 ai = 0.0d0

16 poti = 0.0d0

17 do j=1,n_jp

18 rij%x = xi%x - ep_j(j)%pos%x

19 rij%y = xi%y - ep_j(j)%pos%y

20 rij%z = xi%z - ep_j(j)%pos%z

21 r3_inv = rij%x*rij%x &

22 + rij%y*rij%y &

23 + rij%z*rij%z &

24 + eps2

25 r_inv = 1.0d0/sqrt(r3_inv)

26 r3_inv = r_inv * r_inv

27 r_inv = r_inv * ep_j(j)%mass

28 r3_inv = r3_inv * r_inv

29 ai%x = ai%x - r3_inv * rij%x

30 ai%y = ai%y - r3_inv * rij%y

31 ai%z = ai%z - r3_inv * rij%z

32 poti = poti - r_inv

33 end do

34 f(i)%pot = f(i)%pot + poti

35 f(i)%acc = f(i)%acc + ai

36 end do

37

38 end subroutine calc_gravity_ep_sp

本サンプルコードでは、サブルーチン calc gravity ep spとして実装されている。サブルー
チンの仮引数は、EssentialParticleIの配列、EssentialParticleIの個数、超粒子の配列、超粒
子の個数、Force型の配列である。本サンプルコードでは、FullParticle型がすべてのユーザ
定義型を兼ねているため、引数のForce型は full particle型となっていることに注意して
頂きたい。ここで指定する超粒子型はこの相互作用計算を実施するのに使用するツリーオブ
ジェクトの種別と適合していなければならない。

4.1.3 プログラム本体

本節では、FDPS Fortranインターフェースを用いてN 体計算を行うにあたり、“メイン
ルーチン” f_main()に書かれるべきサブルーチンや関数に関して解説する。ここで、メイン
ルーチンとはっきり書かないのは、次の理由による: FDPS Fortranインターフェースを使
用する場合、ユーザコードは必ずサブルーチン

ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ̂ˆ
f_main()の下に記述されなければならず、

ユーザコードは正しい意味でのメインルーチンを持たない (メインルーチンはインターフェー
スプログラムの C++ソースコード内にある)。しかし、実質的にはサブルーチン f_main()

がメインルーチンの役割を果たす。そのため、敢えて “メインルーチン”という言葉を使った。
メインルーチンという言葉は、それがユーザコードの入り口であることを示すのに適してい
るので、以後、f_main()をメインルーチンと呼ぶことにする。本サンプルコードのメイン
ルーチンは f_main.F90に記述されている。

4.1.3.1 fdps controller型オブジェクトの生成

FDPS Fortran インターフェースにおいて、FDPSのAPIはすべてFortran 2003のクラス
FDPS controllerのメンバ関数として提供される。このクラスは、インターフェースプログ
ラムの 1つである FDPS module.F90の中の、モジュール fdps module内で定義されている。
したがって、ユーザはFDPSのAPIを使用するために、FDPS controller型オブジェクトを
生成しなければならない。本サンプルコードでは、FDPS controller型オブジェクト fdps

ctrlをメインルーチンで生成している:

Listing 4: fdps controller型オブジェクトの生成
1 subroutine f_main ()

2 use fdps_module

3 implicit none

4 !* Local variables

5 type(fdps_controller) :: fdps_ctrl

6

7 ! Do something

8

9 end subroutine f_main

ここに示したコードは実際にサンプルコードから必要な部分だけを取り出したものであるこ
とに注意して頂きたい。
上記の理由から、以下の説明において、FDPSのAPIはこのオブジェクトのメンバ関数と
して呼び出されていることに注意されたい。

4.1.3.2 開始、終了

まずは、FDPSの初期化/開始を行う必要がある。次のように、メインルーチンに記述する。

Listing 5: FDPSの開始
1 call fdps_ctrl%ps_initialize ()

FDPSは、開始したら明示的に終了させる必要がある。今回は、プログラムの終了と同時
に FDPSも終了させるため、メインルーチンの最後に次のように記述する。

Listing 6: FDPSの終了
1 call fdps_ctrl%ps_finalize ()

4.1.3.3 オブジェクトの生成・初期化

FDPSの初期化に成功した場合、ユーザーはコード中で用いるオブジェクトを作成する必
要がある。本節では、オブジェクトの生成/初期化の仕方について解説する。

4.1.3.3.1 オブジェクトの生成

今回の計算では、粒子群オブジェクト、領域情報オブジェクトに加え、重力計算用のツリー
オブジェクトを 1個生成する必要がある。Fortranインターフェースでは、これらオブジェク
トはすべて整数変数に格納された識別番号を使って操作する。したがって、まず識別番号を
格納する整数変数を用意したあとに、オブジェクトを生成するAPIを呼び出す必要がある。
以下にそのコードを記す。これらはサンプルコード f_main.F90のメインルーチン内に記述
されている。

Listing 7: オブジェクトの生成
1 subroutine f_main ()

2 use fdps_module

3 use user_defined_types

4 implicit none

5 !* Local variables

6 integer :: psys_num ,dinfo_num ,tree_num

7

8 !* Create FDPS objects

9 call fdps_ctrl%create_dinfo(dinfo_num)

10 call fdps_ctrl%create_psys(psys_num ,’full_particle ’)

11 call fdps_ctrl%create_tree(tree_num , &

12 "Long ,full_particle ,full_particle ,

full_particle ,Monopole")

13

14 end subroutine f_main

ここでも、実際のサンプルコードから該当部分だけを抜き出していることに注意して頂き
たい。
上に示すように、粒子群オブジェクトを生成する際には FullParticle型に対応する派生
データ型名を文字列として APIの引数に渡す必要がある。同様に、ツリーオブジェクト生
成の際には、ツリーの種別を示す文字列をAPIの引数に渡す必要がある。両APIにおいて、
派生データ型 名は小文字で入力されなければならない。

4.1.3.3.2 領域情報オブジェクトの初期化

ユーザーはオブジェクトを作成したら、そのオブジェクトの初期化を行う必要がある。本
サンプルコードでは周期境界等は用いていないため、領域情報オブジェクトの初期化はAPI

init dinfo を実行するだけでよい:

Listing 8: 領域オブジェクトの初期化
1 call fdps_ctrl%init_dinfo(dinfo_num ,coef_ema)

ここで、API init dinfo の第 2 引数は領域分割に使用される指数移動平均の平滑化係数を
表す。この係数の意味については仕様書に詳しい解説があるので、そちらを参照されたい。

4.1.3.3.3 粒子群オブジェクトの初期化

次に、粒子群オブジェクトの初期化を行う必要がある。粒子群オブジェクトの初期化は、
API init psys で行う:

Listing 9: 粒子群オブジェクトの初期化
1 call fdps_ctrl%init_psys(psys_num)

4.1.3.3.4 ツリーオブジェクトの初期化

次に、ツリーオブジェクトの初期化を行う必要がある。ツリーオブジェクトの初期化はAPI

init tree で行う。このAPIには、引数としてツリーオブジェクト内部で使用する配列の大
きさの初期値を渡す必要がある。これはローカル粒子数 (n loc)程度で十分であるため、そ
のようにセットする:

Listing 10: ツリーオブジェクトの初期化
1 call fdps_ctrl%init_tree(tree_num ,n_loc ,theta , &

2 n_leaf_limit ,n_group_limit)

このAPIには 3つの省略可能引数が存在し、サンプルコードではこれらを省略せずに指定
している:

• theta — ツリー法で力の計算をする場合の見込み角についての基準
• n leaf limit — ツリーを切るのをやめる粒子数の上限
• n group limit — 相互作用リストを共有する粒子数の上限

4.1.3.4 粒子データの初期化

初期条件の設定を行うためには、粒子群オブジェクトに粒子データを入力する必要がある。
(既にAPI init psys で初期化済みの)粒子群オブジェクトに、FullParticle型粒子のデータ
を格納するには、粒子群オブジェクトの API set nptcl loc と get psys fptr を用いて、
次のように行う:

Listing 11: 粒子データの初期化
1 subroutine foo(fdps_ctrl ,psys_num)

2 use fdps_vector

3 use fdps_module

4 use user_defined_types

5 implicit none

6 type(fdps_controller), intent(IN) :: fdps_ctrl

7 integer , intent(IN) :: psys_num

8 !* Local variables

9 integer :: i,nptcl_loc

10 type(full_particle), dimension (:), pointer :: ptcl

11

12 !* Set # of local particles

13 call fdps_ctrl%set_nptcl_loc(psys_num ,nptcl_loc)

14

15 !* Get the pointer to full particle data

16 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

17

18 !* Initialize particle data

19 do i=1,nptcl_loc

20 ptcl(i)%pos = ! Do something

21 end do

22

23 !* Release the pointer

24 nullify(ptcl)

25

26 end subroutine foo

まず、粒子群オブジェクトに粒子データを保存するのに必要なメモリを確保しなければなら
ない。これを行うには API set nptcl loc を実行すればよい。このAPIは指定された粒子
群オブジェクトのローカル粒子数 (自プロセスが管理する粒子数)の値を設定し、かつ、その
粒子数を格納するのに必要なメモリを確保する。粒子データを初期化するためには、確保さ

れたメモリのアドレスを取得しなければならない。これにはAPI get psys fptr を使用す
る。 アドレスは Fortranポインタで受け取る必要がある。そのため、上記の例では、ポイ
ンタを以下のように用意している:� �
type(full_particle), dimension(:), pointer :: ptcl� �

API get psys fptrによってポインタを設定した後は、ポインタを粒子配列のように使用す
ることが可能である。上の例では、粒子データの設定が完了した後、ポインタを組込関数
nullifyによって解放している。

4.1.3.5 ループ

本節では、時間積分ループの中で行わなければならないことについて、解説する。

4.1.3.5.1 領域分割の実行

まずは、粒子分布に基いて、領域分割を実行する。本サンプルコードでは、これを領域情
報オブジェクトのAPI decompose domain allを用いて行っている:

Listing 12: 領域分割の実行
1 if (mod(num_loop ,4) == 0) then

2 call fdps_ctrl%decompose_domain_all(dinfo_num ,psys_num)

3 end if

ここで、計算時間の節約のため、領域分割は 4ループ毎に 1回だけ行うようにしている。

4.1.3.5.2 粒子交換の実行

次に、領域情報に基いて、プロセス間の粒子の情報を交換する。これには、粒子群オブジェ
クトのAPI exchange particle を用いる:

Listing 13: 粒子交換の実行
1 call fdps_ctrl%exchange_particle(psys_num ,dinfo_num)

4.1.3.5.3 相互作用計算の実行

領域分割・粒子交換が終了したら、相互作用の計算を行う。これには、ツリーオブジェク
トのAPI calc force all and write back を用いる:

Listing 14: 相互作用計算の実行
1 subroutien f_main ()

2 use , intrinsic :: iso_c_binding

3 use user_defined_types

4 implicit none

5 !* Local variables

6 type(c_funptr) :: pfunc_ep_ep ,pfunc_ep_sp

7

8 ! Do somehting

9

10 pfunc_ep_ep = c_funloc(calc_gravity_ep_ep)

11 pfunc_ep_sp = c_funloc(calc_gravity_ep_sp)

12 call fdps_ctrl%calc_force_all_and_write_back(tree_num , &

13 pfunc_ep_ep , &

14 pfunc_ep_sp , &

15 psys_num , &

16 dinfo_num)

17

18 ! Do something

19

20 end subroutine f_main

ここで、APIの第 2,3引数には関数 calcForceEpEp, calcForceEpSpの (C言語アドレス注 4)と
しての)関数ポインタを指定する。関数のC言語アドレスは、Fortran 2003で導入された組
込関数 c funlocを使って取得する (この組込み関数はモジュール iso c bindingで提供され
るため、use文を使い、このモジュールを利用可能にしている)。C言語アドレスを格納する
ためには、同じく Fortran 2003で導入された派生データ型 c funptrの変数が必要である。
そのため、本サンプルコードでは、c funptr型変数として、pfunc ep epと pfunc ep spを
用意している。ここに、calc gravity ep epと calc gravity ep spのC言語アドレスを格
納した上で、APIに渡している。

4.1.3.5.4 時間積分

本サンプルコードでは、時間積分をLeapfrog時間積分法によって行う。時間積分は形式的
に、K(∆t

2
)D(∆t)K(∆t

2
)と表される。ここで、∆tは時間刻み、K(·)は速度を指定された時間

だけ時間推進するオペレータ、D(·)は位置を指定された時間だけ時間推進するオペレータで
ある。本サンプルコードにおいて、これらのオペレータは、サブルーチン kickとサブルー
チン driftとして実装している。
時間積分ループの最初で、最初のD(∆t)K(∆t

2
)の計算を行い、粒子の座標と速度の情報を

更新している:

Listing 15: D(∆t)K(∆t
2
)オペレータの計算

1 !* Leapfrog: Kick -Drift

2 call kick(fdps_ctrl ,psys_num ,0.5d0*dt)

3 time_sys = time_sys + dt

4 call drift(fdps_ctrl ,psys_num ,dt)

時間積分ループの次の部分では、力の計算を行い、その後、最後のK(∆t
2
)の計算を行って

いる:

Listing 16: K(∆t
2
)オペレータの計算

注 4)C言語方式で記述されたアドレス情報のこと。

1 !* Leapfrog: Kick

2 call kick(fdps_ctrl ,psys_num ,0.5d0*dt)

4.1.3.6 粒子データの更新

上記で説明した kickや drift等のサブルーチンで、粒子データを更新するためには、粒子
群オブジェクトに格納されている粒子データにアクセスする必要がある。これは、第 4.1.3.4

節で説明した方法とほぼ同様に行う:

Listing 17: 粒子データの更新
1 subroutine foo(fdps_ctrl ,psys_num)

2 use fdps_vector

3 use fdps_module

4 use user_defined_types

5 implicit none

6 type(fdps_controller), intent(IN) :: fdps_ctrl

7 integer , intent(IN) :: psys_num

8 !* Local variables

9 integer :: i,nptcl_loc

10 type(full_particle), dimension (:), pointer :: ptcl

11

12 !* Get # of local particles

13 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

14

15 !* Get the pointer to full particle data

16 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

17

18 !* Initialize or update particle data

19 do i=1,nptcl_loc

20 ptcl(i)%pos = ! Do something

21 end do

22

23 !* Release the pointer

24 nullify(ptcl)

25

26 end subroutine foo

API get psys fptr を使い、粒子群オブジェクトに格納された粒子データのアドレスをポイ
ンタとして受け取る。受け取ったポインタは要素数 nptcl locの粒子配列として振る舞うの
で、一般的な配列同様に値を更新すればよい。

4.1.4 ログファイル

計算が正しく開始すると、標準出力に、時間・エネルギー誤差の 2つが出力される。以下
はその出力の最も最初のステップでの例である。

Listing 18: 標準出力の例
1 time: 0.0000000000E+000, energy error: -0.0000000000E+000

4.2 固定長SPHシミュレーションコード

本節では、前節 (第 3節)で使用した、固定 smoothing lengthでの標準 SPH法のサンプル
コードの詳細について解説する。

4.2.1 ソースファイルの場所と構成

ソースファイルは$(FDPS)/sample/fortran/sph以下にある。 サンプルコードは、次節
で説明するユーザ定義型が記述されたソースコード user_defined.F90と、SPHシミュレー
ションのメインループ等が記述されたソースコード f_main.F90から構成される。この他に、
GCCと Intelコンパイラ用のMakefileである Makefileと Makefile.intelがある。

4.2.2 ユーザー定義型・ユーザ定義関数

本節では、FDPSの機能を用いて SPHの計算を行う際に、ユーザーが記述しなければな
らない派生データ型とサブルーチンについて記述する。

4.2.2.1 FullParticle型

ユーザーはユーザ定義型の 1つFullParticle型を記述しなければならない。FullParticle型
には、シミュレーションを行うにあたって、SPH粒子が持っているべき全ての物理量が含ま
れている。Listing 19に本サンプルコード中で用いる FullParticle型の実装例を示す ( user

defined.F90 を参照)。

Listing 19: FullParticle型
1 !**** Full particle type

2 type , public , bind(c) :: full_particle !$fdps FP

3 !$fdps copyFromForce force_dens (dens ,dens)

4 !$fdps copyFromForce force_hydro (acc ,acc) (eng_dot ,eng_dot) (dt ,dt)

5 real(kind=c_double) :: mass !$fdps charge

6 type(fdps_f64vec) :: pos !$fdps position

7 type(fdps_f64vec) :: vel

8 type(fdps_f64vec) :: acc

9 real(kind=c_double) :: dens

10 real(kind=c_double) :: eng

11 real(kind=c_double) :: pres

12 real(kind=c_double) :: smth !$fdps rsearch

13 real(kind=c_double) :: snds

14 real(kind=c_double) :: eng_dot

15 real(kind=c_double) :: dt

16 integer(kind=c_long_long) :: id

17 type(fdps_f64vec) :: vel_half

18 real(kind=c_double) :: eng_half

19 end type full_particle

SPHサンプルコードではN 体サンプルコードと異なり、FullPartice型が他のユーザ定義
型を兼ねることはない。したがって、この派生データ型が FullParticle型であることを示す
ため、次の指示文を記述している:� �
type, public, bind(c) :: full_particle !$fdps FP� �

SPHシミュレーションにおける相互作用は短距離力である。そのため、必須物理量として探
索半径が加わる。粒子位置等の指定も含め、どのメンバ変数がどの必須物理量に対応してい
るかの指定を次の指示文で行っている:� �
real(kind=c_double) :: mass !$fdps charge

type(fdps_f64vec) :: pos !$fdps position

real(kind=c_double) :: smth !$fdps rsearch� �
N体シミュレーションコードの節で述べたように、メンバ変数が粒子速度であることを指定
するキーワード velocityは予約語でしかないため、本サンプルコードでは指定してない。
FullParticle型はForce型との間でデータコピーを行う。ユーザは指示文を使い、FDPSに
データコピーの仕方を教えなければならない。後述するように本 SPHサンプルコードには 2

つの Force型が存在する。したがって、ユーザはそれぞれの Force型に対して、指示文を記
述する必要がある。本サンプルコードでは、以下のように記述している:� �
!$fdps copyFromForce force_dens (dens,dens)

!$fdps copyFromForce force_hydro (acc,acc) (eng_dot,eng_dot) (dt,dt)� �
4.2.2.2 EssentialParticleI型

ユーザーは EssentialParticleI型を記述しなければならない。EssentialParticleI型には、
Forceの計算を行う際、i粒子が持っているべき全ての物理量をメンバ変数として持っている
必要がある。また、本サンプルコード中では、EssentialParticleJ型も兼ねているため、j粒
子が持っているべき全ての物理量もメンバ変数として持っている必要がある。Listing 20に、
本サンプルコードの EssentialParticleI型の実装例を示す ( user defined.F90 参照):

Listing 20: EssentialParticleI型
1 !**** Essential particle type

2 type , public , bind(c) :: essential_particle !$fdps EPI ,EPJ

3 !$fdps copyFromFP full_particle (id ,id) (pos ,pos) (vel ,vel) (mass ,

mass) (smth ,smth) (dens ,dens) (pres ,pres) (snds ,snds)

4 integer(kind=c_long_long) :: id !$fdps id

5 type(fdps_f64vec) :: pos !$fdps position

6 type(fdps_f64vec) :: vel

7 real(kind=c_double) :: mass !$fdps charge

8 real(kind=c_double) :: smth !$fdps rsearch

9 real(kind=c_double) :: dens

10 real(kind=c_double) :: pres

11 real(kind=c_double) :: snds

12 end type essential_particle

まず、ユーザは指示文を用いて、この派生データ型が EssentialParticleI型かつ Essential-

ParticleJ型であることをFDPSに教えなければならない。本サンプルコードでは、以下のよ
うに記述している:� �
type, public, bind(c) :: essential_particle !$fdps EPI,EPJ� �
次に、ユーザはこの派生データ型のどのメンバ変数がどの必須物理量に対応するのかを指示
文によって指定しなければならない。今回は SPHシミュレーションを行うので探索半径の
指定も必要である。本サンプルコードでは、以下のように記述している:� �
type(fdps_f64vec) :: pos !$fdps position

real(kind=c_double) :: mass !$fdps charge

real(kind=c_double) :: smth !$fdps rsearch� �
EssentialParticleI型とEssentialParticleJ型はFullParticle型からデータを受け取る。ユーザ
はFullParticle型のどのメンバ変数をEssentialParticle?型 (?=I,J)のどのメンバ変数にコピー
するのかを、指示文を用いて指定する必要がある。本サンプルコードでは、以下のように記
述している:� �
!$fdps copyFromFP full_particle (id,id) (pos,pos) (vel,vel) (mass,mass)

(smth,smth) (dens,dens) (pres,pres) (snds,snds)� �
4.2.2.3 Force型

ユーザーはForce型を記述しなければならない。Force型には、Forceの計算を行った際に
その結果として得られる全ての物理量をメンバ変数として持っている必要がある。また、本
サンプルコード中では、Forceの計算は密度の計算と流体相互作用計算の 2つが存在するた
め、Force型は 2つ書く必要がある。Listing 21に、本サンプルコード中で用いるForce型の
実装例を示す。

Listing 21: Force型
1 !**** Force types

2 type , public , bind(c) :: force_dens !$fdps Force

3 !$fdps clear smth=keep

4 real(kind=c_double) :: dens

5 real(kind=c_double) :: smth

6 end type force_dens

7

8 type , public , bind(c) :: force_hydro !$fdps Force

9 !$fdps clear

10 type(fdps_f64vec) :: acc

11 real(kind=c_double) :: eng_dot

12 real(kind=c_double) :: dt

13 end type force_hydro

まず、ユーザはこれらの派生データ型がForce型であることを指示文によって指定する必要
がある。この実装例では、それぞれの派生データ型に対して、以下のように記述している:� �
type, public, bind(c) :: force_dens !$fdps Force

type, public, bind(c) :: force_hydro !$fdps Force� �
これらの派生データ型は Force型であるから、ユーザは必ず、相互作用計算における積算対
象のメンバ変数の初期化方法を指定する必要がある。本サンプルコードでは、積算対象であ
る密度、(圧力勾配による)加速度、エネルギー密度の変化率、時間刻みのみを 0クリアする
指示を出している:� �
!$fdps clear smth=keep

!$fdps clear� �
この例において、Force型 force densには、smoothing lengthを表すメンバ変数smthが用意
されている。本来、固定長SPHでは、Force型に smoothing lengthに対応するメンバを持たせ
る必要はない。しかし、ここでは、ユーザが将来的に可変長SPHに移行することを想定して用
意してある。可変長 SPHの formulationの 1つである Springel [2005,MNRAS,364,1105]の方
法では、密度計算と同時に smoothing lengthを計算する必要がある。そのような formulation

を実装する場合には、この例のように、Force型に smoothing lengthを持たせる必要が生じ
る。本サンプルコードでは固定長 SPHを使うため、smthを 0クリアされないようにしてい
る (0クリアされては 2回目以降の密度計算が破綻するため)。

4.2.2.4 相互作用関数 calcForceEpEp

ユーザーは粒子間相互作用の仕方を記述した相互作用関数 calcForceEpEpを記述しなけれ
ばならない。相互作用関数 calcForceEpEpには、各 Force型に対応する粒子-粒子相互作用
計算の具体的な内容を書く必要がある。Listing 22に、本サンプルコードでの実装を示す (

user defined.F90 を参照)。

Listing 22: 関数 calcForceEpEp
1 !**** Interaction function

2 subroutine calc_density(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

3 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

4 type(essential_particle), dimension(n_ip), intent(in) :: ep_i

5 type(essential_particle), dimension(n_jp), intent(in) :: ep_j

6 type(force_dens), dimension(n_ip), intent(inout) :: f

7 !* Local variables

8 integer(kind=c_int) :: i,j

9 type(fdps_f64vec) :: dr

10

11 do i=1,n_ip

12 f(i)%dens = 0.0d0

13 do j=1,n_jp

14 dr%x = ep_j(j)%pos%x - ep_i(i)%pos%x

15 dr%y = ep_j(j)%pos%y - ep_i(i)%pos%y

16 dr%z = ep_j(j)%pos%z - ep_i(i)%pos%z

17 f(i)%dens = f(i)%dens &

18 + ep_j(j)%mass * W(dr ,ep_i(i)%smth)

19 end do

20 end do

21

22 end subroutine calc_density

23

24 !**** Interaction function

25 subroutine calc_hydro_force(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

26 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

27 type(essential_particle), dimension(n_ip), intent(in) :: ep_i

28 type(essential_particle), dimension(n_jp), intent(in) :: ep_j

29 type(force_hydro), dimension(n_ip), intent(inout) :: f

30 !* Local parameters

31 real(kind=c_double), parameter :: C_CFL =0.3d0

32 !* Local variables

33 integer(kind=c_int) :: i,j

34 real(kind=c_double) :: mass_i ,mass_j ,smth_i ,smth_j , &

35 dens_i ,dens_j ,pres_i ,pres_j , &

36 snds_i ,snds_j

37 real(kind=c_double) :: povrho2_i ,povrho2_j , &

38 v_sig_max ,dr_dv ,w_ij ,v_sig ,AV

39 type(fdps_f64vec) :: pos_i ,pos_j ,vel_i ,vel_j , &

40 dr ,dv ,gradW_ij

41

42 do i=1,n_ip

43 !* Zero -clear

44 v_sig_max = 0.0d0

45 !* Extract i-particle info.

46 pos_i = ep_i(i)%pos

47 vel_i = ep_i(i)%vel

48 mass_i = ep_i(i)%mass

49 smth_i = ep_i(i)%smth

50 dens_i = ep_i(i)%dens

51 pres_i = ep_i(i)%pres

52 snds_i = ep_i(i)%snds

53 povrho2_i = pres_i /( dens_i*dens_i)

54 do j=1,n_jp

55 !* Extract j-particle info.

56 pos_j%x = ep_j(j)%pos%x

57 pos_j%y = ep_j(j)%pos%y

58 pos_j%z = ep_j(j)%pos%z

59 vel_j%x = ep_j(j)%vel%x

60 vel_j%y = ep_j(j)%vel%y

61 vel_j%z = ep_j(j)%vel%z

62 mass_j = ep_j(j)%mass

63 smth_j = ep_j(j)%smth

64 dens_j = ep_j(j)%dens

65 pres_j = ep_j(j)%pres

66 snds_j = ep_j(j)%snds

67 povrho2_j = pres_j /( dens_j*dens_j)

68 !* Compute dr & dv

69 dr%x = pos_i%x - pos_j%x

70 dr%y = pos_i%y - pos_j%y

71 dr%z = pos_i%z - pos_j%z

72 dv%x = vel_i%x - vel_j%x

73 dv%y = vel_i%y - vel_j%y

74 dv%z = vel_i%z - vel_j%z

75 !* Compute the signal velocity

76 dr_dv = dr%x * dv%x + dr%y * dv%y + dr%z * dv%z

77 if (dr_dv < 0.0d0) then

78 w_ij = dr_dv / sqrt(dr%x * dr%x + dr%y * dr%y + dr%z * dr%z

)

79 else

80 w_ij = 0.0d0

81 end if

82 v_sig = snds_i + snds_j - 3.0d0 * w_ij

83 v_sig_max = max(v_sig_max , v_sig)

84 !* Compute the artificial viscosity

85 AV = - 0.5d0*v_sig*w_ij / (0.5d0*( dens_i+dens_j))

86 !* Compute the average of the gradients of kernel

87 gradW_ij = 0.5d0 * (gradW(dr ,smth_i) + gradW(dr ,smth_j))

88 !* Compute the acceleration and the heating rate

89 f(i)%acc%x = f(i)%acc%x - mass_j *( povrho2_i+povrho2_j+AV)*

gradW_ij%x

90 f(i)%acc%y = f(i)%acc%y - mass_j *( povrho2_i+povrho2_j+AV)*

gradW_ij%y

91 f(i)%acc%z = f(i)%acc%z - mass_j *( povrho2_i+povrho2_j+AV)*

gradW_ij%z

92 f(i)%eng_dot = f(i)%eng_dot &

93 + mass_j * (povrho2_i + 0.5d0*AV) &

94 *(dv%x * gradW_ij%x &

95 +dv%y * gradW_ij%y &

96 +dv%z * gradW_ij%z)

97 end do

98 f(i)%dt = C_CFL *2.0d0*smth_i /( v_sig_max*kernel_support_radius)

99 end do

100 ! [IMPORTANT NOTE]

101 ! In the innermost loop , we use the components of vectors

102 ! directly for vector operations because of the following

103 ! reasion. Except for intel compilers with ‘-ipo ‘ option ,

104 ! most of Fortran compilers use function calls to perform

105 ! vector operations like rij = x - ep_j(j)%pos.

106 ! This significantly slow downs the speed of the code.

107 ! By using the components of vector directly , we can avoid

108 ! these function calls.

109

110 end subroutine calc_hydro_force

本 SPHシミュレーションコードでは、2種類の相互作用があるため、calcForceEpEp は 2

つ記述する必要がある。いずれの場合にも、サブルーチンの仮引数は、EssentialParticleIの
配列、EssentialParticleIの個数、EssentialParticleJの配列、EssentialParticleJの個数、Force
型の配列である。

4.2.3 プログラム本体

本節では、FDPSを用いて SPH計算を行う際に、メインルーチンに書かれるべき関数に関
して解説する (本文書におけるメインルーチンの定義については、第 4.1.3節を参照のこと)

。

4.2.3.1 fdps controller型オブジェクトの生成

ユーザは FDPSのAPIを使用するために、FDPS controller型オブジェクトを生成しな
ければならない。本サンプルコードでは、FDPS controller型オブジェクト fdps ctrlをメ
インルーチンで生成している:

Listing 23: fdps controller型オブジェクトの生成
1 subroutine f_main ()

2 use fdps_module

3 implicit none

4 !* Local variables

5 type(fdps_controller) :: fdps_ctrl

6

7 ! Do something

8

9 end subroutine f_main

ここに示したコードは実際にサンプルコードから必要な部分だけを取り出したものであるこ
とに注意して頂きたい。
上記の理由から、以下の説明において、FDPSのAPIはこのオブジェクトのメンバ関数と
して呼び出されていることに注意されたい。

4.2.3.2 開始、終了

まずは、FDPSの初期化/開始を行う必要がある。次のように、メインルーチンに記述する。

Listing 24: FDPSの開始
1 call fdps_ctrl%ps_initialize ()

FDPSは、開始したら明示的に終了させる必要がある。今回は、プログラムの終了と同時
に FDPSも終了させるため、メインルーチンの最後に次のように記述する。

Listing 25: FDPSの終了
1 call fdps_ctrl%PS_Finalize ()

4.2.3.3 オブジェクトの生成・初期化

FDPSの初期化に成功した場合、ユーザーはコード中で用いるオブジェクトを作成する必
要がある。本節では、オブジェクトの生成/初期化の仕方について、解説する。

37



目 次

4.2.3.3.1 オブジェクトの生成

SPHでは、粒子群オブジェクト、領域情報オブジェクトに加え、密度計算用にGather型
の短距離力用ツリーを 1本、流体相互作用計算用に Symmetry型の短距離力用ツリーを 1本
生成する必要がある。以下にそのコードを記す。

Listing 26: オブジェクトの生成
1 subroutine f_main ()

2 use fdps_vector

3 use fdps_module

4 use user_defined_types

5 implicit none

6 !* Local variables

7 integer :: psys_num ,dinfo_num

8 integer :: dens_tree_num ,hydro_tree_num

9

10 !* Create FDPS objects

11 call fdps_ctrl%create_psys(psys_num ,’full_particle ’)

12 call fdps_ctrl%create_dinfo(dinfo_num)

13 call fdps_ctrl%create_tree(dens_tree_num , &

14 "Short ,dens_force ,essential_particle ,

essential_particle ,Gather")

15 call fdps_ctrl%create_tree(hydro_tree_num , &

16 "Short ,hydro_force ,essential_particle ,

essential_particle ,Symmetry")

17

18 end subroutine f_main

ここでも、実際のサンプルコードから該当部分だけを抜き出していることに注意して頂きた
い。API create psys と create tree には、それぞれ、粒子種別とツリー種別を示す文字
列を渡す。これら文字列の中のすべての派生データ型名は小文字で記述されなければなら

ないことに注意して頂きたい。

4.2.3.3.2 領域情報オブジェクトの初期化

ユーザーはオブジェクトを作成したら、そのオブジェクトの初期化を行う必要がある。こ
こでは、まず領域情報オブジェクトの初期化について、解説する。領域情報オブジェクトの
初期化が終わった後、領域情報オブジェクトに周期境界の情報と、境界の大きさをセットす
る必要がある。今回のサンプルコードでは、x, y, z方向に周期境界を用いる。

Listing 27: 領域情報オブジェクトの初期化
1 call fdps_ctrl%init_dinfo(dinfo_num ,coef_ema)

2 call fdps_ctrl%set_boundary_condition(dinfo_num ,fdps_bc_periodic_xyz)

3 call fdps_ctrl%set_pos_root_domain(dinfo_num ,pos_ll ,pos_ul)

4.2.3.3.3 粒子群オブジェクトの初期化

次に、粒子群オブジェクトの初期化を行う必要がある。粒子群オブジェクトの初期化は、
次の一文だけでよい。

Listing 28: 粒子群オブジェクトの初期化
1 call fdps_ctrl%init_psys(psys_num)

4.2.3.3.4 ツリーオブジェクトの初期化

次に、ツリーオブジェクトの初期化を行う必要がある。ツリーオブジェクトの初期化を行
う関数には、オブジェクト内部で使用する配列の大きさの初期値を渡す必要がある。初期値
はローカル粒子数 (n loc)程度で十分であるため、そのようにセットする。

Listing 29: 相互作用ツリークラスの初期化
1 call fdps_ctrl%init_tree(dens_tree_num ,n_loc ,theta , &

2 n_leaf_limit ,n_group_limit)

3 call fdps_ctrl%init_tree(hydro_tree_num ,n_loc ,theta , &

4 n_leaf_limit ,n_group_limit)

4.2.3.4 ループ

本節では、時間積分ループの中で行わなければならないことについて、解説する。

4.2.3.4.1 領域分割の実行

まずは、粒子分布に基いて、領域分割を実行する。これには、領域情報オブジェクトのAPI

decompose domain allを用いる。

Listing 30: 領域分割の実行
1 call fdps_ctrl%decompose_domain_all(dinfo_num ,psys_num)

4.2.3.4.2 粒子交換の実行

次に、領域情報に基いて、プロセス間の粒子の情報を交換する。これには、粒子群オブジェ
クトのAPI exchange particle を用いる。

Listing 31: 粒子交換の実行
1 call fdps_ctrl%exchange_particle(psys_num ,dinfo_num)

4.2.3.4.3 相互作用計算の実行

領域分割・粒子交換が終了したら、相互作用の計算を行う。これには、ツリーオブジェク
トのAPI calc force all and write back を用いる。

Listing 32: 相互作用計算の実行
1 subroutine f_main ()

2 use , intrinsic :: iso_c_binding

3 use user_defined_types

4 implicit none

5 !* Local variables

6 type(c_funptr) :: pfunc_ep_ep

7

8 ! Do something

9

10 pfunc_ep_ep = c_funloc(calc_density)

11 call fdps_ctrl%calc_force_all_and_write_back(tree_num_dens , &

12 pfunc_ep_ep , &

13 psys_num , &

14 dinfo_num)

15 call set_pressure(fdps_ctrl ,psys_num)

16 pfunc_ep_ep = c_funloc(calc_hydro_force)

17 call fdps_ctrl%calc_force_all_and_write_back(tree_num_hydro , &

18 pfunc_ep_ep , &

19 psys_num , &

20 dinfo_num)

21

22 ! Do something

23

24 end subroutine f_main

ここで、APIの第 2引数には関数 calcForceEpEp の (C言語アドレスとしての)関数ポイン
タを指定する。

4.2.4 コンパイル

作業ディレクトリで makeコマンドを打てばよい。Makefileとしては、サンプルコードに
付属の Makefileをそのまま用いる事にする。� �
$ make� �

4.2.5 実行

MPIを使用しないで実行する場合、コマンドライン上で以下のコマンドを実行すればよい。� �
$ ./sph.out� �
もし、MPIを用いて実行する場合は、以下のコマンドを実行すればよい。

� �
$ MPIRUN -np NPROC ./sph.out� �
ここで、MPIRUNには mpirunや mpiexecなどのMPI実行プログラムが、NPROCにはプロセ
ス数が入る。

4.2.6 ログファイル

計算が終了すると、resultフォルダ下にログが出力される。

4.2.7 可視化

ここでは、gnuplotを用いた可視化の方法について解説する。gnuplotで対話モードに入
るために、コマンドラインから gnuplotを起動する。� �
$ gnuplot� �
対話モードに入ったら、gnuplotを用いて可視化を行う。今回は、50番目のスナップショッ
トファイルから、横軸を粒子の x座標、縦軸を密度に取ったグラフを生成する。� �
gnuplot> plot "result/snap00050-proc00000.dat" u 3:9� �
ここで、文字列 procの後の整数はMPIのプロセス番号を表す。

5 サンプルコード

5.1 N体シミュレーション

N体シミュレーションのサンプルコードを以下に示す。このサンプルは第 3, 4節で用いた
N 体シミュレーションのサンプルコードと同じものである。これをカット&ペーストしてコ
ンパイルすれば、正常に動作するN体シミュレーションコードを作ることができる。

Listing 33: N 体シミュレーションのサンプルコード (user defined.F90)

1 !===============================

2 ! MODULE: User defined types

3 !===============================

4 module user_defined_types

5 use , intrinsic :: iso_c_binding

6 use fdps_vector

7 use fdps_super_particle

8 implicit none

9

10 !* Public variables

11 real(kind=c_double), public :: eps_grav ! gravitational softening

12

13 !**** Full particle type

14 type , public , bind(c) :: full_particle !$fdps FP ,EPI ,EPJ ,Force

15 !$fdps copyFromForce full_particle (pot ,pot) (acc ,acc)

16 !$fdps copyFromFP full_particle (id ,id) (mass ,mass) (pos ,pos)

17 !$fdps clear id=keep , mass=keep , pos=keep , vel=keep

18 integer(kind=c_long_long) :: id

19 real(kind=c_double) mass !$fdps charge

20 type(fdps_f64vec) :: pos !$fdps position

21 type(fdps_f64vec) :: vel !$fdps velocity

22 real(kind=c_double) :: pot

23 type(fdps_f64vec) :: acc

24 end type full_particle

25

26 !* The following types are used in PIKG -generated kenrels

27 type , public , bind(c) :: epi_grav

28 type(fdps_f32vec) :: pos

29 end type epi_grav

30

31 type , public , bind(c) :: epj_grav

32 type(fdps_f32vec) :: pos

33 real(kind=c_float) :: mass

34 end type epj_grav

35

36 type , public , bind(c) :: force_grav

37 type(fdps_f32vec) :: acc

38 real(kind=c_float) :: pot

39 end type force_grav

40

41 contains

42

43 !**** Interaction function (particle -particle)

44 #if defined(ENABLE_PHANTOM_GRAPE_X86)

45 subroutine calc_gravity_ep_ep(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

46 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

47 use omp_lib

48 #endif

49 use phantom_grape_g5_x86

50 implicit none

51 integer(c_int), intent(in), value :: n_ip ,n_jp

52 type(full_particle), dimension(n_ip), intent(in) :: ep_i

53 type(full_particle), dimension(n_jp), intent(in) :: ep_j

54 type(full_particle), dimension(n_ip), intent(inout) :: f

55 !* Local variables

56 integer(c_int) :: i,j

57 integer(c_int) :: nipipe ,njpipe ,devid

58 real(c_double), dimension(3,n_ip) :: xi,ai

59 real(c_double), dimension(n_ip) :: pi

60 real(c_double), dimension(3,n_jp) :: xj

61 real(c_double), dimension(n_jp) :: mj

62

63 nipipe = n_ip

64 njpipe = n_jp

65 do i=1,n_ip

66 xi(1,i) = ep_i(i)%pos%x

67 xi(2,i) = ep_i(i)%pos%y

68 xi(3,i) = ep_i(i)%pos%z

69 ai(1,i) = 0.0d0

70 ai(2,i) = 0.0d0

71 ai(3,i) = 0.0d0

72 pi(i) = 0.0d0

73 end do

74 do j=1,n_jp

75 xj(1,j) = ep_j(j)%pos%x

76 xj(2,j) = ep_j(j)%pos%y

77 xj(3,j) = ep_j(j)%pos%z

78 mj(j) = ep_j(j)%mass

79 end do

80 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

81 devid = omp_get_thread_num ()

82 ! [IMPORTANT NOTE]

83 ! The subroutine calc_gravity_pp is called by a OpenMP thread

84 ! in the FDPS. This means that here is already in the parallel

region.

85 ! So , you can use omp_get_thread_num () without !$OMP parallel

directives.

86 ! If you use them , a nested parallel resions is made and the

gravity

87 ! calculation will not be performed correctly.

88 #else

89 devid = 0

90 #endif

91 call g5_set_xmjMC(devid , 0, n_jp , xj, mj)

92 call g5_set_nMC(devid , n_jp)

93 call g5_calculate_force_on_xMC(devid , xi, ai, pi, n_ip)

94 do i=1,n_ip

95 f(i)%acc%x = f(i)%acc%x + ai(1,i)

96 f(i)%acc%y = f(i)%acc%y + ai(2,i)

97 f(i)%acc%z = f(i)%acc%z + ai(3,i)

98 f(i)%pot = f(i)%pot - pi(i)

99 end do

100 end subroutine calc_gravity_ep_ep

101

102 subroutine calc_gravity_ep_sp(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

103 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

104 use omp_lib

105 #endif

106 use phantom_grape_g5_x86

107 implicit none

108 integer(c_int), intent(in), value :: n_ip ,n_jp

109 type(full_particle), dimension(n_ip), intent(in) :: ep_i

110 type(fdps_spj_monopole), dimension(n_jp), intent(in) :: ep_j

111 type(full_particle), dimension(n_ip), intent(inout) :: f

112 !* Local variables

113 integer(c_int) :: i,j

114 integer(c_int) :: nipipe ,njpipe ,devid

115 real(c_double), dimension(3,n_ip) :: xi,ai

116 real(c_double), dimension(n_ip) :: pi

117 real(c_double), dimension(3,n_jp) :: xj

118 real(c_double), dimension(n_jp) :: mj

119

120 nipipe = n_ip

121 njpipe = n_jp

122 do i=1,n_ip

123 xi(1,i) = ep_i(i)%pos%x

124 xi(2,i) = ep_i(i)%pos%y

125 xi(3,i) = ep_i(i)%pos%z

126 ai(1,i) = 0.0d0

127 ai(2,i) = 0.0d0

128 ai(3,i) = 0.0d0

129 pi(i) = 0.0d0

130 end do

131 do j=1,n_jp

132 xj(1,j) = ep_j(j)%pos%x

133 xj(2,j) = ep_j(j)%pos%y

134 xj(3,j) = ep_j(j)%pos%z

135 mj(j) = ep_j(j)%mass

136 end do

137 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

138 devid = omp_get_thread_num ()

139 ! [IMPORTANT NOTE]

140 ! The subroutine calc_gravity_psp is called by a OpenMP thread

141 ! in the FDPS. This means that here is already in the parallel

region.

142 ! So , you can use omp_get_thread_num () without !$OMP parallel

directives.

143 ! If you use them , a nested parallel resions is made and the

gravity

144 ! calculation will not be performed correctly.

145 #else

146 devid = 0

147 #endif

148 call g5_set_xmjMC(devid , 0, n_jp , xj, mj)

149 call g5_set_nMC(devid , n_jp)

150 call g5_calculate_force_on_xMC(devid , xi, ai, pi, n_ip)

151 do i=1,n_ip

152 f(i)%acc%x = f(i)%acc%x + ai(1,i)

153 f(i)%acc%y = f(i)%acc%y + ai(2,i)

154 f(i)%acc%z = f(i)%acc%z + ai(3,i)

155 f(i)%pot = f(i)%pot - pi(i)

156 end do

157 end subroutine calc_gravity_ep_sp

158 #elif defined(ENABLE_PIKG_KERNEL_X86)

159 subroutine calc_gravity_ep_ep(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

160 use , intrinsic :: iso_c_binding

161 use pikg_module_ep_ep

162 implicit none

163 integer(c_int), intent(in), value :: n_ip ,n_jp

164 type(full_particle), dimension(n_ip), intent(in) :: ep_i

165 type(full_particle), dimension(n_jp), intent(in) :: ep_j

166 type(full_particle), dimension(n_ip), intent(inout) :: f

167 !* Local variables

168 integer(c_int) :: i,j

169 type(epi_grav), dimension(n_ip), target :: ep_i_tmp

170 type(epj_grav), dimension(n_jp), target :: ep_j_tmp

171 type(force_grav), dimension(n_ip), target :: f_tmp

172

173 if (n_ip > 0) then

174 do i=1,n_ip

175 ep_i_tmp(i)%pos%x = ep_i(i)%pos%x - ep_i (1)%pos%x

176 ep_i_tmp(i)%pos%y = ep_i(i)%pos%y - ep_i (1)%pos%y

177 ep_i_tmp(i)%pos%z = ep_i(i)%pos%z - ep_i (1)%pos%z

178 f_tmp(i)%acc%x = 0.0

179 f_tmp(i)%acc%y = 0.0

180 f_tmp(i)%acc%z = 0.0

181 f_tmp(i)%pot = 0.0

182 end do

183 do j=1,n_jp

184 ep_j_tmp(j)%pos%x = ep_j(j)%pos%x - ep_i (1)%pos%x

185 ep_j_tmp(j)%pos%y = ep_j(j)%pos%y - ep_i (1)%pos%y

186 ep_j_tmp(j)%pos%z = ep_j(j)%pos%z - ep_i (1)%pos%z

187 ep_j_tmp(j)%mass = ep_j(j)%mass

188 end do

189 call pikg_calc_grav_ep_ep(c_loc(ep_i_tmp), n_ip , &

190 c_loc(ep_j_tmp), n_jp , &

191 c_loc(f_tmp))

192 do i=1,n_ip

193 f(i)%acc%x = f(i)%acc%x + f_tmp(i)%acc%x

194 f(i)%acc%y = f(i)%acc%y + f_tmp(i)%acc%y

195 f(i)%acc%z = f(i)%acc%z + f_tmp(i)%acc%z

196 f(i)%pot = f(i)%pot + f_tmp(i)%pot

197 end do

198 end if

199

200 end subroutine calc_gravity_ep_ep

201

202 subroutine calc_gravity_ep_sp(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

203 use , intrinsic :: iso_c_binding

204 use pikg_module_ep_ep

205 implicit none

206 integer(c_int), intent(in), value :: n_ip ,n_jp

207 type(full_particle), dimension(n_ip), intent(in) :: ep_i

208 type(fdps_spj_monopole), dimension(n_jp), intent(in) :: ep_j

209 type(full_particle), dimension(n_ip), intent(inout) :: f

210 !* Local variables

211 integer(c_int) :: i,j

212 type(epi_grav), dimension(n_ip), target :: ep_i_tmp

213 type(epj_grav), dimension(n_jp), target :: ep_j_tmp

214 type(force_grav), dimension(n_ip), target :: f_tmp

215

216 if (n_ip > 0) then

217 do i=1,n_ip

218 ep_i_tmp(i)%pos%x = ep_i(i)%pos%x - ep_i (1)%pos%x

219 ep_i_tmp(i)%pos%y = ep_i(i)%pos%y - ep_i (1)%pos%y

220 ep_i_tmp(i)%pos%z = ep_i(i)%pos%z - ep_i (1)%pos%z

221 f_tmp(i)%acc%x = 0.0

222 f_tmp(i)%acc%y = 0.0

223 f_tmp(i)%acc%z = 0.0

224 f_tmp(i)%pot = 0.0

225 end do

226 do j=1,n_jp

227 ep_j_tmp(j)%pos%x = ep_j(j)%pos%x - ep_i (1)%pos%x

228 ep_j_tmp(j)%pos%y = ep_j(j)%pos%y - ep_i (1)%pos%y

229 ep_j_tmp(j)%pos%z = ep_j(j)%pos%z - ep_i (1)%pos%z

230 ep_j_tmp(j)%mass = ep_j(j)%mass

231 end do

232 call pikg_calc_grav_ep_ep(c_loc(ep_i_tmp), n_ip , &

233 c_loc(ep_j_tmp), n_jp , &

234 c_loc(f_tmp))

235 do i=1,n_ip

236 f(i)%acc%x = f(i)%acc%x + f_tmp(i)%acc%x

237 f(i)%acc%y = f(i)%acc%y + f_tmp(i)%acc%y

238 f(i)%acc%z = f(i)%acc%z + f_tmp(i)%acc%z

239 f(i)%pot = f(i)%pot + f_tmp(i)%pot

240 end do

241 end if

242

243 end subroutine calc_gravity_ep_sp

244 #else

245 subroutine calc_gravity_ep_ep(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

246 implicit none

247 integer(c_int), intent(in), value :: n_ip ,n_jp

248 type(full_particle), dimension(n_ip), intent(in) :: ep_i

249 type(full_particle), dimension(n_jp), intent(in) :: ep_j

250 type(full_particle), dimension(n_ip), intent(inout) :: f

251 !* Local variables

252 integer(c_int) :: i,j

253 real(c_double) :: eps2 ,poti ,r3_inv ,r_inv

254 type(fdps_f64vec) :: xi ,ai ,rij

255

256 !* Compute force

257 eps2 = eps_grav * eps_grav

258 do i=1,n_ip

259 xi = ep_i(i)%pos

260 ai = 0.0d0

261 poti = 0.0d0

262 do j=1,n_jp

263 rij%x = xi%x - ep_j(j)%pos%x

264 rij%y = xi%y - ep_j(j)%pos%y

265 rij%z = xi%z - ep_j(j)%pos%z

266 r3_inv = rij%x*rij%x &

267 + rij%y*rij%y &

268 + rij%z*rij%z &

269 + eps2

270 r_inv = 1.0d0/sqrt(r3_inv)

271 r3_inv = r_inv * r_inv

272 r_inv = r_inv * ep_j(j)%mass

273 r3_inv = r3_inv * r_inv

274 ai%x = ai%x - r3_inv * rij%x

275 ai%y = ai%y - r3_inv * rij%y

276 ai%z = ai%z - r3_inv * rij%z

277 poti = poti - r_inv

278 ! [IMPORTANT NOTE]

279 ! In the innermost loop , we use the components of vectors

280 ! directly for vector operations because of the following

281 ! reasion. Except for intel compilers with ‘-ipo ‘ option ,

282 ! most of Fortran compilers use function calls to perform

283 ! vector operations like rij = x - ep_j(j)%pos.

284 ! This significantly slow downs the speed of the code.

285 ! By using the components of vector directly , we can avoid

286 ! these function calls.

287 end do

288 f(i)%pot = f(i)%pot + poti

289 f(i)%acc = f(i)%acc + ai

290 end do

291

292 end subroutine calc_gravity_ep_ep

293

294 !**** Interaction function (particle -super particle)

295 subroutine calc_gravity_ep_sp(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

296 implicit none

297 integer(c_int), intent(in), value :: n_ip ,n_jp

298 type(full_particle), dimension(n_ip), intent(in) :: ep_i

299 type(fdps_spj_monopole), dimension(n_jp), intent(in) :: ep_j

300 type(full_particle), dimension(n_ip), intent(inout) :: f

301 !* Local variables

302 integer(c_int) :: i,j

303 real(c_double) :: eps2 ,poti ,r3_inv ,r_inv

304 type(fdps_f64vec) :: xi ,ai ,rij

305

306 eps2 = eps_grav * eps_grav

307 do i=1,n_ip

308 xi = ep_i(i)%pos

309 ai = 0.0d0

310 poti = 0.0d0

311 do j=1,n_jp

312 rij%x = xi%x - ep_j(j)%pos%x

313 rij%y = xi%y - ep_j(j)%pos%y

314 rij%z = xi%z - ep_j(j)%pos%z

315 r3_inv = rij%x*rij%x &

316 + rij%y*rij%y &

317 + rij%z*rij%z &

318 + eps2

319 r_inv = 1.0d0/sqrt(r3_inv)

320 r3_inv = r_inv * r_inv

321 r_inv = r_inv * ep_j(j)%mass

322 r3_inv = r3_inv * r_inv

323 ai%x = ai%x - r3_inv * rij%x

324 ai%y = ai%y - r3_inv * rij%y

325 ai%z = ai%z - r3_inv * rij%z

326 poti = poti - r_inv

327 end do

328 f(i)%pot = f(i)%pot + poti

329 f(i)%acc = f(i)%acc + ai

330 end do

331

332 end subroutine calc_gravity_ep_sp

333 #endif

334

335 end module user_defined_types

Listing 34: N 体シミュレーションのサンプルコード (f main.F90)

1 !-----------------------------------------------------------------------

2 !/////////////////////// < M A I N R O U T I N E > ////////////////////

3 !-----------------------------------------------------------------------

4 subroutine f_main ()

5 use fdps_module

6 #if defined(ENABLE_PHANTOM_GRAPE_X86)

7 use phantom_grape_g5_x86

8 #endif

9 #if defined(ENABLE_PIKG_KERNEL_X86)

10 use pikg_module_ep_ep

11 #endif

12 use user_defined_types

13 implicit none

14 !* Local parameters

15 integer , parameter :: n_tot =2**10

16 ! -(force parameters)

17 real , parameter :: theta = 0.5

18 integer , parameter :: n_leaf_limit = 8

19 integer , parameter :: n_group_limit = 64

20 ! -(domain decomposition)

21 real , parameter :: coef_ema =0.3

22 ! -(timing parameters)

23 double precision , parameter :: time_end = 10.0d0

24 double precision , parameter :: dt = 1.0d0 /128.0 d0

25 double precision , parameter :: dt_diag = 1.0d0/8.0d0

26 double precision , parameter :: dt_snap = 1.0d0

27 !* Local variables

28 integer :: i,j,k,num_loop ,ierr

29 integer :: psys_num ,dinfo_num ,tree_num

30 integer :: n_loc

31 logical :: clear

32 double precision :: ekin0 ,epot0 ,etot0

33 double precision :: ekin1 ,epot1 ,etot1

34 double precision :: time_diag ,time_snap ,time_sys

35 double precision :: r,acc

36 real(c_float) :: eps2

37 type(fdps_controller) :: fdps_ctrl

38 type(full_particle), dimension (:), pointer :: ptcl

39 type(c_funptr) :: pfunc_ep_ep ,pfunc_ep_sp

40 ! -(IO)

41 character(len =64) :: fname

42 integer(c_int) :: np

43

44 !* Initialize FDPS

45 call fdps_ctrl%PS_Initialize ()

46

47 !* Create domain info object

48 call fdps_ctrl%create_dinfo(dinfo_num)

49 call fdps_ctrl%init_dinfo(dinfo_num ,coef_ema)

50

51 !* Create particle system object

52 call fdps_ctrl%create_psys(psys_num ,’full_particle ’)

53 call fdps_ctrl%init_psys(psys_num)

54

55 !* Make an initial condition

56 call setup_IC(fdps_ctrl ,psys_num ,n_tot)

57

58 !* Domain decomposition and exchange particle

59 call fdps_ctrl%decompose_domain_all(dinfo_num ,psys_num)

60 call fdps_ctrl%exchange_particle(psys_num ,dinfo_num)

61 n_loc = fdps_ctrl%get_nptcl_loc(psys_num)

62

63 !* Create tree object

64 call fdps_ctrl%create_tree(tree_num , &

65 "Long ,full_particle ,full_particle ,

full_particle ,Monopole")

66 call fdps_ctrl%init_tree(tree_num ,n_loc ,theta , &

67 n_leaf_limit ,n_group_limit)

68

69 #if defined(ENABLE_PHANTOM_GRAPE_X86)

70 call g5_open ()

71 call g5_set_eps_to_all(eps_grav);

72 #elif defined(ENABLE_PIKG_KERNEL_X86)

73 eps2 = eps_grav * eps_grav

74 call pikg_calc_grav_ep_ep_initialize(eps2)

75 #endif

76

77 !* Compute force at the initial time

78 pfunc_ep_ep = c_funloc(calc_gravity_ep_ep)

79 pfunc_ep_sp = c_funloc(calc_gravity_ep_sp)

80 call fdps_ctrl%calc_force_all_and_write_back(tree_num , &

81 pfunc_ep_ep , &

82 pfunc_ep_sp , &

83 psys_num , &

84 dinfo_num)

85 !* Compute energies at the initial time

86 clear = .true.

87 call calc_energy(fdps_ctrl ,psys_num ,etot0 ,ekin0 ,epot0 ,clear)

88

89 !* Time integration

90 time_diag = 0.0d0

91 time_snap = 0.0d0

92 time_sys = 0.0d0

93 num_loop = 0

94 do

95 !* Output

96 !if (fdps_ctrl%get_rank () == 0) then

97 ! write (* ,50) num_loop ,time_sys

98 ! 50 format(’(num_loop , time_sys) = ’,i5 ,1x,1es25 .16e3)

99 !end if

100 if ( (time_sys >= time_snap) .or. &

101 ((( time_sys + dt) - time_snap) > (time_snap - time_sys)) ) then

102 call output(fdps_ctrl ,psys_num)

103 time_snap = time_snap + dt_snap

104 end if

105

106 !* Compute energies and output the results

107 clear = .true.

108 call calc_energy(fdps_ctrl ,psys_num ,etot1 ,ekin1 ,epot1 ,clear)

109 if (fdps_ctrl%get_rank () == 0) then

110 if ( (time_sys >= time_diag) .or. &

111 ((( time_sys + dt) - time_diag) > (time_diag - time_sys)) )

then

112 write (* ,100) time_sys ,(etot1 -etot0)/etot0

113 100 format("time:␣" ,1es20 .10e3,",␣energy␣error:␣",1es20 .10e3)

114 time_diag = time_diag + dt_diag

115 end if

116 end if

117

118 !* Leapfrog: Kick -Drift

119 call kick(fdps_ctrl ,psys_num ,0.5d0*dt)

120 time_sys = time_sys + dt

121 call drift(fdps_ctrl ,psys_num ,dt)

122

123 !* Domain decomposition & exchange particle

124 if (mod(num_loop ,4) == 0) then

125 call fdps_ctrl%decompose_domain_all(dinfo_num ,psys_num)

126 end if

127 call fdps_ctrl%exchange_particle(psys_num ,dinfo_num)

128

129 !* Force calculation

130 pfunc_ep_ep = c_funloc(calc_gravity_ep_ep)

131 pfunc_ep_sp = c_funloc(calc_gravity_ep_sp)

132 call fdps_ctrl%calc_force_all_and_write_back(tree_num , &

133 pfunc_ep_ep , &

134 pfunc_ep_sp , &

135 psys_num , &

136 dinfo_num)

137 !* Leapfrog: Kick

138 call kick(fdps_ctrl ,psys_num ,0.5d0*dt)

139

140 !* Update num_loop

141 num_loop = num_loop + 1

142

143 !* Termination

144 if (time_sys >= time_end) then

145 exit

146 end if

147 end do

148

149 #if defined(ENABLE_PHANTOM_GRAPE_X86)

150 call g5_close ()

151 #endif

152

153 !* Finalize FDPS

154 call fdps_ctrl%PS_Finalize ()

155

156 end subroutine f_main

157

158 !-----------------------------------------------------------------------

159 !///////////////////////// S U B R O U T I N E /////////////////////////

160 !///////////////////////// < S E T U P _ I C > /////////////////////////

161 !-----------------------------------------------------------------------

162 subroutine setup_IC(fdps_ctrl ,psys_num ,nptcl_glb)

163 use fdps_vector

164 use fdps_module

165 use user_defined_types

166 implicit none

167 type(fdps_controller), intent(IN) :: fdps_ctrl

168 integer , intent(IN) :: psys_num ,nptcl_glb

169 !* Local parameters

170 double precision , parameter :: m_tot =1.0d0

171 double precision , parameter :: rmax =3.0d0,r2max=rmax*rmax

172 !* Local variables

173 integer :: i,j,k,ierr

174 integer :: nprocs ,myrank

175 double precision :: r2,cm_mass

176 type(fdps_f64vec) :: cm_pos ,cm_vel ,pos

177 type(full_particle), dimension (:), pointer :: ptcl

178 character(len =64) :: fname

179

180 !* Get # of MPI processes and rank number

181 nprocs = fdps_ctrl%get_num_procs ()

182 myrank = fdps_ctrl%get_rank ()

183

184 !* Make an initial condition at RANK 0

185 if (myrank == 0) then

186 !* Set # of local particles

187 call fdps_ctrl%set_nptcl_loc(psys_num ,nptcl_glb)

188

189 !* Create an uniform sphere of particles

190 !** get the pointer to full particle data

191 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

192 !** initialize Mersenne twister

193 call fdps_ctrl%MT_init_genrand (0)

194 do i=1, nptcl_glb

195 ptcl(i)%id = i

196 ptcl(i)%mass = m_tot/nptcl_glb

197 do

198 ptcl(i)%pos%x = (2.0d0*fdps_ctrl%MT_genrand_res53 () -1.0d0) *

rmax

199 ptcl(i)%pos%y = (2.0d0*fdps_ctrl%MT_genrand_res53 () -1.0d0) *

rmax

200 ptcl(i)%pos%z = (2.0d0*fdps_ctrl%MT_genrand_res53 () -1.0d0) *

rmax

201 r2 = ptcl(i)%pos*ptcl(i)%pos

202 if ( r2 < r2max ) exit

203 end do

204 ptcl(i)%vel = 0.0d0

205 end do

206

207 !* Correction

208 cm_pos = 0.0d0

209 cm_vel = 0.0d0

210 cm_mass = 0.0d0

211 do i=1, nptcl_glb

212 cm_pos = cm_pos + ptcl(i)%mass * ptcl(i)%pos

213 cm_vel = cm_vel + ptcl(i)%mass * ptcl(i)%vel

214 cm_mass = cm_mass + ptcl(i)%mass

215 end do

216 cm_pos = cm_pos/cm_mass

217 cm_vel = cm_vel/cm_mass

218 do i=1, nptcl_glb

219 ptcl(i)%pos = ptcl(i)%pos - cm_pos

220 ptcl(i)%vel = ptcl(i)%vel - cm_vel

221 end do

222

223 !* Output

224 !fname = ’initial.dat ’

225 !open(unit=9,file=trim(fname),action=’write ’,status=’replace ’, &

226 ! form=’unformatted ’,access=’stream ’)

227 !open(unit=9,file=trim(fname),action=’write ’,status=’replace ’)

228 ! do i=1, nptcl_glb

229 ! !write (9) ptcl(i)%pos%x,ptcl(i)%pos%y,ptcl(i)%pos%z

230 ! write (9,’(3es25 .16e3) ’)ptcl(i)%pos%x,ptcl(i)%pos%y,ptcl(i)%pos

%z

231 ! end do

232 !close(unit =9)

233

234 !* Release the pointer

235 nullify( ptcl )

236

237 else

238 call fdps_ctrl%set_nptcl_loc(psys_num ,0)

239 end if

240

241 !* Set the gravitational softening

242 eps_grav = 1.0d0/32.0 d0

243

244 end subroutine setup_IC

245

246 !-----------------------------------------------------------------------

247 !///////////////////////// S U B R O U T I N E /////////////////////////

248 !///////////////////////// < K I C K > /////////////////////////

249 !-----------------------------------------------------------------------

250 subroutine kick(fdps_ctrl ,psys_num ,dt)

251 use fdps_vector

252 use fdps_module

253 use user_defined_types

254 implicit none

255 type(fdps_controller), intent(IN) :: fdps_ctrl

256 integer , intent(IN) :: psys_num

257 double precision , intent(IN) :: dt

258 !* Local variables

259 integer :: i,nptcl_loc

260 type(full_particle), dimension (:), pointer :: ptcl

261

262 !* Get # of local particles

263 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

264

265 !* Get the pointer to full particle data

266 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

267 do i=1,nptcl_loc

268 ptcl(i)%vel = ptcl(i)%vel + ptcl(i)%acc * dt

269 end do

270 nullify(ptcl)

271

272 end subroutine kick

273

274 !-----------------------------------------------------------------------

275 !///////////////////////// S U B R O U T I N E /////////////////////////

276 !///////////////////////// < D R I F T > /////////////////////////

277 !-----------------------------------------------------------------------

278 subroutine drift(fdps_ctrl ,psys_num ,dt)

279 use fdps_vector

280 use fdps_module

281 use user_defined_types

282 implicit none

283 type(fdps_controller), intent(IN) :: fdps_ctrl

284 integer , intent(IN) :: psys_num

285 double precision , intent(IN) :: dt

286 !* Local variables

287 integer :: i,nptcl_loc

288 type(full_particle), dimension (:), pointer :: ptcl

289

290 !* Get # of local particles

291 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

292

293 !* Get the pointer to full particle data

294 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

295 do i=1,nptcl_loc

296 ptcl(i)%pos = ptcl(i)%pos + ptcl(i)%vel * dt

297 end do

298 nullify(ptcl)

299

300 end subroutine drift

301

302 !-----------------------------------------------------------------------

303 !////////////////////// S U B R O U T I N E //////////////////////

304 !////////////////////// < C A L C _ E N E R G Y > //////////////////////

305 !-----------------------------------------------------------------------

306 subroutine calc_energy(fdps_ctrl ,psys_num ,etot ,ekin ,epot ,clear)

307 use fdps_vector

308 use fdps_module

309 use user_defined_types

310 implicit none

311 type(fdps_controller), intent(IN) :: fdps_ctrl

312 integer , intent(IN) :: psys_num

313 double precision , intent(INOUT) :: etot ,ekin ,epot

314 logical , intent(IN) :: clear

315 !* Local variables

316 integer :: i,nptcl_loc

317 double precision :: etot_loc ,ekin_loc ,epot_loc

318 type(full_particle), dimension (:), pointer :: ptcl

319

320 !* Clear energies

321 if (clear .eqv. .true.) then

322 etot = 0.0d0

323 ekin = 0.0d0

324 epot = 0.0d0

325 end if

326

327 !* Get # of local particles

328 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

329 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

330

331 !* Compute energies

332 ekin_loc = 0.0d0

333 epot_loc = 0.0d0

334 do i=1,nptcl_loc

335 ekin_loc = ekin_loc + ptcl(i)%mass * ptcl(i)%vel * ptcl(i)%vel

336 epot_loc = epot_loc + ptcl(i)%mass * (ptcl(i)%pot + ptcl(i)%mass/

eps_grav)

337 end do

338 ekin_loc = ekin_loc * 0.5d0

339 epot_loc = epot_loc * 0.5d0

340 etot_loc = ekin_loc + epot_loc

341 call fdps_ctrl%get_sum(ekin_loc ,ekin)

342 call fdps_ctrl%get_sum(epot_loc ,epot)

343 call fdps_ctrl%get_sum(etot_loc ,etot)

344

345 !* Release the pointer

346 nullify(ptcl)

347

348 end subroutine calc_energy

349

350 !-----------------------------------------------------------------------

351 !///////////////////////// S U B R O U T I N E /////////////////////////

352 !///////////////////////// < O U T P U T > /////////////////////////

353 !-----------------------------------------------------------------------

354 subroutine output(fdps_ctrl ,psys_num)

355 use fdps_vector

356 use fdps_module

357 use user_defined_types

358 implicit none

359 type(fdps_controller), intent(IN) :: fdps_ctrl

360 integer , intent(IN) :: psys_num

361 !* Local parameters

362 character(len =16), parameter :: root_dir="result"

363 character(len =16), parameter :: file_prefix_1st="snap"

364 character(len =16), parameter :: file_prefix_2nd="proc"

365 !* Local variables

366 integer :: i,nptcl_loc

367 integer :: myrank

368 character(len =5) :: file_num ,proc_num

369 character(len =64) :: cmd ,sub_dir ,fname

370 type(full_particle), dimension (:), pointer :: ptcl

371 !* Static variables

372 integer , save :: snap_num =0

373

374 !* Get the rank number

375 myrank = fdps_ctrl%get_rank ()

376

377 !* Get # of local particles

378 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

379

380 !* Get the pointer to full particle data

381 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

382

383 !* Output

384 write(file_num ,"(i5.5)")snap_num

385 write(proc_num ,"(i5.5)")myrank

386 fname = trim(root_dir) // "/" &

387 // trim(file_prefix_1st) // file_num // "-" &

388 // trim(file_prefix_2nd) // proc_num // ".dat"

389 open(unit=9,file=trim(fname),action=’write ’,status=’replace ’)

390 do i=1, nptcl_loc

391 write (9 ,100) ptcl(i)%id,ptcl(i)%mass , &

392 ptcl(i)%pos%x,ptcl(i)%pos%y,ptcl(i)%pos%z, &

393 ptcl(i)%vel%x,ptcl(i)%vel%y,ptcl(i)%vel%z

394 100 format(i8 ,1x,7e25 .16e3)

395 end do

396 close(unit =9)

397 nullify(ptcl)

398

399 !* Update snap_num

400 snap_num = snap_num + 1

401

402 end subroutine output

5.2 固定長SPHシミュレーション

固定長 SPHシミュレーションのサンプルコードを以下に示す。このサンプルは第 3, 4節
で用いた固定長 SPHシミュレーションのサンプルコードと同じものである。これをカット&

ペーストしてコンパイルすれば、正常に動作する固定長 SPHシミュレーションコードを作
ることができる。

Listing 35: 固定長 SPHシミュレーションのサンプルコード (user defined.F90)

1 !================================

2 ! MODULE: User defined types

3 !================================

4 module user_defined_types

5 use , intrinsic :: iso_c_binding

6 use fdps_vector

7 implicit none

8

9 !* Private parameters

10 real(kind=c_double), parameter , private :: pi=datan (1.0d0)*4.0d0

11 !* Public parameters

12 real(kind=c_double), parameter , public :: kernel_support_radius =2.5d0

13

14 !**** Force types

15 type , public , bind(c) :: force_dens !$fdps Force

16 !$fdps clear smth=keep

17 real(kind=c_double) :: dens

18 real(kind=c_double) :: smth

19 end type force_dens

20

21 type , public , bind(c) :: force_hydro !$fdps Force

22 !$fdps clear

23 type(fdps_f64vec) :: acc

24 real(kind=c_double) :: eng_dot

25 real(kind=c_double) :: dt

26 end type force_hydro

27

28 !**** Full particle type

29 type , public , bind(c) :: full_particle !$fdps FP

30 !$fdps copyFromForce force_dens (dens ,dens)

31 !$fdps copyFromForce force_hydro (acc ,acc) (eng_dot ,eng_dot) (dt ,dt)

32 real(kind=c_double) :: mass !$fdps charge

33 type(fdps_f64vec) :: pos !$fdps position

34 type(fdps_f64vec) :: vel

35 type(fdps_f64vec) :: acc

36 real(kind=c_double) :: dens

37 real(kind=c_double) :: eng

38 real(kind=c_double) :: pres

39 real(kind=c_double) :: smth !$fdps rsearch

40 real(kind=c_double) :: snds

41 real(kind=c_double) :: eng_dot

42 real(kind=c_double) :: dt

43 integer(kind=c_long_long) :: id

44 type(fdps_f64vec) :: vel_half

45 real(kind=c_double) :: eng_half

46 end type full_particle

47

48 !**** Essential particle type

49 type , public , bind(c) :: essential_particle !$fdps EPI ,EPJ

50 !$fdps copyFromFP full_particle (id ,id) (pos ,pos) (vel ,vel) (mass ,

mass) (smth ,smth) (dens ,dens) (pres ,pres) (snds ,snds)

51 integer(kind=c_long_long) :: id !$fdps id

52 type(fdps_f64vec) :: pos !$fdps position

53 type(fdps_f64vec) :: vel

54 real(kind=c_double) :: mass !$fdps charge

55 real(kind=c_double) :: smth !$fdps rsearch

56 real(kind=c_double) :: dens

57 real(kind=c_double) :: pres

58 real(kind=c_double) :: snds

59 end type essential_particle

60

61 !* Public routines

62 public :: W

63 public :: gradW

64 public :: calc_density

65 public :: calc_hydro_force

66

67 contains

68

69 !-------------------------------------------------------------------

70 pure function W(dr,h)

71 implicit none

72 real(kind=c_double) :: W

73 type(fdps_f64vec), intent(in) :: dr

74 real(kind=c_double), intent(in) :: h

75 !* Local variables

76 real(kind=c_double) :: s,s1,s2

77

78 s = dsqrt(dr%x*dr%x &

79 +dr%y*dr%y &

80 +dr%z*dr%z)/h

81 s1 = 1.0d0 - s

82 if (s1 < 0.0d0) s1 = 0.0d0

83 s2 = 0.5d0 - s

84 if (s2 < 0.0d0) s2 = 0.0d0

85 W = (s1*s1*s1) - 4.0d0*(s2*s2*s2)

86 W = W * 16.0d0/(pi*h*h*h)

87

88 end function W

89

90 !-------------------------------------------------------------------

91 pure function gradW(dr ,h)

92 implicit none

93 type(fdps_f64vec) :: gradW

94 type(fdps_f64vec), intent(in) :: dr

95 real(kind=c_double), intent(in) :: h

96 !* Local variables

97 real(kind=c_double) :: dr_abs ,s,s1 ,s2 ,coef

98

99 dr_abs = dsqrt(dr%x*dr%x &

100 +dr%y*dr%y &

101 +dr%z*dr%z)

102 s = dr_abs/h

103 s1 = 1.0d0 - s

104 if (s1 < 0.0d0) s1 = 0.0d0

105 s2 = 0.5d0 - s

57



目 次

106 if (s2 < 0.0d0) s2 = 0.0d0

107 coef = - 3.0d0*(s1*s1) + 12.0d0*(s2*s2)

108 coef = coef * 16.0d0/(pi*h*h*h)

109 coef = coef / (dr_abs*h + 1.0d-6*h)

110 gradW%x = dr%x * coef

111 gradW%y = dr%y * coef

112 gradW%z = dr%z * coef

113

114 end function gradW

115

116 !**** Interaction function

117 subroutine calc_density(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

118 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

119 type(essential_particle), dimension(n_ip), intent(in) :: ep_i

120 type(essential_particle), dimension(n_jp), intent(in) :: ep_j

121 type(force_dens), dimension(n_ip), intent(inout) :: f

122 !* Local variables

123 integer(kind=c_int) :: i,j

124 type(fdps_f64vec) :: dr

125

126 do i=1,n_ip

127 f(i)%dens = 0.0d0

128 do j=1,n_jp

129 dr%x = ep_j(j)%pos%x - ep_i(i)%pos%x

130 dr%y = ep_j(j)%pos%y - ep_i(i)%pos%y

131 dr%z = ep_j(j)%pos%z - ep_i(i)%pos%z

132 f(i)%dens = f(i)%dens &

133 + ep_j(j)%mass * W(dr ,ep_i(i)%smth)

134 end do

135 end do

136

137 end subroutine calc_density

138

139 !**** Interaction function

140 subroutine calc_hydro_force(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

141 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

142 type(essential_particle), dimension(n_ip), intent(in) :: ep_i

143 type(essential_particle), dimension(n_jp), intent(in) :: ep_j

144 type(force_hydro), dimension(n_ip), intent(inout) :: f

145 !* Local parameters

146 real(kind=c_double), parameter :: C_CFL =0.3d0

147 !* Local variables

148 integer(kind=c_int) :: i,j

149 real(kind=c_double) :: mass_i ,mass_j ,smth_i ,smth_j , &

150 dens_i ,dens_j ,pres_i ,pres_j , &

151 snds_i ,snds_j

152 real(kind=c_double) :: povrho2_i ,povrho2_j , &

153 v_sig_max ,dr_dv ,w_ij ,v_sig ,AV

154 type(fdps_f64vec) :: pos_i ,pos_j ,vel_i ,vel_j , &

155 dr ,dv ,gradW_ij

156

157 do i=1,n_ip

158 !* Zero -clear

159 v_sig_max = 0.0d0

160 !* Extract i-particle info.

161 pos_i = ep_i(i)%pos

162 vel_i = ep_i(i)%vel

163 mass_i = ep_i(i)%mass

164 smth_i = ep_i(i)%smth

165 dens_i = ep_i(i)%dens

166 pres_i = ep_i(i)%pres

167 snds_i = ep_i(i)%snds

168 povrho2_i = pres_i /( dens_i*dens_i)

169 do j=1,n_jp

170 !* Extract j-particle info.

171 pos_j%x = ep_j(j)%pos%x

172 pos_j%y = ep_j(j)%pos%y

173 pos_j%z = ep_j(j)%pos%z

174 vel_j%x = ep_j(j)%vel%x

175 vel_j%y = ep_j(j)%vel%y

176 vel_j%z = ep_j(j)%vel%z

177 mass_j = ep_j(j)%mass

178 smth_j = ep_j(j)%smth

179 dens_j = ep_j(j)%dens

180 pres_j = ep_j(j)%pres

181 snds_j = ep_j(j)%snds

182 povrho2_j = pres_j /( dens_j*dens_j)

183 !* Compute dr & dv

184 dr%x = pos_i%x - pos_j%x

185 dr%y = pos_i%y - pos_j%y

186 dr%z = pos_i%z - pos_j%z

187 dv%x = vel_i%x - vel_j%x

188 dv%y = vel_i%y - vel_j%y

189 dv%z = vel_i%z - vel_j%z

190 !* Compute the signal velocity

191 dr_dv = dr%x * dv%x + dr%y * dv%y + dr%z * dv%z

192 if (dr_dv < 0.0d0) then

193 w_ij = dr_dv / sqrt(dr%x * dr%x + dr%y * dr%y + dr%z * dr%z

)

194 else

195 w_ij = 0.0d0

196 end if

197 v_sig = snds_i + snds_j - 3.0d0 * w_ij

198 v_sig_max = max(v_sig_max , v_sig)

199 !* Compute the artificial viscosity

200 AV = - 0.5d0*v_sig*w_ij / (0.5d0*( dens_i+dens_j))

201 !* Compute the average of the gradients of kernel

202 gradW_ij = 0.5d0 * (gradW(dr ,smth_i) + gradW(dr ,smth_j))

203 !* Compute the acceleration and the heating rate

204 f(i)%acc%x = f(i)%acc%x - mass_j *( povrho2_i+povrho2_j+AV)*

gradW_ij%x

205 f(i)%acc%y = f(i)%acc%y - mass_j *( povrho2_i+povrho2_j+AV)*

gradW_ij%y

206 f(i)%acc%z = f(i)%acc%z - mass_j *( povrho2_i+povrho2_j+AV)*

gradW_ij%z

207 f(i)%eng_dot = f(i)%eng_dot &

208 + mass_j * (povrho2_i + 0.5d0*AV) &

209 *(dv%x * gradW_ij%x &

210 +dv%y * gradW_ij%y &

211 +dv%z * gradW_ij%z)

212 end do

213 f(i)%dt = C_CFL *2.0d0*smth_i /( v_sig_max*kernel_support_radius)

214 end do

215 ! [IMPORTANT NOTE]

216 ! In the innermost loop , we use the components of vectors

217 ! directly for vector operations because of the following

218 ! reasion. Except for intel compilers with ‘-ipo ‘ option ,

219 ! most of Fortran compilers use function calls to perform

220 ! vector operations like rij = x - ep_j(j)%pos.

221 ! This significantly slow downs the speed of the code.

222 ! By using the components of vector directly , we can avoid

223 ! these function calls.

224

225 end subroutine calc_hydro_force

226

227 end module user_defined_types

Listing 36: 固定長 SPHシミュレーションのサンプルコード (f main.F90)

1 !-----------------------------------------------------------------------

2 !///////////////////// < M A I N R O U T I N E > /////////////////////

3 !-----------------------------------------------------------------------

4 subroutine f_main ()

5 use fdps_vector

6 use fdps_module

7 use user_defined_types

8 implicit none

9 !* Local parameters

10 ! -(force parameters)

11 real , parameter :: theta = 0.5

12 integer , parameter :: n_leaf_limit = 8

13 integer , parameter :: n_group_limit = 64

14 ! -(domain decomposition)

15 real , parameter :: coef_ema =0.3

16 ! -(IO)

17 integer , parameter :: output_interval =10

18 !* Local variables

19 integer :: i,j,k,ierr

20 integer :: nstep

21 integer :: psys_num ,dinfo_num

22 integer :: tree_num_dens ,tree_num_hydro

23 integer :: n_tot ,n_loc

24 logical :: clear

25 double precision :: time ,dt,end_time

26 type(fdps_f64vec) :: pos_ll ,pos_ul

27 type(fdps_controller) :: fdps_ctrl

28 type(full_particle), dimension (:), pointer :: ptcl

29 type(c_funptr) :: pfunc_ep_ep

30 ! -(IO)

31 character(len =64) :: filename

32 !* External routines

33 double precision , external :: get_timestep

34

35 !* Initialize FDPS

36 call fdps_ctrl%PS_Initialize ()

37

60



目 次

38 !* Make an instance of ParticleSystem and initialize it

39 call fdps_ctrl%create_psys(psys_num ,’full_particle ’)

40 call fdps_ctrl%init_psys(psys_num)

41

42 !* Make an initial condition and initialize the particle system

43 call setup_IC(fdps_ctrl ,psys_num ,end_time ,pos_ll ,pos_ul)

44

45 !* Make an instance of DomainInfo and initialize it

46 call fdps_ctrl%create_dinfo(dinfo_num)

47 call fdps_ctrl%init_dinfo(dinfo_num ,coef_ema)

48 call fdps_ctrl%set_boundary_condition(dinfo_num ,fdps_bc_periodic_xyz)

49 call fdps_ctrl%set_pos_root_domain(dinfo_num ,pos_ll ,pos_ul)

50

51 !* Perform domain decomposition and exchange particles

52 call fdps_ctrl%decompose_domain_all(dinfo_num ,psys_num)

53 call fdps_ctrl%exchange_particle(psys_num ,dinfo_num)

54

55 !* Make two tree structures

56 n_loc = fdps_ctrl%get_nptcl_loc(psys_num)

57 !** dens_tree (used for the density calculation)

58 call fdps_ctrl%create_tree(tree_num_dens , &

59 "Short ,force_dens ,essential_particle ,

essential_particle ,Gather")

60 call fdps_ctrl%init_tree(tree_num_dens ,n_loc ,theta , &

61 n_leaf_limit ,n_group_limit)

62

63 !** hydro_tree (used for the force calculation)

64 call fdps_ctrl%create_tree(tree_num_hydro , &

65 "Short ,force_hydro ,essential_particle ,

essential_particle ,Symmetry")

66 call fdps_ctrl%init_tree(tree_num_hydro ,n_loc ,theta , &

67 n_leaf_limit ,n_group_limit)

68

69 !* Compute density , pressure , acceleration due to pressure gradient

70 pfunc_ep_ep = c_funloc(calc_density)

71 call fdps_ctrl%calc_force_all_and_write_back(tree_num_dens , &

72 pfunc_ep_ep , &

73 psys_num , &

74 dinfo_num)

75 call set_pressure(fdps_ctrl ,psys_num)

76 pfunc_ep_ep = c_funloc(calc_hydro_force)

77 call fdps_ctrl%calc_force_all_and_write_back(tree_num_hydro , &

78 pfunc_ep_ep , &

79 psys_num , &

80 dinfo_num)

81 !* Get timestep

82 dt = get_timestep(fdps_ctrl ,psys_num)

83

84 !* Main loop for time integration

85 nstep = 0; time = 0.0d0

86 do

87 !* Leap frog: Initial Kick & Full Drift

88 call initial_kick(fdps_ctrl ,psys_num ,dt)

89 call full_drift(fdps_ctrl ,psys_num ,dt)

90

91 !* Adjust the positions of the SPH particles that run over

92 ! the computational boundaries.

93 call fdps_ctrl%adjust_pos_into_root_domain(psys_num ,dinfo_num)

94

95 !* Leap frog: Predict

96 call predict(fdps_ctrl ,psys_num ,dt)

97

98 !* Perform domain decomposition and exchange particles again

99 call fdps_ctrl%decompose_domain_all(dinfo_num ,psys_num)

100 call fdps_ctrl%exchange_particle(psys_num ,dinfo_num)

101

102 !* Compute density , pressure , acceleration due to pressure gradient

103 pfunc_ep_ep = c_funloc(calc_density)

104 call fdps_ctrl%calc_force_all_and_write_back(tree_num_dens , &

105 pfunc_ep_ep , &

106 psys_num , &

107 dinfo_num)

108 call set_pressure(fdps_ctrl ,psys_num)

109 pfunc_ep_ep = c_funloc(calc_hydro_force)

110 call fdps_ctrl%calc_force_all_and_write_back(tree_num_hydro , &

111 pfunc_ep_ep , &

112 psys_num , &

113 dinfo_num)

114

115 !* Get a new timestep

116 dt = get_timestep(fdps_ctrl ,psys_num)

117

118 !* Leap frog: Final Kick

119 call final_kick(fdps_ctrl ,psys_num ,dt)

120

121 !* Output result files

122 if (mod(nstep ,output_interval) == 0) then

123 call output(fdps_ctrl ,psys_num ,nstep)

124 call check_cnsrvd_vars(fdps_ctrl ,psys_num)

125 end if

126

127 !* Output information to STDOUT

128 if (fdps_ctrl%get_rank () == 0) then

129 write (* ,200)time ,nstep

130 200 format("================================"/ &

131 "time␣␣=␣",1es25 .16e3/ &

132 "nstep␣=␣",i6/ &

133 "================================")

134 end if

135

136 !* Termination condition

137 if (time >= end_time) exit

138

139 !* Update time & step

140 time = time + dt

141 nstep = nstep + 1

142

143 end do

144 call fdps_ctrl%ps_finalize ()

145 stop 0

146

147 !* Finalize FDPS

148 call fdps_ctrl%PS_Finalize ()

149

150 end subroutine f_main

151

152 !-----------------------------------------------------------------------

153 !///////////////////////// S U B R O U T I N E /////////////////////////

154 !///////////////////////// < S E T U P _ I C > /////////////////////////

155 !-----------------------------------------------------------------------

156 subroutine setup_IC(fdps_ctrl ,psys_num ,end_time ,pos_ll ,pos_ul)

157 use fdps_vector

158 use fdps_module

159 use user_defined_types

160 implicit none

161 type(fdps_controller), intent(IN) :: fdps_ctrl

162 integer , intent(IN) :: psys_num

163 double precision , intent(inout) :: end_time

164 type(fdps_f64vec) :: pos_ll ,pos_ul

165 !* Local variables

166 integer :: i,ii,irank

167 integer :: nprocs , myrank

168 integer :: nx_left , ny_left , nz_left

169 integer :: nx_right , ny_right , nz_right

170 integer :: nptcl_loc , nptcl_glb

171 integer :: nptcl_quot , nptcl_rem

172 integer :: i_first , i_last

173 double precision :: dens_left , dens_right

174 double precision :: eng_left , eng_right

175 double precision :: sv, mass

176 double precision :: x,y,z,dx,dy,dz

177 type(full_particle), dimension (:), pointer :: ptcl

178 character(len =5) :: proc_num

179 character(len =64) :: fname

180

181 !* Get # of MPI processes and rank number

182 nprocs = fdps_ctrl%get_num_procs ()

183 myrank = fdps_ctrl%get_rank ()

184

185 !* Set the box size

186 pos_ll%x = 0.0d0

187 pos_ll%y = 0.0d0

188 pos_ll%z = 0.0d0

189 pos_ul%x = 1.0d0

190 pos_ul%y = pos_ul%x / 8.0d0

191 pos_ul%z = pos_ul%x / 8.0d0

192

193 !* Set the left and right states

194 dens_left = 1.0d0

195 eng_left = 2.5d0

196 dens_right = 0.5d0

197 eng_right = 2.5d0

198

199 !* Set the separation of particle of the left state

200 dx = 1.0d0 / 128.0 d0

201 dy = dx

202 dz = dx

203

204 !* Count # of particles in the computational domain

205 !** (1) Left -half

206 nx_left = 0

207 x = 0.0d0

208 do

209 nx_left = nx_left + 1

210 x = x + dx

211 if (x >= 0.5d0*pos_ul%x) exit

212 end do

213 ny_left = 0

214 y = 0.0d0

215 do

216 ny_left = ny_left + 1

217 y = y + dy

218 if (y >= pos_ul%y) exit

219 end do

220 nz_left = 0

221 z = 0.0d0

222 do

223 nz_left = nz_left + 1

224 z = z + dz

225 if (z >= pos_ul%z) exit

226 end do

227 !** (2) Right -half

228 nx_right = 0

229 x = 0.5d0*pos_ul%x

230 do

231 nx_right = nx_right + 1

232 x = x + (dens_left/dens_right)*dx

233 if (x >= pos_ul%x) exit

234 end do

235 ny_right = 0

236 y = 0.0d0

237 do

238 ny_right = ny_right + 1

239 y = y + dy

240 if (y >= pos_ul%y) exit

241 end do

242 nz_right = 0

243 z = 0.0d0

244 do

245 nz_right = nz_right + 1

246 z = z + dz

247 if (z >= pos_ul%z) exit

248 end do

249 !** (3) calculate # of particles

250 nptcl_glb = (nx_left * ny_left * nz_left) &

251 + (nx_right * ny_right * nz_right)

252 if (myrank == 0) then

253 write (*,*)’nptcl_glb␣=␣’,nptcl_glb

254 end if

255

256 !* Set # of local particles

257 nptcl_quot = nptcl_glb / nprocs

258 nptcl_rem = mod(nptcl_glb , nprocs)

259 if (myrank == 0) then

260 write (*,*)’nptcl_quot␣=␣’,nptcl_quot

261 write (*,*)’nptcl_rem␣␣=␣’,nptcl_rem

262 end if

263 nptcl_loc = nptcl_quot

264 if (myrank < nptcl_rem) nptcl_loc = nptcl_loc + 1

265 call fdps_ctrl%set_nptcl_loc(psys_num , nptcl_loc)

266 i_first = 1 + nptcl_quot * myrank

267 if (myrank > (nptcl_rem -1)) then

268 i_first = i_first + nptcl_rem

269 else

270 i_first = i_first + myrank

271 end if

272 i_last = i_first + nptcl_loc - 1

273 do irank=0, nprocs -1

274 if (irank == myrank) then

275 write(*,"(4(a,1x,i8))")’myrank␣=’,myrank , &

276 ’,␣nptcl_loc␣=’,nptcl_loc , &

277 ’,␣i_first␣=’,i_first , &

278 ’,␣i_last␣=’,i_last

279 end if

280 call fdps_ctrl%barrier ()

281 end do

282

283 !* Set local particles

284 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

285 sv = (pos_ul%x * pos_ul%y * pos_ul%z) / nptcl_glb ! specific volume

286 mass = 0.5d0*( dens_left+dens_right) * sv

287 i = 1

288 !** (1) Left -half

289 x = 0.0d0

290 do

291 y = 0.0d0

292 do

293 z = 0.0d0

294 do

295 if (( i_first <= i) .and. (i <= i_last)) then

296 ii = i - i_first + 1

297 ptcl(ii)%mass = mass

298 ptcl(ii)%pos%x = x

299 ptcl(ii)%pos%y = y

300 ptcl(ii)%pos%z = z

301 ptcl(ii)%dens = dens_left

302 ptcl(ii)%eng = eng_left

303 ptcl(ii)%smth = kernel_support_radius * 0.012 d0

304 ptcl(ii)%id = i

305 end if

306 i = i + 1

307 z = z + dz

308 if (z >= pos_ul%z) exit

309 end do

310 y = y + dy

311 if (y >= pos_ul%y) exit

312 end do

313 x = x + dx

314 if (x >= 0.5d0*pos_ul%x) exit

315 end do

316 !** (2) Right -half

317 x = 0.5d0*pos_ul%x

318 do

319 y = 0.0d0

320 do

321 z = 0.0d0

322 do

323 if (( i_first <= i) .and. (i <= i_last)) then

324 ii = i - i_first + 1

325 ptcl(ii)%mass = mass

326 ptcl(ii)%pos%x = x

327 ptcl(ii)%pos%y = y

328 ptcl(ii)%pos%z = z

329 ptcl(ii)%dens = dens_right

330 ptcl(ii)%eng = eng_right

331 ptcl(ii)%smth = kernel_support_radius * 0.012 d0

332 ptcl(ii)%id = i

333 end if

334 i = i + 1

335 z = z + dz

336 if (z >= pos_ul%z) exit

337 end do

338 y = y + dy

339 if (y >= pos_ul%y) exit

340 end do

341 x = x + (dens_left/dens_right)*dx

342 if (x >= pos_ul%x) exit

343 end do

344

345 !* Check the initial distribution

346 !write(proc_num ,"(i5.5)") myrank

347 !fname = "initial" // proc_num // ".dat"

348 !open(unit=9,file=trim(fname),action=’write ’,status=’replace ’)

349 ! do i=1, nptcl_loc

350 ! write (9,’(3es25 .16e3) ’)ptcl(i)%pos%x, &

351 ! ptcl(i)%pos%y, &

352 ! ptcl(i)%pos%z

353 ! end do

354 !close(unit =9)

355

356 !* Set the end time

357 end_time = 0.12d0

358

359 !* Inform to STDOUT

360 if (fdps_ctrl%get_rank () == 0) then

361 write (*,*)"setup ..."

362 end if

363 !call fdps_ctrl%ps_finalize ()

364 !stop 0

365

366 end subroutine setup_IC

367

368 !-----------------------------------------------------------------------

369 !///////////////////// S U B R O U T I N E /////////////////////

370 !///////////////////// < G E T _ T I M E S T E P > /////////////////////

371 !-----------------------------------------------------------------------

372 function get_timestep(fdps_ctrl ,psys_num)

373 use fdps_vector

374 use fdps_module

375 use user_defined_types

376 implicit none

377 real(kind=c_double) :: get_timestep

378 type(fdps_controller), intent(in) :: fdps_ctrl

379 integer , intent(in) :: psys_num

380 !* Local variables

381 integer :: i,nptcl_loc

382 type(full_particle), dimension (:), pointer :: ptcl

383 real(kind=c_double) :: dt_loc

384

385 !* Get # of local particles

386 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

387

388 !* Get the pointer to full particle data

389 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

390 dt_loc = 1.0d30

391 do i=1,nptcl_loc

392 dt_loc = min(dt_loc , ptcl(i)%dt)

393 end do

394 nullify(ptcl)

395

396 !* Reduction

397 call fdps_ctrl%get_min_value(dt_loc ,get_timestep)

398

399 end function get_timestep

400

401 !-----------------------------------------------------------------------

402 !///////////////////// S U B R O U T I N E /////////////////////

403 !///////////////////// < I N I T I A L _ K I C K > /////////////////////

404 !-----------------------------------------------------------------------

405 subroutine initial_kick(fdps_ctrl ,psys_num ,dt)

406 use fdps_vector

407 use fdps_module

408 use user_defined_types

409 implicit none

410 type(fdps_controller), intent(in) :: fdps_ctrl

411 integer , intent(in) :: psys_num

412 double precision , intent(in) :: dt

413 !* Local variables

414 integer :: i,nptcl_loc

415 type(full_particle), dimension (:), pointer :: ptcl

416

417 !* Get # of local particles

418 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

419

420 !* Get the pointer to full particle data

421 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

422 do i=1,nptcl_loc

423 ptcl(i)%vel_half = ptcl(i)%vel + 0.5d0 * dt * ptcl(i)%acc

424 ptcl(i)%eng_half = ptcl(i)%eng + 0.5d0 * dt * ptcl(i)%eng_dot

425 end do

426 nullify(ptcl)

427

428 end subroutine initial_kick

429

430 !-----------------------------------------------------------------------

431 !/////////////////////// S U B R O U T I N E ///////////////////////

432 !/////////////////////// < F U L L _ D R I F T > ///////////////////////

433 !-----------------------------------------------------------------------

434 subroutine full_drift(fdps_ctrl ,psys_num ,dt)

435 use fdps_vector

436 use fdps_module

437 use user_defined_types

438 implicit none

439 type(fdps_controller), intent(in) :: fdps_ctrl

440 integer , intent(in) :: psys_num

441 double precision , intent(in) :: dt

442 !* Local variables

443 integer :: i,nptcl_loc

444 type(full_particle), dimension (:), pointer :: ptcl

445

446 !* Get # of local particles

447 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

448

449 !* Get the pointer to full particle data

450 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

451 do i=1,nptcl_loc

452 ptcl(i)%pos = ptcl(i)%pos + dt * ptcl(i)%vel_half

453 end do

454 nullify(ptcl)

455

456 end subroutine full_drift

457

458 !-----------------------------------------------------------------------

459 !///////////////////////// S U B R O U T I N E /////////////////////////

460 !///////////////////////// < P R E D I C T > /////////////////////////

461 !-----------------------------------------------------------------------

462 subroutine predict(fdps_ctrl ,psys_num ,dt)

463 use fdps_vector

464 use fdps_module

465 use user_defined_types

466 implicit none

467 type(fdps_controller), intent(in) :: fdps_ctrl

468 integer , intent(in) :: psys_num

469 double precision , intent(in) :: dt

470 !* Local variables

471 integer :: i,nptcl_loc

472 type(full_particle), dimension (:), pointer :: ptcl

473

474 !* Get # of local particles

475 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

476

477 !* Get the pointer to full particle data

478 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

479 do i=1,nptcl_loc

480 ptcl(i)%vel = ptcl(i)%vel + dt * ptcl(i)%acc

481 ptcl(i)%eng = ptcl(i)%eng + dt * ptcl(i)%eng_dot

482 end do

483 nullify(ptcl)

484

485 end subroutine predict

486

487 !-----------------------------------------------------------------------

488 !/////////////////////// S U B R O U T I N E ///////////////////////

489 !/////////////////////// < F I N A L _ K I C K > ///////////////////////

490 !-----------------------------------------------------------------------

491 subroutine final_kick(fdps_ctrl ,psys_num ,dt)

492 use fdps_vector

493 use fdps_module

494 use user_defined_types

495 implicit none

496 type(fdps_controller), intent(in) :: fdps_ctrl

497 integer , intent(in) :: psys_num

498 double precision , intent(in) :: dt

499 !* Local variables

500 integer :: i,nptcl_loc

501 type(full_particle), dimension (:), pointer :: ptcl

502

503 !* Get # of local particles

504 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

505

506 !* Get the pointer to full particle data

507 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

508 do i=1,nptcl_loc

509 ptcl(i)%vel = ptcl(i)%vel_half + 0.5d0 * dt * ptcl(i)%acc

510 ptcl(i)%eng = ptcl(i)%eng_half + 0.5d0 * dt * ptcl(i)%eng_dot

511 end do

512 nullify(ptcl)

513

514 end subroutine final_kick

515

516 !-----------------------------------------------------------------------

517 !///////////////////// S U B R O U T I N E /////////////////////

518 !///////////////////// < S E T _ P R E S S U R E > /////////////////////

519 !-----------------------------------------------------------------------

520 subroutine set_pressure(fdps_ctrl ,psys_num)

521 use fdps_vector

522 use fdps_module

523 use user_defined_types

524 implicit none

525 type(fdps_controller), intent(in) :: fdps_ctrl

526 integer , intent(in) :: psys_num

527 !* Local parameters

528 double precision , parameter :: hcr =1.4d0

529 !* Local variables

530 integer :: i,nptcl_loc

531 type(full_particle), dimension (:), pointer :: ptcl

532

533 !* Get # of local particles

534 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

535

536 !* Get the pointer to full particle data

537 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

538 do i=1,nptcl_loc

539 ptcl(i)%pres = (hcr - 1.0d0) * ptcl(i)%dens * ptcl(i)%eng

540 ptcl(i)%snds = dsqrt(hcr * ptcl(i)%pres / ptcl(i)%dens)

541 end do

542 nullify(ptcl)

543

544 end subroutine set_pressure

545

546 !-----------------------------------------------------------------------

547 !///////////////////////// S U B R O U T I N E /////////////////////////

548 !///////////////////////// < O U T P U T > /////////////////////////

549 !-----------------------------------------------------------------------

550 subroutine output(fdps_ctrl ,psys_num ,nstep)

551 use fdps_vector

552 use fdps_module

553 use user_defined_types

554 implicit none

555 type(fdps_controller), intent(IN) :: fdps_ctrl

556 integer , intent(IN) :: psys_num

557 integer , intent(IN) :: nstep

558 !* Local parameters

559 character(len =16), parameter :: root_dir="result"

560 character(len =16), parameter :: file_prefix_1st="snap"

561 character(len =16), parameter :: file_prefix_2nd="proc"

562 !* Local variables

563 integer :: i,nptcl_loc

564 integer :: myrank

565 character(len =5) :: file_num ,proc_num

566 character(len =64) :: cmd ,sub_dir ,fname

567 type(full_particle), dimension (:), pointer :: ptcl

568

569 !* Get the rank number

570 myrank = fdps_ctrl%get_rank ()

571

572 !* Get # of local particles

573 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

574

575 !* Get the pointer to full particle data

576 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

577

578 !* Output

579 write(file_num ,"(i5.5)")nstep

580 write(proc_num ,"(i5.5)")myrank

581 fname = trim(root_dir) // "/" &

582 // trim(file_prefix_1st) // file_num // "-" &

583 // trim(file_prefix_2nd) // proc_num // ".dat"

584 open(unit=9,file=trim(fname),action=’write ’,status=’replace ’)

585 do i=1, nptcl_loc

586 write (9 ,100) ptcl(i)%id,ptcl(i)%mass , &

587 ptcl(i)%pos%x,ptcl(i)%pos%y,ptcl(i)%pos%z, &

588 ptcl(i)%vel%x,ptcl(i)%vel%y,ptcl(i)%vel%z, &

589 ptcl(i)%dens ,ptcl(i)%eng ,ptcl(i)%pres

590 100 format(i8 ,1x,10 e25 .16e3)

591 end do

592 close(unit =9)

593 nullify(ptcl)

594

595 end subroutine output

596

597 !-----------------------------------------------------------------------

598 !//////////////// S U B R O U T I N E ////////////////

599 !//////////////// < C H E C K _ C N S R V D _ V A R S > ////////////////

600 !-----------------------------------------------------------------------

601 subroutine check_cnsrvd_vars(fdps_ctrl ,psys_num)

602 use fdps_vector

603 use fdps_module

604 use user_defined_types

605 implicit none

606 type(fdps_controller), intent(in) :: fdps_ctrl

607 integer , intent(in) :: psys_num

608 !* Local variables

609 integer :: i,nptcl_loc

610 type(full_particle), dimension (:), pointer :: ptcl

611 type(fdps_f64vec) :: mom_loc ,mom

612 real(kind=c_double) :: eng_loc ,eng

613

614 !* Get # of local particles

615 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

616

617 !* Get the pointer to full particle data

618 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

619 mom_loc = 0.0d0; eng_loc = 0.0d0

620 do i=1,nptcl_loc

621 mom_loc = mom_loc + ptcl(i)%vel * ptcl(i)%mass

622 eng_loc = eng_loc + ptcl(i)%mass &

623 *(ptcl(i)%eng &

624 +0.5d0*ptcl(i)%vel*ptcl(i)%vel)

625 end do

626 nullify(ptcl)

627

628 !* Reduction & output

629 call fdps_ctrl%get_sum(eng_loc ,eng)

630 call fdps_ctrl%get_sum(mom_loc%x,mom%x)

631 call fdps_ctrl%get_sum(mom_loc%y,mom%y)

632 call fdps_ctrl%get_sum(mom_loc%z,mom%z)

633 if (fdps_ctrl%get_rank () == 0) then

634 write (* ,100) eng

635 write (* ,100) mom%x

636 write (* ,100) mom%y

637 write (* ,100) mom%z

638 100 format (1es25 .16e3)

639 end if

640

641 end subroutine check_cnsrvd_vars

6 拡張機能の解説

6.1 P3Mコード

本節では、FDPSの拡張機能 Particle Mesh (以下、PMと省略する)の使用方法につい
て、P3M(Particle-Particle-Particle-Mesh)法のサンプルコードを用いて解説を行う。このサ
ンプルコードでは、塩化ナトリウム (NaCl)結晶の系全体の結晶エネルギーをP3M法で計算
し、結果を解析解と比較する。P3M法では、力、及び、ポテンシャルエネルギーの計算を、
Particle-Particle(PP)パートとParticle-Mesh(PM)パートに splitして行われる。このサンプ
ルコードではPPパートをFDPS標準機能を用いて計算し、PMパートをFDPS拡張機能を
用いて計算する。なお、拡張機能PMの仕様の詳細は、仕様書 9.2節で説明されているので、
そちらも参照されたい。

6.1.1 サンプルコードの場所と作業ディレクトリ

サンプルコードの場所は、$(FDPS)/sample/fortran/p3mである。まずは、そこに移動
する。� �
$ cd (FDPS)/sample/fortran/p3m� �
サンプルコードはユーザ定義型と相互作用関数が実装されたuser defined.F90、ユーザコー
ドのそれ以外の部分が実装された f main.F90、GCCと intelコンパイラ用のMakefileであ
る Makefileと Makefile.intelから構成される。

6.1.2 ユーザー定義型

本節では、FDPSの機能を用いて P3M法の計算を行うにあたって、ユーザーが記述しな
ければならない派生データ型について記述する。

6.1.2.1 FullParticle型

ユーザーは FullParticle型を記述しなければならない。Listing 37に、サンプルコードの
FullParticle型を示す。FullParticle型には、計算を行うにあたって、粒子が持っているべき
全ての物理量が含まれている必要がある。

Listing 37: FullParticle型
1 type , public , bind(c) :: nbody_fp !$fdps FP

2 !$fdps copyFromForce nbody_pp_results (pot ,pot) (agrv ,agrv)

3 !$fdps copyFromForcePM agrv_pm

4 integer(kind=c_long_long) :: id

5 real(kind=c_double) :: m !$fdps charge

6 real(kind=c_double) :: rc !$fdps rsearch

7 type(fdps_f64vec) :: x !$fdps position

73



目 次

8 type(fdps_f64vec) :: v,v_half

9 type(fdps_f64vec) :: agrv

10 real(kind=c_double) :: pot

11 type(fdps_f32vec) :: agrv_pm

12 real(kind=c_float) :: pot_pm

13 end type nbody_fp

この 派生データ型が FullParticle型であることを示すため、次の指示文を記述している:� �
type, public, bind(c) :: nbody_fp !$fdps FP� �

P3Mシミュレーションにおける相互作用はカットオフを持つ長距離力である。そのため、必
須物理量としてカットオフ半径が加わる。現在のFDPSの仕様では、カットオフ半径の指定
は探索半径の指定 (§ 4.2参照)と同様に行う。次の指示文は、どのメンバ変数がどの必須物
理量に対応するかを指定するものである:� �
real(kind=c_double) :: m !$fdps charge

real(kind=c_double) :: rc !$fdps rsearch

type(fdps_f64vec) :: x !$fdps position� �
FullParticle型は Force型との間でデータコピーを行う。ユーザは指示文を使い、FDPSに
データコピーの仕方を教えなければならない。また、拡張機能PMを用いて相互作用計算を
行う場合、FullParticle型には PMモジュールで計算した相互作用計算の結果をどのメンバ
変数で受け取るのか指示する指示文を記述する必要がある。本サンプルコードでは、以下の
ように記述している:� �
!$fdps copyFromForce nbody_pp_results (pot,pot) (agrv,agrv)

!$fdps copyFromForcePM agrv_pm� �
6.1.2.2 EssentialParticleI型

ユーザーはEssentialParticleI型を記述しなければならない。EssentialParticleI型には、PP
パートのForce計算を行う際、i粒子が持っているべき全ての物理量をメンバ変数として持っ
ている必要がある。また、本チュートリアル中では、EssentialParticleJ型も兼ねているため、
j粒子が持っているべき全ての物理量もメンバ変数として持っている必要がある。Listing 38

に、サンプルコードの EssentialParticleI型を示す。

Listing 38: EssentialParticleI型
1 type , public , bind(c) :: nbody_ep !$fdps EPI ,EPJ

2 !$fdps copyFromFP nbody_fp (id ,id) (m,m) (rc ,rc) (x,x)

3 integer(kind=c_long_long) :: id

4 real(kind=c_double) :: m !$fdps charge

5 real(kind=c_double) :: rc !$fdps rsearch

6 type(fdps_f64vec) :: x !$fdps position

7 end type nbody_ep

まず、ユーザは指示文を用いて、この派生データ型が EssentialParticleI型かつ Essential-

ParticleJ型であることをFDPSに教えなければならない。本サンプルコードでは、以下のよ
うに記述している:� �
type, public, bind(c) :: nbody_ep !$fdps EPI,EPJ� �
次に、ユーザはこの派生データ型のどのメンバ変数がどの必須物理量に対応するのかを指
示文によって指定しなければならない。今回は相互作用がカットオフ有りの長距離力である
ため、カットオフ半径の指定も必要である。本サンプルコードでは、以下のように記述して
いる:� �
real(kind=c_double) :: m !$fdps charge

real(kind=c_double) :: rc !$fdps rsearch

type(fdps_f64vec) :: x !$fdps position� �
EssentialParticleI 型と EssentialParticleJ 型は FullParticle 型からデータを受け取る。ユー
ザは FullParticle 型のどのメンバ変数を EssentialParticle?型 (?=I,J) のどのメンバ変数にコ
ピーするのかを、指示文を用いて指定する必要がある。本サンプルコードでは、以下のよう
に記述している:� �
!$fdps copyFromFP nbody_fp (id,id) (m,m) (rc,rc) (x,x)� �

6.1.2.3 Force型

ユーザーは Force型を記述しなければならない。Force型は、PPパートの Forceの計算を
行った際にその結果として得られる全ての物理量をメンバ変数として持っている必要があ
る。本サンプルコードの Force型を Listing 39に示す。このサンプルコードでは、Forceは
Coulomb相互作用計算のみであるため、Force型が 1つ用意されている。

Listing 39: Force型
1 type , public , bind(c) :: nbody_pp_results !$fdps Force

2 !$fdps clear

3 real(kind=c_double) :: pot

4 type(fdps_f64vec) :: agrv

5 end type nbody_pp_results

まず、ユーザはこの派生データ型が Force型であることを指示文によって指定する必要が
ある。本サンプルコードでは、以下のように記述している:� �
type, public, bind(c) :: nbody_pp_results !$fdps Force� �
この派生データ型は Force型であるから、ユーザは必ず、相互作用計算における積算対象の
メンバ変数の初期化方法を指定する必要がある。本サンプルコードでは Force型のすべての
メンバ変数にデフォルトの初期化方法を指定するため、単に、キーワード clearの指示文を

記述している:� �
!$fdps clear� �

6.1.2.4 相互作用関数 calcForceEpEp

ユーザーは相互作用関数 calcForceEpEpを記述しなければならない。calcForceEpEpには、
PPパートのForceの計算の具体的な内容を書く必要がある。calcForceEpEpは、サブルーチ
ン として実装されなければならない。引数は、EssentialParticleIの配列、EssentialParticleI

の個数、EssentialParticleJの配列、EssentialParticleJの個数、Force型の配列である。本サ
ンプルコードの calcForceEpEpの実装を Listing 40に示す。

Listing 40: 相互作用関数 calcForceEpEp

1 subroutine calc_force_ep_ep(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

2 integer(c_int), intent(in), value :: n_ip ,n_jp

3 type(nbody_ep), dimension(n_ip), intent(in) :: ep_i

4 type(nbody_ep), dimension(n_jp), intent(in) :: ep_j

5 type(nbody_pp_results), dimension(n_ip), intent(inout) :: f

6 !* Local variables

7 integer(c_int) :: i,j

8 real(c_double) :: rij ,rinv ,rinv3 ,xi

9 type(fdps_f64vec) :: dx

10

11 do i=1,n_ip

12 do j=1,n_jp

13 dx%x = ep_i(i)%x%x - ep_j(j)%x%x

14 dx%y = ep_i(i)%x%y - ep_j(j)%x%y

15 dx%z = ep_i(i)%x%z - ep_j(j)%x%z

16 rij = dsqrt(dx%x * dx%x &

17 +dx%y * dx%y &

18 +dx%z * dx%z)

19 if ((ep_i(i)%id == ep_j(j)%id) .and. (rij == 0.0d0)) cycle

20 rinv = 1.0d0/rij

21 rinv3 = rinv*rinv*rinv

22 xi = 2.0d0*rij/ep_i(i)%rc

23 f(i)%pot = f(i)%pot + ep_j(j)%m * S2_pcut(xi) * rinv

24 f(i)%agrv%x = f(i)%agrv%x + ep_j(j)%m * S2_fcut(xi) * rinv3 *

dx%x

25 f(i)%agrv%y = f(i)%agrv%y + ep_j(j)%m * S2_fcut(xi) * rinv3 *

dx%y

26 f(i)%agrv%z = f(i)%agrv%z + ep_j(j)%m * S2_fcut(xi) * rinv3 *

dx%z

27 end do

28 !* Self -interaction term

29 f(i)%pot = f(i)%pot - ep_i(i)%m * (208.0 d0 /(70.0 d0*ep_i(i)%rc))

30 end do

31

32 end subroutine calc_force_ep_ep

P3M法のPPパートは、(距離に関する)カットオフ付きの 2体相互作用である。そのため、
ポテンシャルと加速度の計算にカットオフ関数 (S2 pcut(), S2 fcut())が含まれていること

に注意されたい。ここで、各カットオフ関数は、粒子の形状関数 S(r)が S2(r)のときのカッ
トオフ関数である必要がある。ここで、S2(r)はHockney & Eastwood (1988)の式 (8.3)で
定義される形状関数で、以下の形を持つ:

S2(r) =


48

πa4

(a
2
− r

)
r < a/2,

0 otherwise.
(1)

ここで、rは粒子からの距離、aは形状関数のスケール長である。粒子の電荷量を qとすれ
ば、この粒子が作る電荷密度分布 ρ(r)は ρ(r) = q S2(r)と表現される。これは rに関して線
形な密度分布を仮定していることを意味する。PPパートのカットオフ関数が S2(r)を仮定
したものでなければならない理由は、PMパートのカットオフ関数が S2型形状関数を仮定
して実装されているためである (PMパートとPPパートのカットオフ関数は同じ形状関数に
基づく必要がある)。
カットオフ関数はユーザが定義する必要がある。サンプルコードの冒頭にS2 pcut()と S2

fcut()の実装例がある。これらの関数では、Hockney & Eastwood (1988)の式 (8-72),(8-75)

が使用されている。カットオフ関数は、PP相互作用が以下の形となるように定義されている:

ΦPP(r) =
m

|r − r′|
S2 pcut(ξ) (2)

fPP(r) =
m(r − r′)

|r − r′|3
S2 fcut(ξ) (3)

ここで、ξ = 2|r − r′|/aである。本サンプルコードでは aを変数 rc で表現している。
Hockney & Eastwood (1988)の式 (8-75)を見ると、r = 0のとき、メッシュポテンシャル

ϕmが次の有限値を持つことがわかる (ここで、1/4πε0の因子は省略した):

ϕm(0) =
208

70a
(4)

この項はサンプルコードの i粒子のループの最後で考慮されている:

1 f(i)%pot = f(i)%pot - ep_i(i)%m * (208.0 d0 /(70.0 d0*ep_i(i)%rc))

この項を考慮しないと解析解と一致しないことに注意する必要がある。

6.1.2.5 相互作用関数 calcForceEpSp

ユーザーは相互作用関数 calcForceEpSpを記述しなければならない注 5)。calcForceEpSpに
は、粒子-超粒子間相互作用計算の具体的な内容を書く必要がある。calcForceEpSp は、サブ
ルーチン として実装されなければならない。引数は、EssentialParticleIの配列、Essential-

ParticleIの個数、超粒子型の配列、超粒子型の個数、Force型の配列である。本サンプルコー
ドの calcForceEpSpの実装を Listing 41に示す。

注 5)冒頭で述べたように、本サンプルコードでは相互作用計算に P3M法を用いる。FDPSの枠組み内で、これ
を実現するため、後述するように、見込み角の基準値 θを 0に指定して相互作用計算を行う。このため、粒子-
超粒子相互作用は発生しないはずである。しかしながら、API calc force all and write back に、粒子-超
粒子間相互作用を計算する関数のアドレスを渡す必要があるため、相互作用関数 calcForceEpSp を定義する必
要がある。

Listing 41: 相互作用関数 calcForceEpSp

1 subroutine calc_force_ep_sp(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

2 integer(c_int), intent(in), value :: n_ip ,n_jp

3 type(nbody_ep), dimension(n_ip), intent(in) :: ep_i

4 type(fdps_spj_monopole_cutoff), dimension(n_jp), intent(in) :: ep_j

5 type(nbody_pp_results), dimension(n_ip), intent(inout) :: f

6 !* Local variables

7 integer(c_int) :: i,j

8 real(c_double) :: rij ,rinv ,rinv3 ,xi

9 type(fdps_f64vec) :: dx

10

11 do i=1,n_ip

12 do j=1,n_jp

13 dx%x = ep_i(i)%x%x - ep_j(j)%pos%x

14 dx%y = ep_i(i)%x%y - ep_j(j)%pos%y

15 dx%z = ep_i(i)%x%z - ep_j(j)%pos%z

16 rij = dsqrt(dx%x * dx%x &

17 +dx%y * dx%y &

18 +dx%z * dx%z)

19 rinv = 1.0d0/rij

20 rinv3 = rinv*rinv*rinv

21 xi = 2.0d0*rij/ep_i(i)%rc

22 f(i)%pot = f(i)%pot + ep_j(j)%mass * S2_pcut(xi) * rinv

23 f(i)%agrv%x = f(i)%agrv%x + ep_j(j)%mass * S2_fcut(xi) * rinv3

* dx%x

24 f(i)%agrv%y = f(i)%agrv%y + ep_j(j)%mass * S2_fcut(xi) * rinv3

* dx%y

25 f(i)%agrv%z = f(i)%agrv%z + ep_j(j)%mass * S2_fcut(xi) * rinv3

* dx%z

26 end do

27 end do

28

29 end subroutine calc_force_ep_sp

6.1.3 プログラム本体

本節では、サンプルコード本体について解説を行う。詳細な説明に入る前に、サンプルコー
ドの内容と全体構造について説明を与える。6.1節で述べたように、このサンプルコードで
はNaCl結晶の結晶エネルギーを P3M法によって計算し解析解と比較する。NaCl結晶は一
様格子状に並んだ粒子として表現される。NaとClは互い違いに並んでおり、Naに対応する
粒子は正の電荷を、Clに対応する粒子は負の電荷を持っている。この粒子で表現された結晶
を、大きさが [0, 1)3の周期境界ボックスの中に配置し、結晶エネルギーを計算する。結晶エ
ネルギーの計算精度は周期境界ボックスの中の粒子数や粒子の配置に依存するはずなので、
サンプルコードでは、これらを変えてエネルギーの相対誤差を調べ、結果をファイルに出力
する内容となっている。
コードの全体構造は以下のようになっている:

(1) FDPSで使用するオブジェクトの生成と初期化
(2) 指定された粒子数と配置の結晶を生成 (メインルーチンの setup NaCl crystal() )

(3) 各粒子のポテンシャルエネルギーを P3M法で計算 ( メインルーチン内 )

(4) 系全体のエネルギーを計算し、解析解と比較 (メインルーチンの calc energy error())

(5) (2)～(4)を繰り返す

以下で、個々について詳しく説明を行う。

6.1.3.1 fdps controller型オブジェクトの生成

FDPS Fortran インターフェースにおいて、FDPSのAPIはすべてFortran 2003のクラス
FDPS controllerのメンバ関数として提供される。このクラスは、インターフェースプログ
ラムの 1つである FDPS module.F90の中の、モジュール fdps module内で定義されている。
したがって、ユーザはFDPSのAPIを使用するために、FDPS controller型オブジェクトを
生成しなければならない。本サンプルコードでは、FDPS controller型オブジェクト fdps

ctrlをメインルーチンで生成している:

Listing 42: fdps controller型オブジェクトの生成
1 subroutine f_main ()

2 use fdps_module

3 implicit none

4 !* Local variables

5 type(fdps_controller) :: fdps_ctrl

6

7 ! Do something

8

9 end subroutine f_main

ここに示したコードは実際にサンプルコードから必要な部分だけを取り出したものであるこ
とに注意して頂きたい。
上記の理由から、以下の説明において、FDPSのAPIはこのオブジェクトのメンバ関数と
して呼び出されていることに注意されたい。

6.1.3.2 開始、終了

まずは、FDPSの初期化/開始を行う必要がある。次のように、メインルーチンに記述する。

Listing 43: FDPSの開始
1 fdps_ctrl%ps_initialize ();

FDPSは、開始したら明示的に終了させる必要がある。今回は、プログラムの終了と同時
に FDPSも終了させるため、メインルーチンの最後に次のように記述する。

Listing 44: FDPSの終了
1 fdps_ctrl%ps_finalize ();

79



目 次

6.1.3.3 オブジェクトの生成と初期化

FDPSの初期化に成功した場合、ユーザーはコード中で用いるオブジェクトを作成する必
要がある。本節では、オブジェクトの生成/初期化の仕方について、解説する。

6.1.3.3.1 オブジェクトの生成

P3M法の計算では、粒子群クラス、領域クラスに加え、PPパートの計算用の treeを 1本、
さらに PMパートの計算に必要な ParticleMeshオブジェクトの生成が必要である。

Listing 45: オブジェクトの生成
1 call fdps_ctrl%create_psys(psys_num ,’nbody_fp ’)

2 call fdps_ctrl%create_dinfo(dinfo_num)

3 call fdps_ctrl%create_pm(pm_num)

4 call fdps_ctrl%create_tree(tree_num , &

5 "Long ,nbody_pp_results ,nbody_ep ,nbody_ep ,

MonopoleWithCutoff")

ここに示したコードは実際にサンプルコードから該当箇所だけを取り出したものであること
に注意して頂きたい。

6.1.3.3.2 オブジェクトの初期化

ユーザーはオブジェクトを生成したら、そのオブジェクトを使用する前に、初期化を行う
必要がある。以下で、各オブジェクトの初期化の仕方について解説を行う。

(i) 粒子群オブジェクトの初期化 粒子群オブジェクトの初期化は、以下のように行う:

Listing 46: 粒子群オブジェクトの初期化
1 call fdps_ctrl%init_psys(psys_num)

サンプルコードではメインルーチンの冒頭で呼び出されている。

(ii) 領域オブジェクトの初期化 領域オブジェクトの初期化は、以下のように行う:

Listing 47: 領域オブジェクトの初期化
1 call fdps_ctrl%init_dinfo(dinfo_num ,coef_ema)

サンプルコードではメインルーチンの冒頭で呼び出されている。

初期化が完了した後、領域オブジェクトには境界条件と境界の大きさをセットする必要があ
る。サンプルコードでは、この作業は粒子分布を決定するサブルーチン setup NaCl crystal

の中で行われている:

1 call fdps_ctrl%set_boundary_condition(dinfo_num ,fdps_bc_periodic_xyz)

2 pos_ll%x = 0.0d0; pos_ll%y = 0.0d0; pos_ll%z = 0.0d0

3 pos_ul%x = 1.0d0; pos_ul%y = 1.0d0; pos_ul%z = 1.0d0

4 call fdps_ctrl%set_pos_root_domain(dinfo_num ,pos_ll ,pos_ul)

(iii) ツリーオブジェクトの初期化 相互作用ツリーオブジェクトの初期化も、API init -

tree を使って、以下のように行う:

Listing 48: ツリーオブジェクトの初期化
1 call fdps_ctrl%init_tree(tree_num ,3* nptcl_loc ,theta , &

2 n_leaf_limit ,n_group_limit)

ツリーオブジェクトの API init tree には引数として、大雑把な粒子数を渡す必要があ
る。これはAPIの第 2 引数として指定する。上記の例では、APIが呼ばれた時点でのロー
カル粒子数の 3倍の値がセットされるようになっている。一方、APIの第 3 引数は 省略可
能引数で 、tree法で力を計算するときの opening angle criterion θを指定する。本サンプル
コードではPPパートの計算で粒子-超粒子相互作用を発生させないようにするため、θ = 0

を指定している。

(iv) ParticleMeshオブジェクトの初期化 特に明示的に初期化を行う必要はない。

6.1.3.4 粒子分布の生成

本節では、粒子分布を生成するサブルーチン setup NaCl crystal の動作とその中で呼ば
れているFDPSのAPIについて解説を行う。このサブルーチンは、周期境界ボックスの 1次
元あたりの粒子数と、原点 (0, 0, 0)に最も近い粒子の座標を引数として、3次元粒子分布を
生成する。サンプルコードでは、これらのパラメータは派生データ型 crystal parameters

のオブジェクト NaCl paramsを使って渡されている:

1 ! In user_defined.F90

2 type , public , bind(c) :: crystal_parameters

3 integer(kind=c_int) :: nptcl_per_side

4 type(fdps_f64vec) :: pos_vertex

5 end type crystal_parameters

6 ! In f_main.F90

7 type(crystal_parameters) :: NaCl_params

8 call setup_NaCl_crystal(fdps_ctrl , &

9 psys_num , &

10 dinfo_num , &

11 NaCl_params)

setup NaCl crystal の前半部分において、渡されたパラメータを使って粒子分布を生成
している。この結晶の系全体のエネルギーは

E = −Nαm2

R0

(5)

と解析的に書ける。ここで、Nは分子の総数 (原子の数は2N個)、mは粒子の電荷量、R0は最
隣接原子間距離、αはマーデリング (Madelung)定数である。NaCl結晶の場合、α ≈ 1.747565

である (例えば、キッテル著「固体物理学入門 (第 8版)」を参照せよ)。計算結果をこの解析
解と比較するとき、系全体のエネルギーが粒子数に依存しては不便である。そこで、サンプ
ルコードでは、系全体のエネルギーが粒子数に依存しないように、粒子の電荷量mを

2Nm2

R0

= 1 (6)

となるようにスケールしていることに注意されたい。
粒子分布の生成後、FDPSのAPIを使って、領域分割と粒子交換を行っている。以下で、
これらのAPIについて解説する。

6.1.3.4.1 領域分割の実行

粒子分布に基いて領域分割を実行するには、領域オブジェクトのAPI decompose domain

all を使用する:

Listing 49: 領域分割の実行
1 call fdps_ctrl%decompose_domain_all(dinfo_num ,psys_num)

6.1.3.4.2 粒子交換の実行

領域情報に基いてプロセス間の粒子の情報を交換するには、粒子群オブジェクトの API

exchange particle を使用する:

Listing 50: 粒子交換の実行
1 call fdps_ctrl%exchange_particle(psys_num ,dinfo_num)

6.1.3.5 相互作用計算の実行

粒子分布を決定し、領域分割・粒子交換が終了したら、相互作用の計算を行う。サンプル
コードでは、この作業をメインルーチンで行っている:

Listing 51: 相互作用計算の実行
1 !* [4] Compute force and potential with P^{3}M method

2 !* [4-1] Get the pointer to FP and # of local particles

3 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

4 call fdps_ctrl%get_psys_fptr(psys_num ,ptcl)

5 !* [4-2] PP part

6 pfunc_ep_ep = c_funloc(calc_force_ep_ep)

7 pfunc_ep_sp = c_funloc(calc_force_ep_sp)

8 call fdps_ctrl%calc_force_all_and_write_back(tree_num , &

9 pfunc_ep_ep , &

10 pfunc_ep_sp , &

11 psys_num , &

12 dinfo_num)

13 !* [4-3] PM part

14 call fdps_ctrl%calc_pm_force_all_and_write_back(pm_num , &

15 psys_num , &

16 dinfo_num)

17 do i=1, nptcl_loc

18 pos32 = ptcl(i)%x

19 call fdps_ctrl%get_pm_potential(pm_num ,pos32 ,ptcl(i)%pot_pm)

20 end do

21 !* [4-4] Compute the total acceleration and potential

22 do i=1, nptcl_loc

23 ptcl(i)%pot = ptcl(i)%pot - ptcl(i)%pot_pm

24 ptcl(i)%agrv = ptcl(i)%agrv - ptcl(i)%agrv_pm

25 end do

PPパートの相互作用計算には API calc force all and write back 、PMパートの相
互作用計算にはAPI calc pm force all and write back を用いている。PMパートの計算
の後に、加速度とポテンシャルの総和を計算している。この総和計算を引

˜̃˜̃˜
き
˜̃˜̃˜
算
˜̃˜̃˜
で
˜̃˜̃˜
行ってい

ることに注意して頂きたい。引き算を使用する理由は、FDPSの拡張機能PMは重力を想定
してポテンシャルを計算するからである。すなわち、拡張機能PMでは、電荷m(> 0)は正
のポテンシャルを作るべきところを、質量m > 0の重力ポテンシャルとして計算する。この
ポテンシャルは負値である。したがって、拡張機能 PMを Coulomb相互作用で使用するに
は符
˜̃˜̃˜
号
˜̃˜̃˜
反
˜̃˜̃˜
転
˜̃˜̃˜
が
˜̃˜̃˜
必
˜̃˜̃˜
要
˜̃˜̃˜
となる。

6.1.3.6 エネルギー相対誤差の計算

エネルギーの相対誤差の計算は関数 calc energy errorで行っている。ここでは、解析
解の値としては E0 ≡ 2E = −1.7475645946332を採用した。これは、PM3(Particle-Mesh

Multipole Method)で数値的に求めた値である。

6.1.4 コンパイル

本サンプルコードではFFTWライブラリ (http://www.fftw.org)を使用するため、ユー
ザ環境に FFTW3をインストールする必要がある。コンパイルは、付属のMakefile内の変
数 FFTW LOCと FDPS LOCに、FFTWとFDPSのインストール先のPATHをそれぞれ指定し、
makeコマンドを実行すればよい。� �
$ make� �
コンパイルがうまく行けば、workディレクトリに実行ファイル p3m.xが作成されているは
ずである。

6.1.5 実行

FDPS拡張機能の仕様から、本サンプルコードはプロセス数が 2以上のMPI実行でなけ
れば、正常に動作しない。そこで、以下のコマンドでプログラムを実行する:� �
$ MPIRUN -np NPROC ./p3m.x� �
ここで、“MPIRUN”にはmpirunやmpiexecなどのmpi実行プログラムが、“NPROC”には
プロセス数が入る。

図 3: 1辺あたりの粒子数とエネルギー相対誤差の関係 (メッシュ数は 163、カットオフ半径
は 3/16)

6.1.6 結果の確認

計算が終了すると、workフォルダ下にエネルギーの相対誤差を記録したファイルが出力
される。結果をプロットすると、図 3のようになる。

7 より実用的なアプリケーションの解説

前節までは、比較的単純なサンプルコードを用いて、FDPSの基本的な機能について解説
を行ってきた。しかしながら、実際の研究では、複数の粒子種の取り扱いが必要になる等、
より複雑なアプリケーションを書く必要がある。そこで、本節では、より実用的なサンプル
コードを使って、FDPSの他の機能について解説を行う。記述を簡潔にするため、読者は前

節までの内容を理解しているものと仮定する。

7.1 N体/SPHコード

本節では、より実用的なアプリケーションの一例として円盤銀河の時間発展を計算する
N 体/SPHサンプルコードの解説を行う。このサンプルコードは、重力のみで相互作用する
ダークマターと星をN 体粒子、重力及び流体相互作用を行うガスを SPH粒子として表現す
る。重力計算はTree法で、流体計算は SPH法を用いて行う。SPH法はSpringel & Hernquist

[2002, MNRAS, 333, 649]及び Springel [2005, MNRAS, 364, 1105]で提案された方法 (以下、
簡単のため、Springelの方法と呼ぶ)を使用している。本解説を読むことにより、ユーザは、
FDPSで複数の粒子種を扱う方法を学ぶことができる。
以下では、まずはじめにコードの使用方法について説明を行う。次に Springelの方法につ
いて簡単な解説を行った後、サンプルコードの中身について具体的に解説する。

7.1.1 コードの使用方法

前節で述べた通り、本コードは円盤銀河のN 体/SPHシミュレーションを行うコードであ
る。ダークマターと星の初期条件は、銀河の初期条件を作成するソフトウェアMAGI (Miki

& Umemura [2018, MNRAS, 475, 2269])で作成したファイルを読み込んで設定する。一方、
ガスの初期条件は本コードの内部で作成する。したがって、以下の手順で本コードを使用
する。

• ディレクトリ$(FDPS)/sample/fortran/nbody+sphに移動

• カレントディレクトリにある Makefileを編集

• MAGIを使って初期条件に対応する粒子データを生成し、./magi data/dat以下に
配置

• コマンドライン上で makeを実行

• nbodysph.outファイルの実行

• 結果の解析

以下、順に説明していく。

7.1.1.1 ディレクトリ移動

サンプルコードの場所は、$(FDPS)/sample/fortran/nbody+sphである。まずは、そこに
移動する。

7.1.1.2 サンプルコードのファイル構成

以下はサンプルコードのファイル構成である。� �
$ ls | awk ’{print $0}’

Makefile

Makefile.K

Makefile.intel

Makefile.ofp

f_main.F90

ic.F90

job.K.sh

job.ofp.sh

leapfrog.F90

macro_defs.h

magi_data/

mathematical_constants.F90

physical_constants.F90

test.py

tipsy_file_reader.cpp

tipsy_file_reader.h

user_defined.F90� �
各ソースファイルの内容について簡単に解説を行う。まず、ic.F90には初期条件を作成す
るサブルーチンが実装されている。初期条件は円盤銀河の他、複数用意されている (後述)。
leapfrog.F90には粒子の軌道の時間積分を Leapfrog法を用いて行うサブルーチンが実装さ
れている。macro defs.hには計算を制御するためのマクロが定義されている。f main.F90

はアプリケーションのメインルーチンが実装されている。mathematical constants.F90に
は数学定数が、physical constants.F90には物理定数が定義されている。tipsy file -

reader.cpp 及び tipsy file reader.h にはMAGIが出力した粒子データを読むための関
数が定義されている。user defined.F90には、ユーザ定義型や相互作用関数が実装されて
いる。
ディレクトリmagi dataには、銀河の初期条件を作成するソフトウェアMAGIに入力するパ
ラメータファイル (magi data/cfg/*)とMAGIを動作させるスクリプト (magi data/sh/run.sh)

が格納されている。

7.1.1.3 Makefileの編集

Makefileの編集項目は以下の通りである。

• 変数 CXXに使用するC++コンパイラを代入する。

• 変数 FCに使用する Fortranコンパイラを代入する。

• 変数 CXXFLAGSにC++コンパイラのコンパイルオプションを指定する。

• 変数 FCFLAGSに Fortranコンパイラのコンパイルオプションを指定する。

• 本コードでは計算を制御するためにいくつかのマクロを用意している。表 1にマクロ
名とその定義の対応を示した。また、本コードには、INITIAL_CONDITIONの値に応じ
て自動的に設定されて使用されるマクロも存在する。これらは一般に変更する必要は
ないが、詳細は macro_defs.hを参照して頂きたい。

• 本コードでは重力計算にx86版Phantom-GRAPEライブラリを使用することができる。
Phantom-GRAPEライブラリを使用する場合、Makefileの変数 use_phantom_grape_

x86の値を yesにする。

OpenMPやMPIの使用/不使用の指定に関しては、第 3節を参照して頂きたい。

マクロ名 定義

INITIAL_CONDITION 初期条件の種類の指定、或いは、コードの動
作の指定に使用されるマクロ。0から 3までの
いずれかの値を取る必要がある。値に応じて、
次のように指定される。 0:円盤銀河の初期条
件を選択、1:Cold collapse 問題の初期条件を
選択、2:Evrard test問題の初期条件を選択、3:
ガラス状に分布した SPH粒子データを生成す
るモードで実行ファイルを作成

ENABLE_VARIABLE_SMOOTHING_LENGTH smoothing lengthが可変/固定を制御するマク
ロ。定義されている場合、可変となりSpringel

の方法でSPH計算が行われる。未定義の場合、
固定長カーネルの SPHコードとなる。

USE_ENTROPY 流体の熱力学状態を記述する独立変数として
エントロピーを使うか単位質量あたりの内部
エネルギーを使用するかを指定するマクロ。定
義されている場合エントロピーが用いられる。
但し、後述するマクロISOTHERMAL_EOSが定義
されている場合には、単位質量あたりの内部
エネルギーが強制的に使用される (圧力の計算
に内部エネルギーを使用する)。

USE_BALSARA_SWITCH Balsara switch (Balsara [1995, JCP, 121, 357])

の使用/不使用を制御するマクロ。定義されて
いる場合は使用する。

USE_PRESCR_OF_THOMAS_COUCHMAN_1992 Thomas & Couchman [1992, MNRAS,257, 11]

で提案されたSPH計算の tensile不安定を防ぐ
簡便な方法を使用するかを制御するマクロ。定
義されている場合は使用する。

ISOTHERMAL_EOS 流体を等温で取り扱うかどうかを指定するマ
クロ。定義されている場合は等温で扱い、未
定義の場合にはエントロピー方程式、或いは、
内部エネルギー方程式が解いて、熱力学的状
態を時間発展させる。

READ_DATA_WITH_BYTESWAP MAGIの粒子データを読み込む際に基本デー
タ型単位で byteswap してデータを読み込む
かを制御するマクロ。定義されている場合は
byteswapする。

表 1: コンパイル時マクロの種類と定義

7.1.1.4 MAGIを使った粒子データの生成

前述した通り、ユーザは事前に銀河の初期条件を作成するソフトウェアMAGIを使い、以
下に指定する手順でデータを作成する必要がある。MAGIを利用できないユーザは、指定
するサイトからこちらが用意したデータをダウンロードすることも可能である。以下、各場
合について詳しく述べる。

MAGIを使ってデータ作成を行う場合 以下の手順でデータ作成を行う。

1. https://bitbucket.org/ymiki/magiからMAGIをダウンロードし、Webの “How

to compile MAGI”に記載された手順に従って、適当な場所にインストールする。
但し、本サンプルコードは TIPSYファイルの粒子データ読み込みしかサポート
していないため、MAGIは USE_TIPSY_FORMAT=ONの状態でビルドされている必
要がある。

2. ./magi data/sh/run.shを開き、変数 MAGI INSTALL DIRにコマンド magiがイ
ンストールされたディレクトリを、変数 NTOTに希望するN 体粒子の総数をセッ
トする (ダークマターと星への振り分けはMAGIが自動的に行う)。

3. ./magi data/cfg/*を編集し、ダークマターと銀河のモデルを指定する。指定方
法の詳細は上記サイトか、或いは、Miki & Umemura [2018, MNRAS, 475, 2269]

の第 2.4節を参照のこと。デフォルトの銀河モデル (以下、デフォルトモデル)は
次の 4成分から構成される:

(i) ダークマターハロー (NFW profile, M = 1012 M⊙, rs = 21.5 kpc, rc =

200 kpc, ∆c = 10 kpc)

(ii) バルジ (Kingモデル, M = 5× 1010 M⊙, rs = 0.7 kpc, W0 = 5)

(iii) thick disk (Sérsic profile, M = 2.5 × 1010 M⊙, rs = 3.5 kpc, n = 1.5, zd =

1 kpc, QT,min = 1.0)

(iv) thin disk (exponential disk, M = 2.5 × 1010 M⊙, rs = 3.5 kpc, zd = 0.5 kpc,

QT,min = 1.0)

デフォルトモデルでは、2つの星円盤は barモードに対してわずかに不安定であ
るため、弱い棒状構造を持つ渦巻き銀河になることが期待される初期条件となっ
ている。MAGIの最新のリリース (version 1.1.1 [2019年 7月 19日時点])では、
従来のリリースとデフォルトの動作モードが変更されたため、thick disk と thin

disk のパラメータ指定の仕方が FDPS 5.0d 以前から変わっていることに注意さ
れたい。具体的には、従来のMAGIでは disk 成分の速度分散をパラメータ f を
通して指定する形になっていたが (FDPS 5.0d 以前に付属するサンプルコードで
は、f = 0.125)、最新のリリースでは Toomore Q value の最小値 QT,min を通し
て指定する方式になっている。

4. ディレクトリ magi dataに移動し、以下のコマンドを実行:� �
$ ./sh/run.sh� �

5. MAGIが正しく終了しているなら、magi_data/dat以下に、拡張子が tipsyの
粒子データが生成されているはずである。

データをダウンロードする場合 以下のサイトからダウンロードし、./magi data/dat/以下
に置く。各粒子データの銀河モデルはすべてデフォルトモデルで、粒子数だけ異なる。

• N = 221: https://v2.jmlab.jp/owncloud/index.php/s/XnzvW5XAYwfqZYQ/

download?path=%2Fmagi_data%2FGalaxy%2F21&files=Galaxy.tipsy

• N = 222: https://v2.jmlab.jp/owncloud/index.php/s/XnzvW5XAYwfqZYQ/

download?path=%2Fmagi_data%2FGalaxy%2F22&files=Galaxy.tipsy

• N = 223: https://v2.jmlab.jp/owncloud/index.php/s/XnzvW5XAYwfqZYQ/

download?path=%2Fmagi_data%2FGalaxy%2F23&files=Galaxy.tipsy

• N = 224: https://v2.jmlab.jp/owncloud/index.php/s/XnzvW5XAYwfqZYQ/

download?path=%2Fmagi_data%2FGalaxy%2F24&files=Galaxy.tipsy

7.1.1.5 makeの実行

makeコマンドを実行する。

7.1.1.6 実行

実行方法は以下の通りである。

• MPIを使用しない場合、コマンドライン上で以下のコマンドを実行する� �
$ ./nbodysph.out� �

• MPIを使用する場合、コマンドライン上で以下のコマンドを実行する� �
$ MPIRUN -np NPROC ./nbodysph.out� �
ここで、MPIRUNには mpirunや mpiexecなどが、NPROCには使用するMPIプロセスの
数が入る。

7.1.1.7 結果の解析

ディレクトリresultにN体粒子とSPH粒子の粒子データファイル “nbody0000x-proc0000y.dat”

と “sph0000x-proc0000y.dat”が出力される。ここで xは時刻に対応する整数、yはMPIのプ
ロセス番号 (ランク番号)を表す。 N 体粒子データの出力フォーマットは、1列目から順に
粒子の ID, 粒子の質量、位置の x, y, z座標、粒子の x, y, z軸方向の速度となっている。一
方、SPH粒子データの出力フォーマットは、1列目から順に粒子の ID, 粒子の質量、位置の

x, y, z座標、粒子の x, y, z軸方向の速度、密度、単位質量あたりの内部エネルギー、エント
ロピー、圧力となっている。
図 4は、N 体粒子数 221、SPH粒子数 218で円盤銀河のシミュレーションを行ったときの

T = 0.46における星分布とガス分布である。

図 4: T = 0.46における星分布 (左)とガス分布 (右) (計算は以下の条件で実施した: N 体粒
子数 221、SPH粒子数 218、等温、ガス温度 104 K、平均分子量 µ = 0.5)

以下では、まず Springelの方法について解説し、その後、サンプルコードの実装について
説明していく。

7.1.2 Springelの方法

Springel & Hernquist [2002, MNRAS, 333, 649]では、smoothing lengthが可変な場合で
も、系のエネルギーとエントロピーが保存するようなスキーム (具体的には運動方程式)を
定式化した。以下、彼らの定式化を手短に説明する。導出方針としては、smoothing length

も独立変数とみて系のLagrangianを立て、それを粒子数個の拘束条件の下、Euler-Lagrange

方程式を解く、というものである。
具体的には、彼らは系の Lagrangianとして次のようなもの選んだ:

L(q, q̇) =
1

2

N∑
i=1

miṙ
2
i −

1

γ − 1

N∑
i=1

miAiρ
γ−1
i (7)

ここで、q = (r1, ..., rN , h1, ...hN)であり、下付きの整数はすべて粒子番号を表す。riは位置、
hiは smoothing length、miは質量、γは比熱比、ρiは密度、Aiはエントロピー関数と呼ば
れ、単位質量あたりの内部エネルギー uiと次の関係がある:

ui =
Ai

γ − 1
ργ−1
i (8)

式 (7)の第 1項目は運動エネルギー、第 2項目は内部エネルギーを表す。この Lagrangianを
そのままEuler-Lagrangian方程式を使って解くと、4N 個の方程式になってしまうので、彼
らは次のN 個の拘束条件を導入した。

ϕi =
4π

3
h3
i ρi −mNneigh = 0 (9)

ここで、mは SPH粒子の平均質量注 6)、Nneighは近傍粒子数 (定数)である。この拘束条件の
下、Lagrangeの未定乗数法を使って、Euler-Lagrange方程式をとけば、以下の運動方程式が
得られる:

dvi

dt
= −

N∑
j=1

mj

[
fi
Pi

ρ2i
∇iW (rij, hi) + fj

Pj

ρ2j
∇iW (rij, hj)

]
(10)

ここで、Piは圧力、rij = |ri − rj|、W はカーネル関数、fiは∇h termと呼ばれる量で、

fi =

(
1 +

hi

3ρi

∂ρi
∂hi

)−1

(11)

と定義される。
系の熱力学的状態はエントロピーAiを独立変数として記述される。断熱過程の場合、エン
トロピーは衝撃波以外のところでは流れに沿って一定である。Springel [2005, MNRAS, 364,

1105]では、衝撃波でのエントロピー増加と速度の変化を人工粘性を使って次のようにモデ
ル化している:

dAi

dt
=

1

2

γ − 1

ργ−1
i

N∑
j=1

mjΠijvij · ∇iW ij (12)

dvi

dt

∣∣∣∣
visc

= −
N∑
j=1

mjΠij∇iW ij (13)

ここで、vij = vi − vj、viは速度、W ij = 1
2
(W (rij, hi) + W (rij, hj))である。Πij に関して

は、原論文を参照して頂きたい。
したがって、SPH計算の手順は次のようになる:� �
(1) 式 (9)と以下の式を無矛盾に解き、密度 ρiと hiを決定する。

ρi =
N∑
j=1

mjW (rij, hi) (14)

(2) 式 (11)で定義される∇h term を計算する。
(3) 式 (10)、(12)、(13)の右辺を計算する。
(4) SPH粒子の位置、速度、エントロピーを時間積分する。� �
以下、まずユーザ定義クラスと相互作用関数の実装について解説を行い、次にメインルー
チンの実装について解説を行う。複数粒子種の取扱は後者で解説する。
注 6)拘束条件に使用していることから、定数扱いであることに注意。

7.1.3 ユーザー定義型

本サンプルコードのユーザ定義型はすべて user_defined.F90 に定義されている。はじめ
に用意されているユーザ定義型の種類について簡単に説明しておく。冒頭で述べたように、
本サンプルコードは 2種類の粒子 (N 体粒子, SPH粒子)を扱う。そのため、FullParticle型
も 2種類用意している (N 体粒子用に 派生データ型 fp nbody を, SPH粒子用に 派生データ
型 fp sph )。相互作用は重力相互作用と流体相互作用の 2種類ある。そのため、Force型を
3種類用意している (重力計算用に 派生データ型 force grav を、密度計算用に 派生データ
型 force dens を、そして、圧力勾配による加速度 (以下、単に圧力勾配加速度)の計算用に
派生データ型 force hydro を; 第 4節も参照のこと)。本サンプルコードでは、簡単のため、
EssentialParticleI型と EssentialParticleJ型を 1つの派生データ型で兼ねることにし (以下、
単にEssentialParticle型)、密度計算と圧力勾配加速度計算に同じEssentialParticle型を使用
する。したがって、EssentialParticle型の種類は 2種類となっている (重力計算用に派生デー
タ型 ep grav を、SPH計算用に 派生データ型 ep hydro )。
以下、各ユーザ定義型の実装について説明する。

7.1.3.1 FullParticle型

まず、N 体粒子用の FullParticle型である 派生データ型 fp nbody について解説する。こ
の派生データ型には、N 体粒子が持っているべき全ての物理量が含まれている。Listing 52

に、この派生データ型の実装を示す。メンバ変数 の構成は、第 3-4節で紹介したN 体計算
サンプルコードとほぼ同じであり、詳細はそちらを参照されたい。

Listing 52: FullParticle型 (派生データ型 fp nbody)

1 !**** Full particle type

2 type , public , bind(c) :: fp_nbody !$fdps FP

3 !$fdps copyFromForce force_grav (acc ,acc) (pot ,pot)

4 integer(kind=c_long_long) :: id !$fdps id

5 real(kind=c_double) :: mass !$fdps charge

6 type(fdps_f64vec) :: pos !$fdps position

7 type(fdps_f64vec) :: vel

8 type(fdps_f64vec) :: acc

9 real(kind=c_double) :: pot

10 end type fp_nbody

次に、SPH粒子用のFullParticle型である派生データ型 fp sphについて解説する。この派
生データ型には、SPH粒子が持っているべき全ての物理量が含まれている。Listing 53に、こ
の派生データ型の実装を示す。メンバ変数の内、主要な変数の意味は次の通りである: id (識
別番号)、mass (質量)、pos (位置 [ri])、vel (速度 [vi])、acc grav (重力加速度)、pot grav

(重力ポテンシャル)、acc hydro (圧力勾配加速度)、dens (密度 [ρi])、eng (単位質量あたり
の内部エネルギー [ui])、ent (エントロピー関数 [以下、単にエントロピー][Ai])、pres (圧力
[Pi])、smth (smoothing length注 7)[hi])、gradh (∇h term[fi])、divv ((∇ · v)i、ここで下付き
の iは粒子 iの位置での微分を示している)、rotv ((∇× v)i)、 balsw (Balsara switchのた

注 7)カーネル関数が 0になる距離と定義。

めの係数で、定義式は Balsara [1995, JCP, 121, 357]の f(a))、snds (音速)、eng dot (eng

の時間変化率)、ent dot (entの時間変化率)、dt (この粒子の軌道を時間積分するときに許
される最大の時間刻み幅)。
以下の点に注意して頂きたい。

• SPH粒子が関わる相互作用計算は、重力計算、密度計算、圧力勾配加速度計算の 3種類あ
るので、それに応じて copyFromForce指示文も 3つ用意されている。

Listing 53: FullParticle型 (派生データ型 fp sph)

1 type , public , bind(c) :: fp_sph !$fdps FP

2 !$fdps copyFromForce force_grav (acc ,acc_grav) (pot ,pot_grav)

3 !$fdps copyFromForce force_dens (flag ,flag) (dens ,dens) (smth ,smth)

(gradh ,gradh) (divv ,divv) (rotv ,rotv)

4 !$fdps copyFromForce force_hydro (acc ,acc_hydro) (eng_dot ,eng_dot) (

ent_dot ,ent_dot) (dt ,dt)

5 integer(kind=c_long_long) :: id !$fdps id

6 real(kind=c_double) :: mass !$fdps charge

7 type(fdps_f64vec) :: pos !$fdps position

8 type(fdps_f64vec) :: vel

9 type(fdps_f64vec) :: acc_grav

10 real(kind=c_double) :: pot_grav

11 type(fdps_f64vec) :: acc_hydro

12 integer(kind=c_int) :: flag

13 real(kind=c_double) :: dens

14 real(kind=c_double) :: eng

15 real(kind=c_double) :: ent

16 real(kind=c_double) :: pres

17 real(kind=c_double) :: smth

18 real(kind=c_double) :: gradh

19 real(kind=c_double) :: divv

20 type(fdps_f64vec) :: rotv

21 real(kind=c_double) :: balsw

22 real(kind=c_double) :: snds

23 real(kind=c_double) :: eng_dot

24 real(kind=c_double) :: ent_dot

25 real(kind=c_double) :: dt

26 type(fdps_f64vec) :: vel_half

27 real(kind=c_double) :: eng_half

28 real(kind=c_double) :: ent_half

29 end type fp_sph

7.1.3.2 EssentialParticle型

まず、重力計算用のEssentialParticle型である 派生データ型 ep grav について解説する。
この派生データ型には、重力計算を行う際、i粒子と j粒子が持っているべき全ての物理量
をメンバ変数として持たせている。Listing 54に、この派生データ型の実装を示す。 ユー
ザはEssentialParticle型の定義部に、FullParticle型からのデータコピーの方法を指示するた
め指示文 (copyFromFP指示文)を書く必要があるが、本サンプルコードでは粒子種が 2種類
のため、2つの copyFromFP指示文が実装されていることに注意されたい。

Listing 54: EssentialParticle型 (派生データ型 ep grav)

1 type , public , bind(c) :: ep_grav !$fdps EPI ,EPJ

2 !$fdps copyFromFP fp_nbody (id ,id) (mass ,mass) (pos ,pos)

3 !$fdps copyFromFP fp_sph (id ,id) (mass ,mass) (pos ,pos)

4 integer(kind=c_long_long) :: id !$fdps id

5 real(kind=c_double) :: mass !$fdps charge

6 type(fdps_f64vec) :: pos !$fdps position

7 end type ep_grav

次に、密度計算と圧力勾配加速度計算用の EssentialParticle型である 派生データ型 ep

hydroについて解説する。この派生データ型には、密度計算と圧力勾配加速度計算を行う際、
i粒子と j粒子が持つべき全ての物理量をメンバ変数として持たせている。Listing 55に、こ
の派生データ型の実装を示す。

Listing 55: EssentialParticle型 (派生データ型 ep hydro)

1 type , public , bind(c) :: ep_hydro !$fdps EPI ,EPJ

2 !$fdps copyFromFP fp_sph (id ,id) (pos ,pos) (vel ,vel) (mass ,mass) (

smth ,smth) (dens ,dens) (pres ,pres) (gradh ,gradh) (snds ,snds) (

balsw ,balsw)

3 integer(kind=c_long_long) :: id !$fdps id

4 type(fdps_f64vec) :: pos !$fdps position

5 type(fdps_f64vec) :: vel

6 real(kind=c_double) :: mass !$fdps charge

7 real(kind=c_double) :: smth !$fdps rsearch

8 real(kind=c_double) :: dens

9 real(kind=c_double) :: pres

10 real(kind=c_double) :: gradh

11 real(kind=c_double) :: snds

12 real(kind=c_double) :: balsw

13 end type ep_hydro

7.1.3.3 Force型

まず、重力計算用の Force型である 派生データ型 force grav について解説する。この派
生データ型は、重力計算を行った際にその結果として得られる全ての物理量をメンバ変数と
して持っている必要がある。Listing 56にこの派生データ型の実装を示す。

Listing 56: Force型 (派生データ型 force grav)

1 type , public , bind(c) :: force_grav !$fdps Force

2 !$fdps clear

3 type(fdps_f64vec) :: acc

4 real(kind=c_double) :: pot

5 end type force_grav

次に、密度計算用のForce型である 派生データ型 force dens について解説する。この派
生データ型は、密度計算を行った際にその結果として得られる全ての物理量をメンバ変数と
して持っている必要がある。Listing 57に、この派生データ型の実装を示す。本サンプルコー
ドの SPH法では、smoothing lengthは固定ではなく、密度に応じて変化する。そのため、メ
ンバ変数として smthを持つ。また、密度計算と同時に、∇h termおよび (∇ · v)i、(∇× v)i

の計算を同時行うため、メンバ変数に gradh, divv, rotvを持つ。メンバ変数 flagは ρiと
hiを決定するイテレーション計算が収束したかどうかの結果を格納する変数である (詳細は、
相互作用関数の節を参照のこと)。

Listing 57: Force型 (派生データ型 force dens)

1 type , public , bind(c) :: force_dens !$fdps Force

2 !$fdps clear smth=keep

3 integer(kind=c_int) :: flag

4 real(kind=c_double) :: dens

5 real(kind=c_double) :: smth

6 real(kind=c_double) :: gradh

7 real(kind=c_double) :: divv

8 type(fdps_f64vec) :: rotv

9 end type force_dens

最後に、圧力勾配加速度計算用の Force型である 派生データ型 force hydro について解
説する。この派生データ型は、圧力勾配加速度計算を行った際にその結果として得られる全
ての物理量をメンバ変数として持っている必要がある。Listing 58に、この派生データ型の
実装を示す。

Listing 58: Force型 (派生データ型 force hydro)

1 type , public , bind(c) :: force_hydro !$fdps Force

2 !$fdps clear

3 type(fdps_f64vec) :: acc

4 real(kind=c_double) :: eng_dot

5 real(kind=c_double) :: ent_dot

6 real(kind=c_double) :: dt

7 end type force_hydro

7.1.4 相互作用関数

本サンプルコードで使用する相互作用関数はすべて user_defined.F90 に実装されてい
る。全部で 4種類あり、重力計算 (粒子間相互作用及び粒子-超粒子間相互作用)、密度計算、
圧力勾配加速度計算に使用される。以下、順に説明していく。

7.1.4.1 重力計算

重力計算用の相互作用関数は サブルーチン calc gravity ep ep及び calc gravity ep

sp として実装されている。Listing 59にこれらの実装を示す。実装は第 3-4節で紹介したN

体計算サンプルコードのものとほぼ同じであり、詳細はそちらを参照されたい。

Listing 59: 相互作用関数 (重力計算用)

1 #if defined(ENABLE_PHANTOM_GRAPE_X86)

2 subroutine calc_gravity_ep_ep(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

3 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

4 use omp_lib

5 #endif

6 use phantom_grape_g5_x86

7 implicit none

8 integer(c_int), intent(in), value :: n_ip ,n_jp

9 type(ep_grav), dimension(n_ip), intent(in) :: ep_i

10 type(ep_grav), dimension(n_jp), intent(in) :: ep_j

11 type(force_grav), dimension(n_ip), intent(inout) :: f

12 !* Local variables

13 integer(c_int) :: i,j

14 integer(c_int) :: nipipe ,njpipe ,devid

15 real(c_double), dimension(3,n_ip) :: xi,ai

16 real(c_double), dimension(n_ip) :: pi

17 real(c_double), dimension(3,n_jp) :: xj

18 real(c_double), dimension(n_jp) :: mj

19

20 nipipe = n_ip

21 njpipe = n_jp

22 do i=1,n_ip

23 xi(1,i) = ep_i(i)%pos%x

24 xi(2,i) = ep_i(i)%pos%y

25 xi(3,i) = ep_i(i)%pos%z

26 ai(1,i) = 0.0d0

27 ai(2,i) = 0.0d0

28 ai(3,i) = 0.0d0

29 pi(i) = 0.0d0

30 end do

31 do j=1,n_jp

32 xj(1,j) = ep_j(j)%pos%x

33 xj(2,j) = ep_j(j)%pos%y

34 xj(3,j) = ep_j(j)%pos%z

35 mj(j) = ep_j(j)%mass

36 end do

37 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

38 devid = omp_get_thread_num ()

39 ! [IMPORTANT NOTE]

40 ! The subroutine calc_gravity_ep_ep is called by a OpenMP thread

41 ! in the FDPS. This means that here is already in the parallel

region.

42 ! So , you can use omp_get_thread_num () without !$OMP parallel

directives.

43 ! If you use them , a nested parallel resions is made and the

gravity

44 ! calculation will not be performed correctly.

45 #else

46 devid = 0

47 #endif

48 call g5_set_xmjMC(devid , 0, n_jp , xj, mj)

49 call g5_set_nMC(devid , n_jp)

50 call g5_calculate_force_on_xMC(devid , xi, ai, pi, n_ip)

51 do i=1,n_ip

52 f(i)%acc%x = f(i)%acc%x + ai(1,i)

53 f(i)%acc%y = f(i)%acc%y + ai(2,i)

54 f(i)%acc%z = f(i)%acc%z + ai(3,i)

55 f(i)%pot = f(i)%pot - pi(i)

56 end do

57 end subroutine calc_gravity_ep_ep

58

59 subroutine calc_gravity_ep_sp(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

60 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

61 use omp_lib

62 #endif

63 use phantom_grape_g5_x86

64 implicit none

65 integer(c_int), intent(in), value :: n_ip ,n_jp

66 type(ep_grav), dimension(n_ip), intent(in) :: ep_i

67 type(fdps_spj_monopole), dimension(n_jp), intent(in) :: ep_j

68 type(force_grav), dimension(n_ip), intent(inout) :: f

69 !* Local variables

70 integer(c_int) :: i,j

71 integer(c_int) :: nipipe ,njpipe ,devid

72 real(c_double), dimension(3,n_ip) :: xi,ai

73 real(c_double), dimension(n_ip) :: pi

74 real(c_double), dimension(3,n_jp) :: xj

75 real(c_double), dimension(n_jp) :: mj

76

77 nipipe = n_ip

78 njpipe = n_jp

79 do i=1,n_ip

80 xi(1,i) = ep_i(i)%pos%x

81 xi(2,i) = ep_i(i)%pos%y

82 xi(3,i) = ep_i(i)%pos%z

83 ai(1,i) = 0.0d0

84 ai(2,i) = 0.0d0

85 ai(3,i) = 0.0d0

86 pi(i) = 0.0d0

87 end do

88 do j=1,n_jp

89 xj(1,j) = ep_j(j)%pos%x

90 xj(2,j) = ep_j(j)%pos%y

91 xj(3,j) = ep_j(j)%pos%z

92 mj(j) = ep_j(j)%mass

93 end do

94 #if defined(PARTICLE_SIMULATOR_THREAD_PARALLEL) && defined(_OPENMP)

95 devid = omp_get_thread_num ()

96 ! [IMPORTANT NOTE]

97 ! The subroutine calc_gravity_ep_sp is called by a OpenMP thread

98 ! in the FDPS. This means that here is already in the parallel

region.

99 ! So , you can use omp_get_thread_num () without !$OMP parallel

directives.

100 ! If you use them , a nested parallel resions is made and the

gravity

101 ! calculation will not be performed correctly.

102 #else

103 devid = 0

104 #endif

105 call g5_set_xmjMC(devid , 0, n_jp , xj, mj)

106 call g5_set_nMC(devid , n_jp)

107 call g5_calculate_force_on_xMC(devid , xi, ai, pi, n_ip)

108 do i=1,n_ip

109 f(i)%acc%x = f(i)%acc%x + ai(1,i)

98



目 次

110 f(i)%acc%y = f(i)%acc%y + ai(2,i)

111 f(i)%acc%z = f(i)%acc%z + ai(3,i)

112 f(i)%pot = f(i)%pot - pi(i)

113 end do

114 end subroutine calc_gravity_ep_sp

115 #else

116 subroutine calc_gravity_ep_ep(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

117 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

118 type(ep_grav), dimension(n_ip), intent(in) :: ep_i

119 type(ep_grav), dimension(n_jp), intent(in) :: ep_j

120 type(force_grav), dimension(n_ip), intent(inout) :: f

121 !* Local variables

122 integer(kind=c_int) :: i,j

123 real(kind=c_double) :: eps2 ,poti ,r3_inv ,r_inv

124 type(fdps_f64vec) :: xi ,ai ,rij

125 !* Compute force

126 eps2 = eps_grav * eps_grav

127 do i=1,n_ip

128 xi%x = ep_i(i)%pos%x

129 xi%y = ep_i(i)%pos%y

130 xi%z = ep_i(i)%pos%z

131 ai%x = 0.0d0

132 ai%y = 0.0d0

133 ai%z = 0.0d0

134 poti = 0.0d0

135 do j=1,n_jp

136 rij%x = xi%x - ep_j(j)%pos%x

137 rij%y = xi%y - ep_j(j)%pos%y

138 rij%z = xi%z - ep_j(j)%pos%z

139 r3_inv = rij%x*rij%x &

140 + rij%y*rij%y &

141 + rij%z*rij%z &

142 + eps2

143 r_inv = 1.0d0/dsqrt(r3_inv)

144 r3_inv = r_inv * r_inv

145 r_inv = r_inv * ep_j(j)%mass

146 r3_inv = r3_inv * r_inv

147 ai%x = ai%x - r3_inv * rij%x

148 ai%y = ai%y - r3_inv * rij%y

149 ai%z = ai%z - r3_inv * rij%z

150 poti = poti - r_inv

151 end do

152 f(i)%acc%x = f(i)%acc%x + ai%x

153 f(i)%acc%y = f(i)%acc%y + ai%y

154 f(i)%acc%z = f(i)%acc%z + ai%z

155 f(i)%pot = f(i)%pot + poti

156 end do

157 end subroutine calc_gravity_ep_ep

158

159 subroutine calc_gravity_ep_sp(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

160 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

161 type(ep_grav), dimension(n_ip), intent(in) :: ep_i

162 type(fdps_spj_monopole), dimension(n_jp), intent(in) :: ep_j

163 type(force_grav), dimension(n_ip), intent(inout) :: f

164 !* Local variables

165 integer(kind=c_int) :: i,j

166 real(kind=c_double) :: eps2 ,poti ,r3_inv ,r_inv

167 type(fdps_f64vec) :: xi ,ai ,rij

168 !* Compute force

169 eps2 = eps_grav * eps_grav

170 do i=1,n_ip

171 xi%x = ep_i(i)%pos%x

172 xi%y = ep_i(i)%pos%y

173 xi%z = ep_i(i)%pos%z

174 ai%x = 0.0d0

175 ai%y = 0.0d0

176 ai%z = 0.0d0

177 poti = 0.0d0

178 do j=1,n_jp

179 rij%x = xi%x - ep_j(j)%pos%x

180 rij%y = xi%y - ep_j(j)%pos%y

181 rij%z = xi%z - ep_j(j)%pos%z

182 r3_inv = rij%x*rij%x &

183 + rij%y*rij%y &

184 + rij%z*rij%z &

185 + eps2

186 r_inv = 1.0d0/dsqrt(r3_inv)

187 r3_inv = r_inv * r_inv

188 r_inv = r_inv * ep_j(j)%mass

189 r3_inv = r3_inv * r_inv

190 ai%x = ai%x - r3_inv * rij%x

191 ai%y = ai%y - r3_inv * rij%y

192 ai%z = ai%z - r3_inv * rij%z

193 poti = poti - r_inv

194 end do

195 f(i)%acc%x = f(i)%acc%x + ai%x

196 f(i)%acc%y = f(i)%acc%y + ai%y

197 f(i)%acc%z = f(i)%acc%z + ai%z

198 f(i)%pot = f(i)%pot + poti

199 end do

200 end subroutine calc_gravity_ep_sp

201 #endif

7.1.4.2 密度計算

密度計算用の相互作用関数は サブルーチン calc density として実装されている。Listing

60に、実装を示す。実装はマクロ ENABLE VARIABLE SMOOTHING LENGTHが定義されているか
否かで分かれる。このマクロが未定義の場合には、固定長カーネルコードとなり、実装は第
3-4節で紹介した SPHサンプルコードとほぼ同じであるので、そちらを参照されたい。以下、
このマクロが定義されている場合の実装について解説する。
第 7.1.2節で説明したように、密度 ρiと smoothing length hiは式 (14)と式 (9)を無矛盾
に解いて決定する必要がある。これには 2つの方程式を反復的に解く必要がある。このイテ
レーションを無限 do-enddo ループの中で行っている。本サンプルコードでは ρiと hiの計算
を効率的に行うため、smoothing lengthの値を定数 scf smth 倍してから密度計算を実行し
ている 。このため、定数倍する前の smoothing lengthの値を hi,0とすると、このイテレー

ションの間に hiを 0から hmax,alw ≡ scf smth × hi,0までの間なら変化させてもよいことに
なる。なぜなら、j粒子リストの取りこぼしは発生しないからである。逆にこの範囲でイテ
レーションが収束しなければ、求めたい hiは hmax,alwよりも大きいということになり、既存
の j粒子リストでは ρiと hiを決定できないということになる。この場合、hi,0を大きくし
た上で、密度計算をやり直す必要がある。この外側のイテレーションは f_main.F90の サブ
ルーチン calc density wrapper で行われている。このサブルーチンの詳細は第 7.1.5節で
行う。
無限 do-enddo ループの後には、∇h termの計算、(∇ · v)i及び (∇× v)iの計算を行って
いる。

Listing 60: 相互作用関数 (密度計算用)

1 subroutine calc_density(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

2 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

3 type(ep_hydro), dimension(n_ip), intent(in) :: ep_i

4 type(ep_hydro), dimension(n_jp), intent(in) :: ep_j

5 type(force_dens), dimension(n_ip), intent(inout) :: f

6 !* Local parameters

7 real(kind=c_double), parameter :: eps =1.0d-6

8 !* Local variables

9 integer(kind=c_int) :: i,j

10 integer(kind=c_int) :: n_unchanged

11 real(kind=c_double) :: M,M_trgt

12 real(kind=c_double) :: dens ,drho_dh

13 real(kind=c_double) :: h,h_max_alw ,h_L ,h_U ,dh ,dh_prev

14 type(fdps_f64vec) :: dr ,dv ,gradW_i

15

16 #if defined(ENABLE_VARIABLE_SMOOTHING_LENGTH)

17 real(kind=c_double), dimension(n_jp) :: mj ,rij

18 M_trgt = mass_avg * N_neighbor

19 do i=1,n_ip

20 dens = 0.0d0

21 h_max_alw = ep_i(i)%smth ! maximum allowance

22 h = h_max_alw / SCF_smth

23 ! Note that we increase smth by a factor of scf_smth

24 ! before calling calc_density ().

25 h_L = 0.0d0

26 h_U = h_max_alw

27 dh_prev = 0.0d0

28 n_unchanged = 0

29 ! Software cache

30 do j=1,n_jp

31 mj(j) = ep_j(j)%mass

32 dr%x = ep_i(i)%pos%x - ep_j(j)%pos%x

33 dr%y = ep_i(i)%pos%y - ep_j(j)%pos%y

34 dr%z = ep_i(i)%pos%z - ep_j(j)%pos%z

35 rij(j) = dsqrt(dr%x * dr%x &

36 +dr%y * dr%y &

37 +dr%z * dr%z)

38 end do

39 iteration_loop: do

40 ! Calculate density

41 dens = 0.0d0

42 do j=1,n_jp

43 dens = dens + mj(j) * W(rij(j), h)

44 end do

45 ! Check if the current value of the smoohting length

satisfies

46 ! Eq.(5) in Springel (2005).

47 M = 4.0d0 * pi * h * h * h * dens / 3.0d0

48 if ((h < h_max_alw) .and. (dabs(M/M_trgt - 1.0d0) < eps))

then

49 ! In this case , Eq.(5) holds within a specified accuracy

.

50 f(i)%flag = 1

51 f(i)%dens = dens

52 f(i)%smth = h

53 exit iteration_loop

54 end if

55 if (((h == h_max_alw) .and. (M < M_trgt)) .or. (n_unchanged

== 4)) then

56 ! In this case , we skip this particle forcibly.

57 ! In order to determine consistently the density

58 ! and the smoohting length for this particle ,

59 ! we must re -perform calcForceAllAndWriteBack ().

60 f(i)%flag = 0

61 f(i)%dens = dens

62 f(i)%smth = h_max_alw

63 exit iteration_loop

64 end if

65 ! Update h_L & h_U

66 if (M < M_trgt) then

67 if (h_L < h) h_L = h

68 else if (M_trgt < M) then

69 if (h < h_U) h_U = h

70 end if

71 dh = h_U - h_L

72 if (dh == dh_prev) then

73 n_unchanged = n_unchanged + 1

74 else

75 dh_prev = dh

76 n_unchanged = 0

77 end if

78 ! Update smoothing length

79 h = ((3.0d0 * M_trgt)/(4.0d0 * pi * dens))**(1.0 d0/3.0d0)

80 if ((h <= h_L) .or. (h == h_U)) then

81 ! In this case , we switch to the bisection search.

82 ! The inclusion of ’=’ in the if statement is very

83 ! important to escape a limit cycle.

84 h = 0.5d0 * (h_L + h_U)

85 else if (h_U < h) then

86 h = h_U

87 end if

88 end do iteration_loop

89 ! Calculate grad -h term

90 if (f(i)%flag == 1) then

91 drho_dh = 0.0d0

92 do j=1,n_jp

93 drho_dh = drho_dh + mj(j) * dWdh(rij(j), h)

94 end do

95 f(i)%gradh = 1.0d0 / (1.0d0 + (h * drho_dh) / (3.0d0 * dens)

)

96 else

97 f(i)%gradh = 1.0d0 ! dummy value

98 end if

99 ! Compute \div v & \rot v for Balsara switch

100 #if defined(USE_BALSARA_SWITCH)

101 do j=1,n_jp

102 dr%x = ep_i(i)%pos%x - ep_j(j)%pos%x

103 dr%y = ep_i(i)%pos%y - ep_j(j)%pos%y

104 dr%z = ep_i(i)%pos%z - ep_j(j)%pos%z

105 dv%x = ep_i(i)%vel%x - ep_j(j)%vel%x

106 dv%y = ep_i(i)%vel%y - ep_j(j)%vel%y

107 dv%z = ep_i(i)%vel%z - ep_j(j)%vel%z

108 gradW_i = gradW(dr, f(i)%smth)

109 f(i)%divv = f(i)%divv - mj(j) * (dv%x * gradW_i%x &

110 +dv%y * gradW_i%y &

111 +dv%z * gradW_i%z)

112 f(i)%rotv%x = f(i)%rotv%x - mj(j) * (dv%y * gradW_i%z - dv%z

* gradW_i%y)

113 f(i)%rotv%y = f(i)%rotv%y - mj(j) * (dv%z * gradW_i%x - dv%x

* gradW_i%z)

114 f(i)%rotv%z = f(i)%rotv%z - mj(j) * (dv%x * gradW_i%y - dv%y

* gradW_i%x)

115 end do

116 f(i)%divv = f(i)%divv / f(i)%dens

117 f(i)%rotv%x = f(i)%rotv%x / f(i)%dens

118 f(i)%rotv%y = f(i)%rotv%y / f(i)%dens

119 f(i)%rotv%z = f(i)%rotv%z / f(i)%dens

120 #endif

121 end do

122 #else

123 double precision :: mj ,rij

124 do i=1,n_ip

125 f(i)%dens = 0.0d0

126 do j=1,n_jp

127 dr%x = ep_j(j)%pos%x - ep_i(i)%pos%x

128 dr%y = ep_j(j)%pos%y - ep_i(i)%pos%y

129 dr%z = ep_j(j)%pos%z - ep_i(i)%pos%z

130 rij = dsqrt(dr%x * dr%x &

131 +dr%y * dr%y &

132 +dr%z * dr%z)

133 f(i)%dens = f(i)%dens &

134 + ep_j(j)%mass * W(rij ,ep_i(i)%smth)

135 end do

136 f(i)%smth = ep_i(i)%smth

137 f(i)%gradh = 1.0d0

138 ! Compute \div v & \rot v for Balsara switch

139 #if defined(USE_BALSARA_SWITCH)

140 do j=1,n_jp

141 mj = ep_j(j)%mass

142 dr%x = ep_i(i)%pos%x - ep_j(j)%pos%x

143 dr%y = ep_i(i)%pos%y - ep_j(j)%pos%y

144 dr%z = ep_i(i)%pos%z - ep_j(j)%pos%z

145 dv%x = ep_i(i)%vel%x - ep_j(j)%vel%x

146 dv%y = ep_i(i)%vel%y - ep_j(j)%vel%y

147 dv%z = ep_i(i)%vel%z - ep_j(j)%vel%z

148 gradW_i = gradW(dr, f(i)%smth)

149 f(i)%divv = f(i)%divv - mj * (dv%x * gradW_i%x &

150 +dv%y * gradW_i%y &

151 +dv%z * gradW_i%z)

152 f(i)%rotv%x = f(i)%rotv%x - mj * (dv%y * gradW_i%z - dv%z *

gradW_i%y)

153 f(i)%rotv%y = f(i)%rotv%y - mj * (dv%z * gradW_i%x - dv%x *

gradW_i%z)

154 f(i)%rotv%z = f(i)%rotv%z - mj * (dv%x * gradW_i%y - dv%y *

gradW_i%x)

155 end do

156 f(i)%divv = f(i)%divv / f(i)%dens

157 f(i)%rotv%x = f(i)%rotv%x / f(i)%dens

158 f(i)%rotv%y = f(i)%rotv%y / f(i)%dens

159 f(i)%rotv%z = f(i)%rotv%z / f(i)%dens

160 #endif

161 end do

162 #endif

7.1.4.3 圧力勾配加速度計算

圧力勾配加速度用の相互作用関数は サブルーチン calc hydro force として実装されてい
る。Listing 61に、実装を示す。このサブルーチンでは、式 (10)、(12)、(13)の右辺の計算、
及び、Springel [2005, MNRAS, 364, 1105]の式 (16)に従って dtの計算を行っている (dtに
ついては 派生データ型 fp sph の説明を参照のこと)。

Listing 61: 相互作用関数 (圧力勾配加速度計算用)

1 !**** Interaction function

2 subroutine calc_hydro_force(ep_i ,n_ip ,ep_j ,n_jp ,f) bind(c)

3 integer(kind=c_int), intent(in), value :: n_ip ,n_jp

4 type(ep_hydro), dimension(n_ip), intent(in) :: ep_i

5 type(ep_hydro), dimension(n_jp), intent(in) :: ep_j

6 type(force_hydro), dimension(n_ip), intent(inout) :: f

7 !* Local variables

8 integer(kind=c_int) :: i,j

9 real(kind=c_double) :: mass_i ,mass_j ,smth_i ,smth_j , &

10 dens_i ,dens_j ,pres_i ,pres_j , &

11 gradh_i ,gradh_j ,balsw_i ,balsw_j , &

12 snds_i ,snds_j

13 real(kind=c_double) :: povrho2_i ,povrho2_j , &

14 v_sig_max ,dr_dv ,w_ij ,v_sig ,AV

15 type(fdps_f64vec) :: pos_i ,pos_j ,vel_i ,vel_j , &

16 dr ,dv ,gradW_i ,gradW_j ,gradW_ij

17 do i=1,n_ip

18 !* Zero -clear

19 v_sig_max = 0.0d0

20 !* Extract i-particle info.

21 pos_i = ep_i(i)%pos

22 vel_i = ep_i(i)%vel

23 mass_i = ep_i(i)%mass

24 smth_i = ep_i(i)%smth

25 dens_i = ep_i(i)%dens

26 pres_i = ep_i(i)%pres

27 gradh_i = ep_i(i)%gradh

28 balsw_i = ep_i(i)%balsw

29 snds_i = ep_i(i)%snds

30 povrho2_i = pres_i /( dens_i*dens_i)

31 do j=1,n_jp

32 !* Extract j-particle info.

33 pos_j%x = ep_j(j)%pos%x

34 pos_j%y = ep_j(j)%pos%y

35 pos_j%z = ep_j(j)%pos%z

36 vel_j%x = ep_j(j)%vel%x

37 vel_j%y = ep_j(j)%vel%y

38 vel_j%z = ep_j(j)%vel%z

39 mass_j = ep_j(j)%mass

40 smth_j = ep_j(j)%smth

41 dens_j = ep_j(j)%dens

42 pres_j = ep_j(j)%pres

43 gradh_j = ep_j(j)%gradh

44 balsw_j = ep_j(j)%balsw

45 snds_j = ep_j(j)%snds

46 povrho2_j = pres_j /( dens_j*dens_j)

47 !* Compute dr & dv

48 dr%x = pos_i%x - pos_j%x

49 dr%y = pos_i%y - pos_j%y

50 dr%z = pos_i%z - pos_j%z

51 dv%x = vel_i%x - vel_j%x

52 dv%y = vel_i%y - vel_j%y

53 dv%z = vel_i%z - vel_j%z

54 !* Compute the signal velocity

55 dr_dv = dr%x * dv%x + dr%y * dv%y + dr%z * dv%z

56 if (dr_dv < 0.0d0) then

57 w_ij = dr_dv / sqrt(dr%x * dr%x + dr%y * dr%y + dr%z * dr%z

)

58 else

59 w_ij = 0.0d0

60 end if

61 v_sig = snds_i + snds_j - 3.0d0 * w_ij

62 v_sig_max = max(v_sig_max , v_sig)

63 !* Compute the artificial viscosity

64 AV = - 0.5d0*v_sig*w_ij / (0.5d0*( dens_i+dens_j)) * 0.5d0*(

balsw_i+balsw_j)

65 !* Compute the average of the gradients of kernel

66 gradW_i = gradW(dr,smth_i)

67 gradW_j = gradW(dr,smth_j)

68 gradW_ij%x = 0.5d0 * (gradW_i%x + gradW_j%x)

69 gradW_ij%y = 0.5d0 * (gradW_i%y + gradW_j%y)

70 gradW_ij%z = 0.5d0 * (gradW_i%z + gradW_j%z)

71 !* Compute the acceleration and the heating rate

72 f(i)%acc%x = f(i)%acc%x - mass_j *( gradh_i * povrho2_i *

gradW_i%x &

73 +gradh_j * povrho2_j *

gradW_j%x &

74 +AV * gradW_ij%x)

75 f(i)%acc%y = f(i)%acc%y - mass_j *( gradh_i * povrho2_i *

gradW_i%y &

76 +gradh_j * povrho2_j *

gradW_j%y &

77 +AV * gradW_ij%y)

78 f(i)%acc%z = f(i)%acc%z - mass_j *( gradh_i * povrho2_i *

gradW_i%z &

79 +gradh_j * povrho2_j *

gradW_j%z &

80 +AV * gradW_ij%z)

81 f(i)%eng_dot = f(i)%eng_dot

&

82 + mass_j * gradh_i * povrho2_i * (dv%x * gradW_i%

x &

83 +dv%y * gradW_i%

y &

84 +dv%z * gradW_i%

z) &

85 + mass_j * 0.5d0 * AV * (dv%x * gradW_ij%x

&

86 +dv%y * gradW_ij%y

&

87 +dv%z * gradW_ij%z)

88 f(i)%ent_dot = f(i)%ent_dot &

89 + 0.5 * mass_j * AV * (dv%x * gradW_ij%x &

90 +dv%y * gradW_ij%y &

91 +dv%z * gradW_ij%z)

92 end do

93 f(i)%ent_dot = f(i)%ent_dot &

94 * (specific_heat_ratio - 1.0d0) &

95 / dens_i **( specific_heat_ratio - 1.0d0)

96 f(i)%dt = CFL_hydro *2.0d0*smth_i/v_sig_max

97 end do

7.1.5 プログラム本体

本節では、主に f_main.F90に実装されたサンプルコード本体について解説を行う。詳細
な説明に入る前に、サンプルコードの内容と全体構造について説明を与える。7.1節冒頭で
述べたように、このサンプルコードでは円盤銀河のN 体/SPHシミュレーションを行うもの
であるが、初期条件としては円盤銀河の他、簡単なテスト計算用の初期条件も用意されてい
る。具体的に以下の 4つの場合に対応している:

(a) 円盤銀河用の初期条件。この初期条件はコンパイルオプション時に-DINITIAL CONDI-

TION=0が指定された場合に選択される。初期条件作成は ic.F90のサブルーチンgalaxy -

ICで行われる。ダークマターと星の分布は事前にMAGIで作成されたファイルを読み
込んで設定される。一方、ガスの初期分布はこのサブルーチン内部で生成される。デ
フォルトでは粒子数 218で exponential disk (M = 1010 M⊙, Rs = 7 kpc [scale radius],

Rt = 12.5 kpc [truncation radius], zd = 0.4 kpc [scale height], zt = 1 kpc [truncation

height])が生成される。
(b) Cold collapse問題用の初期条件。この初期条件はコンパイルオプション時に-DINITIAL

CONDITION=1が指定された場合に選択される。初期条件作成は ic.F90 のサブルーチン
cold collapse test ICで行われる。

(c) Evrard test (Evrard [1988,MNRAS,235,911]の第 3.3節)用の初期条件。この初期条件
はコンパイルオプション時に-DINITIAL CONDITION=2が指定された場合に選択される。
初期条件作成は ic.F90 のサブルーチン Evrard test ICで行われる。作成方法は 2つ
あり、サブルーチンの最後の引数の値を手動で 0か 1にして指定する。0の場合、格子
状に並んだ SPH粒子からEvrard球の密度分布を作成する。1の場合、ガラス状に分布
した SPH粒子からEvrard球の密度分布を作成する。1を選択するためには、事前に次
項で説明するモードで SPH粒子のデータを作成しておく必要がある。

(d) [−1, 1)3の立方体中に一様密度のガラス状の SPH粒子分布を作成するための初期条件/

動作モード。この初期条件はコンパイルオプション時に-DINITIAL CONDITION=3が指
定された場合に選択される。初期条件作成は ic.F90 のサブルーチン make glass ICで
行われる。

コード全体の構造は以下のようになっている:

(1) FDPSで使用するオブジェクトの生成と初期化
(2) (必要であれば)Phantom-GRAPEライブラリの初期化
(3) 初期条件ファイルの読み込み、或いは、初期条件の作成
(4) 終了時刻まで粒子の運動を計算

以下で、個々について詳しく説明を行う。

7.1.5.1 fdps controller型オブジェクトの生成

ユーザは FDPSのAPIを使用するために、FDPS controller型オブジェクトを生成しな
ければならない。本サンプルコードでは、FDPS controller型オブジェクト fdps ctrlをメ
インルーチンで生成している:

Listing 62: fdps controller型オブジェクトの生成
1 subroutine f_main ()

2 use fdps_module

3 implicit none

4 !* Local variables

5 type(fdps_controller) :: fdps_ctrl

6

7 ! Do something

8

9 end subroutine f_main

ここに示したコードは実際にサンプルコードから必要な部分だけを取り出したものであるこ
とに注意して頂きたい。上記の理由から、以下の説明において、FDPSの APIはこのオブ
ジェクトのメンバ関数として呼び出されていることに注意されたい。

7.1.5.2 開始、終了

まずは、FDPSの初期化/開始を行う必要がある。次のように、メインルーチンに記述する。

Listing 63: FDPSの開始
1 call fdps_ctrl%ps_initialize ();

FDPSは、開始したら明示的に終了させる必要がある。今回は、プログラムの終了と同時
に FDPSも終了させるため、メインルーチンの最後に次のように記述する。

Listing 64: FDPSの終了
1 call fdps_ctrl%ps_finalize ();

7.1.5.3 オブジェクトの生成と初期化

FDPSの初期化に成功した場合、ユーザーはコード中で用いるオブジェクトを作成する必
要がある。本節では、オブジェクトの生成/初期化の仕方について、解説する。

7.1.5.3.1 粒子群オブジェクトの生成と初期化

本サンプルコードでは、N体粒子と SPH粒子のデータを異なる粒子群オブジェクトを用い
て管理する。 2つの整数 psys num nbodyと psys num sphは、それぞれ、N体粒子と SPH

粒子の粒子群オブジェクトの識別番号を格納する変数である。これら 2つの整数を使い、粒
子群オブジェクトを生成・初期化を以下のように行っている。

Listing 65: 粒子群オブジェクトの生成・初期化
1 call fdps_ctrl%create_psys(psys_num_nbody ,’fp_nbody ’)

2 call fdps_ctrl%init_psys(psys_num_nbody)

3 call fdps_ctrl%create_psys(psys_num_sph ,’fp_sph ’)

4 call fdps_ctrl%init_psys(psys_num_sph)

7.1.5.3.2 領域情報オブジェクトの生成と初期化

本サンプルコードでは、計算領域の分割を、N 体粒子と SPH粒子を合わせた粒子全体が
等分割されるように行うこととする。この場合、必要な領域情報オブジェクトは 1つである。
したがって、本コードでは領域情報オブジェクトの識別番号を格納する整数変数 dinfo num

を用意し、それを用いて生成・初期化を次のように行っている。

Listing 66: 領域情報オブジェクトの生成・初期化
1 call fdps_ctrl%create_dinfo(dinfo_num)

2 call fdps_ctrl%init_dinfo(dinfo_num ,coef_ema)

7.1.5.3.3 ツリーオブジェクトの生成と初期化

本サンプルコードでは、重力計算用、密度計算、圧力勾配加速度計算のそれぞれに 1つず
つツリーを用意している。ツリーオブジェクトの初期化の際には、API init tree の第 2

引数に計算で使用する大雑把な粒子数を渡す必要がある。重力計算用のツリーオブジェクト
(変数 tree num gravを介して制御される) では、ローカル粒子数の 3倍の値を渡している。
一方、密度計算と圧力勾配加速度計算に使用されるツリーオブジェクト (それぞれ変数 tree

num densと tree num hydroを介して制御される) では、ローカルの SPH粒子数の 3倍の値
を渡している。

Listing 67: ツリーオブジェクトの生成・初期化
1 nptcl_loc_sph = max(fdps_ctrl%get_nptcl_loc(psys_num_sph) ,1)

2 nptcl_loc_nbody = fdps_ctrl%get_nptcl_loc(psys_num_nbody)

3 nptcl_loc_all = nptcl_loc_nbody + nptcl_loc_sph

4 !** tree for gravity calculation

5 call fdps_ctrl%create_tree(tree_num_grav , &

6 "Long ,force_grav ,ep_grav ,ep_grav ,Monopole")

7 call fdps_ctrl%init_tree(tree_num_grav , 3* nptcl_loc_all , theta , &

8 n_leaf_limit , n_group_limit)

9 !** tree for the density calculation

10 call fdps_ctrl%create_tree(tree_num_dens , &

11 "Short ,force_dens ,ep_hydro ,ep_hydro ,Gather")

12 call fdps_ctrl%init_tree(tree_num_dens , 3* nptcl_loc_sph , theta , &

13 n_leaf_limit , n_group_limit)

14

15 !** tree for the hydrodynamic force calculation

16 call fdps_ctrl%create_tree(tree_num_hydro , &

17 "Short ,force_hydro ,ep_hydro ,ep_hydro ,

Symmetry")

18 call fdps_ctrl%init_tree(tree_num_hydro , 3* nptcl_loc_sph , theta , &

19 n_leaf_limit , n_group_limit)

7.1.5.4 初期条件の設定

初期条件の設定は サブルーチン setup IC で行われる。このサブルーチンはマクロ INI-

TIAL CONDITIONの値に応じて、内部でさらに別のサブルーチンを呼び出しており、呼び出
されるサブルーチンとマクロの値の対応は、既に述べた通りである。引数の time dump, dt

dump, time endは、データ出力の最初の時刻、出力時間間隔、シミュレーション終了時間を
表す変数であり、個々の初期条件作成関数の中で設定すべきものである。また、境界条件、
重力ソフトニングの値 (eps grav)、系に許される最大の時間刻み (dt max)も設定する必要
がある (dt maxに関しては必ずしも設定する必要はない)。

Listing 68: 初期条件の設定
1 call setup_IC(psys_num_nbody , psys_num_sph , dinfo_num , &

2 time_dump , dt_dump , time_end)

以下、円盤銀河の初期条件を設定する サブルーチン galaxy IC について、留意事項を述
べておく。

• MAGIが作成する粒子データはMAGIのコード内単位系で出力される。単位系の情報
はMAGIを実行したときに出力されるファイル doc/unit.txtに記述されている。こ
のファイルに記載された単位質量、単位長さ、単位時間の値と、定数 magi unit mass,

magi unit leng, magi unit timeは一致させなければならない。

• 関数が読み込むファイルは ./magi_data/dat/Galaxy.tipsyである。別なファイルを
読み込ませたい場合、手動でソースコードを変更する必要がある。

• 関数が生成するガス分布はR (≡
√
x2 + y2)方向と z方向に exponential な密度分布を

持つガス円盤である。それぞれの方向のスケール長が変数 Rs, zdで、分布を打ち切る
距離は変数 Rt, ztである。

• 初期のガスの熱力学的状態はガス温度 tempと水素原子に対する平均分子量 muを与え
て指定する。コンパイル時マクロ USE ENTROPYが定義済み/未定義に関わらず、粒子の
熱力学的状態は単位質量あたりの内部エネルギーとして与える必要がある (fp sphの
メンバ変数 eng)。USE ENTROPYが定義済みの場合、メインルーチン f_main()で呼び
出されている サブルーチン set entropy によって、計算された密度と内部エネルギー
の初期値から初期エントロピーが自動的に決定される。未定義の場合、ここで設定し
た engの値がそのまま内部エネルギーの初期値となる。

7.1.5.5 領域分割の実行

複数の粒子種がある場合に、これらを合わせた粒子分布に基づいて領域分割を実行する
には、領域情報オブジェクト用の 2つの API collect sample particle と decompose -

domainを併用する必要がある。まず、API collect sample particleでそれぞれの粒子群
オブジェクトからサンプル粒子を集める。こ

粒子群オブジェクトに対して、このAPIの呼び出しが終わったら、API decompose domain

で領域分割を実行する。

Listing 69: 領域分割の実行
1 call fdps_ctrl%collect_sample_particle(dinfo_num , psys_num_nbody , clear)

2 call fdps_ctrl%collect_sample_particle(dinfo_num , psys_num_sph , unclear)

3 call fdps_ctrl%decompose_domain(dinfo_num)

7.1.5.6 粒子交換の実行

先程計算した領域情報に基いてプロセス間の粒子の情報を交換するには、粒子群オブジェ
クト用API exchange particle を使用する:

Listing 70: 粒子交換の実行
1 call fdps_ctrl%exchange_particle(psys_num_nbody ,dinfo_num)

2 call fdps_ctrl%exchange_particle(psys_num_sph ,dinfo_num)

7.1.5.7 相互作用計算の実行

領域分割・粒子交換が完了したら、計算開始時の加速度を決定するため、相互作用計算を
行う必要がある。以下に、本サンプルコードにおける初期条件作成後最初の相互作用計算の
実装を示す。最初に重力計算をし、その後、密度計算・圧力勾配加速度計算を行っている。

Listing 71: 相互作用計算の実行
1 !** Gravity calculation

2 t_start = fdps_ctrl%get_wtime ()

3 #if defined(ENABLE_GRAVITY_INTERACT)

4 call fdps_ctrl%set_particle_local_tree(tree_num_grav , psys_num_nbody)

5 call fdps_ctrl%set_particle_local_tree(tree_num_grav , psys_num_sph ,

unclear)

6 pfunc_ep_ep = c_funloc(calc_gravity_ep_ep)

7 pfunc_ep_sp = c_funloc(calc_gravity_ep_sp)

8 call fdps_ctrl%calc_force_making_tree(tree_num_grav , &

9 pfunc_ep_ep , &

10 pfunc_ep_sp , &

11 dinfo_num)

12 nptcl_loc_nbody = fdps_ctrl%get_nptcl_loc(psys_num_nbody)

13 call fdps_ctrl%get_psys_fptr(psys_num_nbody , ptcl_nbody)

14 do i=1, nptcl_loc_nbody

15 call fdps_ctrl%get_force(tree_num_grav , i, f_grav)

16 ptcl_nbody(i)%acc%x = f_grav%acc%x

17 ptcl_nbody(i)%acc%y = f_grav%acc%y

18 ptcl_nbody(i)%acc%z = f_grav%acc%z

19 ptcl_nbody(i)%pot = f_grav%pot

20 end do

21 offset = nptcl_loc_nbody

22 nptcl_loc_sph = fdps_ctrl%get_nptcl_loc(psys_num_sph)

23 call fdps_ctrl%get_psys_fptr(psys_num_sph , ptcl_sph)

24 do i=1, nptcl_loc_sph

25 call fdps_ctrl%get_force(tree_num_grav , i + offset , f_grav)

26 ptcl_sph(i)%acc_grav%x = f_grav%acc%x

27 ptcl_sph(i)%acc_grav%y = f_grav%acc%y

28 ptcl_sph(i)%acc_grav%z = f_grav%acc%z

29 ptcl_sph(i)%pot_grav = f_grav%pot

30 end do

31 #endif

32 t_grav = fdps_ctrl%get_wtime () - t_start

33 !** SPH calculations

34 t_start = fdps_ctrl%get_wtime ()

35 #if defined(ENABLE_HYDRO_INTERACT)

36 call calc_density_wrapper(psys_num_sph , dinfo_num , tree_num_dens)

37 call set_entropy(psys_num_sph)

38 call set_pressure(psys_num_sph)

39 pfunc_ep_ep = c_funloc(calc_hydro_force)

40 call fdps_ctrl%calc_force_all_and_write_back(tree_num_hydro , &

41 pfunc_ep_ep , &

42 psys_num_sph , &

43 dinfo_num)

44 #endif

45 t_hydro = fdps_ctrl%get_wtime () - t_start

まず重力計算の方法について説明する。重力計算は、N 体粒子と SPH粒子の両方が関わ
る。このような複数の粒子種の間で 1つの相互作用計算を行うには、ツリーオブジェクト用
のAPI set particle local tree と calc force making tree を合わせて使用する必要が
ある。まず、各粒子群オブジェクトに対して、API set particle local treeを使って、粒
子情報をツリーオブジェクトに渡す。こ

重力計算に関係するすべての粒子群オブジェクトに対して、このAPIの呼び

出しが完了したら、API calc force making tree で相互作用計算を行う。相互作用計算の
結果を取得するためには、API get force を使う。この APIは引数に整数 iを取り、API

set particle local treeで i番目に読み込んだ粒子が受ける相互作用を返す。したがって、
2種類目以降の粒子種の相互作用の結果を取得する場合、適切にオフセット値を指定する必
要があることに注意されたい。
次に密度計算と圧力勾配加速度計算について説明する。これらの計算は 1粒子種しか関
わらないため、本チュートリアルでこれまで使ってきた API calc force all and write

back が使用できる。圧力勾配加速度に関しては、 サブルーチン f main 内でこのAPIを直
接呼び出している。一方、密度計算は、第 7.1.4節でも述べた通り、ρiとhiのイテレーション
計算が収束しなかったときのための対処が必要であり、これをサブルーチン calc density

wrapperの中で行っている。実装は次のようになっている。実装はマクロENABLE_VARIABLE_

SMOOTHING_LENGTHが定義済みか未定義かで分岐しており、未定義の場合には固定長カーネ
ルの SPHコードとなるので、単に、API calc force all and write back を 1回だけ実行
している。一方、上記マクロが定義済みの場合、すべての粒子の ρiと hiが無矛盾に決定さ
れるまで、API calc force all and write back を繰り返し実行する。各粒子が収束した
かの情報は派生データ型 fp sphのメンバ変数 flagに格納されており、値が 1のときに収束
していることを示す。flagが 1を取る粒子数が全 SPH粒子数に一致したときに計算を終わ
らせている。

Listing 72: サブルーチン calc density wrapperの実装
1 subroutine calc_density_wrapper(psys_num ,dinfo_num ,tree_num)

2 use fdps_vector

3 use fdps_module

4 use user_defined_types

5 implicit none

6 integer , intent(in) :: psys_num ,dinfo_num ,tree_num

7 !* Local variables

8 integer :: i,nptcl_loc ,nptcl_glb

9 integer :: n_compl_loc ,n_compl

10 type(fdps_controller) :: fdps_ctrl

11 type(fp_sph), dimension (:), pointer :: ptcl

12 type(c_funptr) :: pfunc_ep_ep

13

14 #if defined(ENABLE_VARIABLE_SMOOTHING_LENGTH)

15 nptcl_loc = fdps_ctrl%get_nptcl_loc(psys_num)

16 nptcl_glb = fdps_ctrl%get_nptcl_glb(psys_num)

17 call fdps_ctrl%get_psys_fptr(psys_num , ptcl)

18 pfunc_ep_ep = c_funloc(calc_density)

19 ! Determine the density and the smoothing length

20 ! so that Eq.(6) in Springel (2005) holds within a specified accuracy.

21 do

22 ! Increase smoothing length

23 do i=1, nptcl_loc

24 ptcl(i)%smth = scf_smth * ptcl(i)%smth

25 end do

26 ! Compute density , etc.

27 call fdps_ctrl%calc_force_all_and_write_back(tree_num , &

28 pfunc_ep_ep , &

29 psys_num , &

30 dinfo_num)

31 ! Check convergence

32 n_compl_loc = 0; n_compl = 0

33 do i=1, nptcl_loc

34 if (ptcl(i)%flag == 1) n_compl_loc = n_compl_loc + 1

35 end do

36 call fdps_ctrl%get_sum(n_compl_loc , n_compl)

37 if (n_compl == nptcl_glb) exit

38 end do

39 !* Release the pointer

40 nullify(ptcl)

41 #else

42 pfunc_ep_ep = c_funloc(calc_density)

43 call fdps_ctrl%calc_force_all_and_write_back(tree_num , &

44 pfunc_ep_ep , &

45 psys_num , &

46 dinfo_num)

47 #endif

48

49 end subroutine calc_density_wrapper

サブルーチン set entropy は、初期条件作成後 1回だけ呼び出されるサブルーチンで、エ
ントロピーの初期値をセットする。式 (8)から、エントロピーを計算するには初期密度が必要
である。そのため、サブルーチン calc density wrapper の後に配置されている。サブルー
チン set entropy では、計算された密度と uiの初期値を使って、エントロピーをセットす
る。これ以降は、エントロピーが独立変数となる。

7.1.5.8 時間積分ループ

本サンプルコードでは、時間積分を Leapfrog時間積分法によって行っている (この方法に
関しては、第 4.1.3.5.4節を参照されたい)。粒子位置を時間推進するD(·)オペレータはサブ
ルーチン full drift、粒子速度を時間推進するK(·)オペレータはサブルーチン initial

kick, final kickとして実装されている。

8 ユーザーサポート

FDPSを使用したコード開発に関する相談は fdps-support<at>mail.jmlab.jp で受け付け
ています (<at>は@に変更お願い致します)。以下のような場合は各項目毎の対応をお願いし
ます。

8.1 コンパイルできない場合

ユーザーには以下の情報提供をお願いします。

• コンパイル環境

• コンパイル時に出力されるエラーメッセージ

• ソースコード (可能ならば)

8.2 コードがうまく動かない場合

ユーザーには以下の情報提供をお願いします。

• 実行環境

• 実行時に出力されるエラーメッセージ

• ソースコード (可能ならば)

8.3 その他

思い通りの性能がでない場合やその他の相談なども、上のメールアドレスにお知らせくだ
さい。

9 ライセンス

MITライセンスに準ずる。標準機能のみ使用する場合は、Iwasawa et al. (PASJ, 68, 54)、
Namekata et al. (PASJ, 70, 70)の引用をお願いします。
拡張機能のParticle MeshクラスはGreeMコード (開発者：石山智明、似鳥啓吾) (Ishiyama,

Fukushige & Makino 2009, Publications of the Astronomical Society of Japan, 61, 1319;

Ishiyama, Nitadori & Makino, 2012 SC’12 Proceedings of the International Conference on

High Performance Computing, Networking Stroage and Analysis, No. 5)のモジュールを使用
している。GreeMコードはYoshikawa & Fukushige (2005, Publications of the Astronomical

Society of Japan, 57, 849)で書かれたコードをベースとしている。Particle Meshクラスを使
用している場合は、上記３つの文献の引用をお願いします。
拡張機能のうち x86版 Phantom-GRAPEを使用する場合は Tanikawa et al.(2012, New

Astronomy, 17, 82)とTanikawa et al.(2012, New Astronomy, 19, 74)の引用をお願いします。

Copyright (c) <2015-> <FDPS development team>

Permission is hereby granted, free of charge, to any person obtaining a copy of this soft-

ware and associated documentation files (the “Software”), to deal in the Software without

restriction, including without limitation the rights to use, copy, modify, merge, publish, dis-

tribute, sublicense, and/or sell copies of the Software, and to permit persons to whom the

Software is furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all copies or

substantial portions of the Software.

THE SOFTWARE IS PROVIDED “AS IS”, WITHOUT WARRANTY OF ANY KIND,

EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES

OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONIN-

FRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS

BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN

AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR

IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN

THE SOFTWARE.

115


	1 変更記録
	2 概要
	3 入門：サンプルコードを動かしてみよう
	3.1 動作環境
	3.2 必要なソフトウェア
	3.2.1 標準機能
	3.2.1.1 逐次処理
	3.2.1.2 並列処理
	3.2.1.2.1 OpenMP
	3.2.1.2.2 MPI
	3.2.1.2.3 MPI+OpenMP


	3.2.2 拡張機能
	3.2.2.1 Particle Mesh


	3.3 インストール
	3.3.1 取得方法
	3.3.1.1 最新バージョン
	3.3.1.2 過去のバージョン

	3.3.2 インストール方法

	3.4 サンプルコードの使用方法
	3.4.1 重力N体シミュレーションコード
	3.4.1.1 概要
	3.4.1.2 ディレクトリ移動
	3.4.1.3 Makefileの編集
	3.4.1.4 makeの実行
	3.4.1.5 実行
	3.4.1.6 結果の解析
	3.4.1.7 x86版Phantom-GRAPEを使う場合
	3.4.1.8 PIKGを使う場合

	3.4.2 SPHシミュレーションコード
	3.4.2.1 概要
	3.4.2.2 ディレクトリ移動
	3.4.2.3 Makefileの編集
	3.4.2.4 makeの実行
	3.4.2.5 実行
	3.4.2.6 結果の解析



	4 サンプルコードの解説
	4.1 N体シミュレーションコード
	4.1.1 ソースファイルの場所と構成
	4.1.2 ユーザー定義型・ユーザ定義関数
	4.1.2.1 FullParticle型
	4.1.2.2 相互作用関数 calcForceEpEp
	4.1.2.3 相互作用関数 calcForceEpSp

	4.1.3 プログラム本体
	4.1.3.1 fdps_controller型オブジェクトの生成
	4.1.3.2 開始、終了
	4.1.3.3 オブジェクトの生成・初期化
	4.1.3.3.1 オブジェクトの生成
	4.1.3.3.2 領域情報オブジェクトの初期化
	4.1.3.3.3 粒子群オブジェクトの初期化
	4.1.3.3.4 ツリーオブジェクトの初期化

	4.1.3.4 粒子データの初期化
	4.1.3.5 ループ
	4.1.3.5.1 領域分割の実行
	4.1.3.5.2 粒子交換の実行
	4.1.3.5.3 相互作用計算の実行
	4.1.3.5.4 時間積分

	4.1.3.6 粒子データの更新

	4.1.4 ログファイル

	4.2 固定長SPHシミュレーションコード
	4.2.1 ソースファイルの場所と構成
	4.2.2 ユーザー定義型・ユーザ定義関数
	4.2.2.1 FullParticle型
	4.2.2.2 EssentialParticleI型
	4.2.2.3 Force型
	4.2.2.4 相互作用関数 calcForceEpEp

	4.2.3 プログラム本体
	4.2.3.1 fdps_controller型オブジェクトの生成
	4.2.3.2 開始、終了
	4.2.3.3 オブジェクトの生成・初期化
	4.2.3.3.1 オブジェクトの生成
	4.2.3.3.2 領域情報オブジェクトの初期化
	4.2.3.3.3 粒子群オブジェクトの初期化
	4.2.3.3.4 ツリーオブジェクトの初期化

	4.2.3.4 ループ
	4.2.3.4.1 領域分割の実行
	4.2.3.4.2 粒子交換の実行
	4.2.3.4.3 相互作用計算の実行


	4.2.4 コンパイル
	4.2.5 実行
	4.2.6 ログファイル
	4.2.7 可視化


	5 サンプルコード
	5.1 N体シミュレーション
	5.2 固定長SPHシミュレーション

	6 拡張機能の解説
	6.1 P3Mコード
	6.1.1 サンプルコードの場所と作業ディレクトリ
	6.1.2 ユーザー定義型
	6.1.2.1 FullParticle型
	6.1.2.2 EssentialParticleI型
	6.1.2.3 Force型
	6.1.2.4 相互作用関数 calcForceEpEp
	6.1.2.5 相互作用関数 calcForceEpSp

	6.1.3 プログラム本体
	6.1.3.1 fdps_controller型オブジェクトの生成
	6.1.3.2 開始、終了
	6.1.3.3 オブジェクトの生成と初期化
	6.1.3.3.1 オブジェクトの生成
	6.1.3.3.2 オブジェクトの初期化

	6.1.3.4 粒子分布の生成
	6.1.3.4.1 領域分割の実行
	6.1.3.4.2 粒子交換の実行

	6.1.3.5 相互作用計算の実行
	6.1.3.6 エネルギー相対誤差の計算

	6.1.4 コンパイル
	6.1.5 実行
	6.1.6 結果の確認


	7 より実用的なアプリケーションの解説
	7.1 N体/SPHコード
	7.1.1 コードの使用方法
	7.1.1.1 ディレクトリ移動
	7.1.1.2 サンプルコードのファイル構成
	7.1.1.3 Makefileの編集
	7.1.1.4 MAGIを使った粒子データの生成
	7.1.1.5 makeの実行
	7.1.1.6 実行
	7.1.1.7 結果の解析

	7.1.2 Springelの方法
	7.1.3 ユーザー定義型
	7.1.3.1 FullParticle型
	7.1.3.2 EssentialParticle型
	7.1.3.3 Force型

	7.1.4 相互作用関数
	7.1.4.1 重力計算
	7.1.4.2 密度計算
	7.1.4.3 圧力勾配加速度計算

	7.1.5 プログラム本体
	7.1.5.1 fdps_controller型オブジェクトの生成
	7.1.5.2 開始、終了
	7.1.5.3 オブジェクトの生成と初期化
	7.1.5.3.1 粒子群オブジェクトの生成と初期化
	7.1.5.3.2 領域情報オブジェクトの生成と初期化
	7.1.5.3.3 ツリーオブジェクトの生成と初期化

	7.1.5.4 初期条件の設定
	7.1.5.5 領域分割の実行
	7.1.5.6 粒子交換の実行
	7.1.5.7 相互作用計算の実行
	7.1.5.8 時間積分ループ



	8 ユーザーサポート
	8.1 コンパイルできない場合
	8.2 コードがうまく動かない場合
	8.3 その他

	9 ライセンス

